

# Introduction to Ada

Raphaël Amiard  
Gustavo A. Hoffmann

**LEARN.**  
ADACORE.COM



# **Introduction to Ada**

***Release 2025-12***

**Raphaël Amiard  
and Gustavo A. Hoffmann**

**Dec 27, 2025**



# CONTENTS:

<b>1</b>	<b>Introduction</b>	<b>3</b>
1.1	History . . . . .	3
1.2	Ada today . . . . .	3
1.3	Philosophy . . . . .	4
1.4	SPARK . . . . .	4
<b>2</b>	<b>Imperative language</b>	<b>5</b>
2.1	Hello world . . . . .	5
2.2	Imperative language - If/Then/Else . . . . .	6
2.3	Imperative language - Loops . . . . .	9
2.3.1	For loops . . . . .	9
2.3.2	Bare loops . . . . .	10
2.3.3	While loops . . . . .	11
2.4	Imperative language - Case statement . . . . .	12
2.5	Imperative language - Declarative regions . . . . .	14
2.6	Imperative language - conditional expressions . . . . .	16
2.6.1	If expressions . . . . .	16
2.6.2	Case expressions . . . . .	17
<b>3</b>	<b>Subprograms</b>	<b>19</b>
3.1	Subprograms . . . . .	19
3.1.1	Subprogram calls . . . . .	20
3.1.2	Nested subprograms . . . . .	21
3.1.3	Function calls . . . . .	22
3.2	Parameter modes . . . . .	24
3.3	Subprogram calls . . . . .	24
3.3.1	In parameters . . . . .	24
3.3.2	In out parameters . . . . .	25
3.3.3	Out parameters . . . . .	26
3.3.4	Forward declaration of subprograms . . . . .	27
3.4	Renaming . . . . .	27
<b>4</b>	<b>Modular programming</b>	<b>31</b>
4.1	Packages . . . . .	31
4.2	Using a package . . . . .	33
4.3	Package body . . . . .	33
4.4	Child packages . . . . .	35
4.4.1	Child of a child package . . . . .	37
4.4.2	Multiple children . . . . .	38
4.4.3	Visibility . . . . .	39
4.5	Renaming . . . . .	41
<b>5</b>	<b>Strongly typed language</b>	<b>43</b>
5.1	What is a type? . . . . .	43
5.2	Integers . . . . .	43


5.2.1 Operational semantics . . . . .	45
5.3 Unsigned types . . . . .	46
5.4 Enumerations . . . . .	47
5.5 Floating-point types . . . . .	48
5.5.1 Basic properties . . . . .	48
5.5.2 Precision of floating-point types . . . . .	49
5.5.3 Range of floating-point types . . . . .	50
5.6 Strong typing . . . . .	52
5.7 Derived types . . . . .	54
5.8 Subtypes . . . . .	56
5.8.1 Subtypes as type aliases . . . . .	58
<b>6 Records</b>	<b>61</b>
6.1 Record type declaration . . . . .	61
6.2 Aggregates . . . . .	62
6.3 Component selection . . . . .	62
6.4 Renaming . . . . .	63
<b>7 Arrays</b>	<b>67</b>
7.1 Array type declaration . . . . .	67
7.2 Indexing . . . . .	70
7.3 Simpler array declarations . . . . .	71
7.4 Range attribute . . . . .	72
7.5 Unconstrained arrays . . . . .	74
7.6 Predefined array type: String . . . . .	75
7.7 Restrictions . . . . .	77
7.8 Returning unconstrained arrays . . . . .	78
7.9 Declaring arrays (2) . . . . .	79
7.10 Array slices . . . . .	80
7.11 Renaming . . . . .	81
<b>8 More about types</b>	<b>85</b>
8.1 Aggregates: A primer . . . . .	85
8.2 Overloading and qualified expressions . . . . .	86
8.3 Character types . . . . .	89
<b>9 Access types (pointers)</b>	<b>91</b>
9.1 Overview . . . . .	91
9.2 Allocation (by type) . . . . .	93
9.3 Dereferencing . . . . .	94
9.4 Other features . . . . .	94
9.5 Mutually recursive types . . . . .	95
<b>10 More about records</b>	<b>97</b>
10.1 Dynamically sized record types . . . . .	97
10.2 Records with discriminant . . . . .	98
10.3 Variant records . . . . .	100
<b>11 Fixed-point types</b>	<b>103</b>
11.1 Decimal fixed-point types . . . . .	103
11.1.1 Decimal delta . . . . .	103
11.1.2 Decimal digits . . . . .	104
11.1.3 Decimal delta and digits . . . . .	106
11.1.4 Requirements for the delta . . . . .	107
11.2 Ordinary fixed-point types . . . . .	107
11.3 Fixed-point vs. floating-point types . . . . .	108
<b>12 Privacy</b>	<b>111</b>
12.1 Basic encapsulation . . . . .	111

12.2 Abstract data types . . . . .	112
12.3 Limited types . . . . .	114
12.4 Child packages & privacy . . . . .	115
<b>13 Generics</b>	<b>121</b>
13.1 Introduction . . . . .	121
13.2 Formal type declaration . . . . .	121
13.3 Formal object declaration . . . . .	122
13.4 Generic body definition . . . . .	122
13.5 Generic instantiation . . . . .	123
13.6 Generic packages . . . . .	124
13.7 Formal subprograms . . . . .	126
13.8 Example: I/O instances . . . . .	127
13.9 Example: ADTs . . . . .	129
13.10 Example: Swap . . . . .	131
13.11 Example: Reversing . . . . .	134
13.12 Example: Test application . . . . .	137
<b>14 Exceptions</b>	<b>141</b>
14.1 Exception declaration . . . . .	141
14.2 Raising an exception . . . . .	141
14.3 Handling an exception . . . . .	142
14.4 Predefined exceptions . . . . .	144
<b>15 Tasking</b>	<b>145</b>
15.1 Tasks . . . . .	145
15.1.1 Simple task . . . . .	145
15.1.2 Simple synchronization . . . . .	146
15.1.3 Delay . . . . .	148
15.1.4 Synchronization: rendezvous . . . . .	149
15.1.5 Select loop . . . . .	150
15.1.6 Cycling tasks . . . . .	152
15.2 Protected objects . . . . .	155
15.2.1 Simple object . . . . .	156
15.2.2 Entries . . . . .	157
15.3 Task and protected types . . . . .	158
15.3.1 Task types . . . . .	158
15.3.2 Protected types . . . . .	160
<b>16 Design by contracts</b>	<b>163</b>
16.1 Pre- and postconditions . . . . .	163
16.2 Predicates . . . . .	166
16.3 Type invariants . . . . .	170
<b>17 Interfacing with C</b>	<b>173</b>
17.1 Multi-language project . . . . .	173
17.2 Type convention . . . . .	173
17.3 Foreign subprograms . . . . .	174
17.3.1 Calling C subprograms in Ada . . . . .	174
17.3.2 Calling Ada subprograms in C . . . . .	176
17.4 Foreign variables . . . . .	177
17.4.1 Using C global variables in Ada . . . . .	177
17.4.2 Using Ada variables in C . . . . .	178
17.5 Generating bindings . . . . .	180
17.5.1 Adapting bindings . . . . .	181
<b>18 Object-oriented programming</b>	<b>187</b>
18.1 Derived types . . . . .	188
18.2 Tagged types . . . . .	189

18.3 Classwide types . . . . .	191
18.4 Dispatching operations . . . . .	192
18.5 Dot notation . . . . .	194
18.6 Private & Limited . . . . .	195
18.7 Classwide access types . . . . .	197
<b>19 Standard library: Containers</b>	<b>201</b>
19.1 Vectors . . . . .	201
19.1.1 Instantiation . . . . .	201
19.1.2 Initialization . . . . .	202
19.1.3 Appending and prepending elements . . . . .	202
19.1.4 Accessing first and last elements . . . . .	204
19.1.5 Iterating . . . . .	205
19.1.6 Finding and changing elements . . . . .	209
19.1.7 Inserting elements . . . . .	211
19.1.8 Removing elements . . . . .	212
19.1.9 Other Operations . . . . .	215
19.2 Sets . . . . .	217
19.2.1 Initialization and iteration . . . . .	218
19.2.2 Operations on elements . . . . .	219
19.2.3 Other Operations . . . . .	221
19.3 Indefinite maps . . . . .	223
19.3.1 Hashed maps . . . . .	224
19.3.2 Ordered maps . . . . .	225
19.3.3 Complexity . . . . .	226
<b>20 Standard library: Dates &amp; Times</b>	<b>229</b>
20.1 Date and time handling . . . . .	229
20.1.1 Delaying using date . . . . .	230
20.2 Real-time . . . . .	233
20.2.1 Benchmarking . . . . .	233
<b>21 Standard library: Strings</b>	<b>237</b>
21.1 String operations . . . . .	237
21.2 Limitation of fixed-length strings . . . . .	242
21.3 Bounded strings . . . . .	242
21.4 Unbounded strings . . . . .	245
<b>22 Standard library: Files and streams</b>	<b>247</b>
22.1 Text I/O . . . . .	247
22.2 Sequential I/O . . . . .	250
22.3 Direct I/O . . . . .	252
22.4 Stream I/O . . . . .	254
<b>23 Standard library: Numerics</b>	<b>257</b>
23.1 Elementary Functions . . . . .	257
23.2 Random Number Generation . . . . .	258
23.3 Complex Types . . . . .	260
23.4 Vector and Matrix Manipulation . . . . .	262
<b>24 Appendices</b>	<b>267</b>
24.1 Appendix A: Generic Formal Types . . . . .	267
24.1.1 Indefinite version . . . . .	268
24.2 Appendix B: Containers . . . . .	269

Copyright © 2018 – 2024, AdaCore

This book is published under a CC BY-SA license, which means that you can copy, redistribute, remix, transform, and build upon the content for any purpose, even commercially, as long as you give appropriate credit, provide a link to the license, and indicate if changes were made. If you remix, transform, or build upon the material, you must distribute your contributions under the same license as the original. You can find license details [on this page<sup>1</sup>](http://creativecommons.org/licenses/by-sa/4.0)



This course will teach you the basics of the Ada programming language and is intended for those who already have a basic understanding of programming techniques. You will learn how to apply those techniques to programming in Ada.

This document was written by Raphaël Amiard and Gustavo A. Hoffmann, with review from Richard Kenner.

### Note

The code examples in this course use a 50-column limit, which greatly improves the readability of the code on devices with a small screen size. This constraint, however, leads to an unusual coding style. For instance, instead of calling `Put_Line` in a single line, we have this:

```
Put_Line
  (" is in the northeast quadrant");
```

or this:

```
Put_Line (" (X => "
  & Integer'Image (P.X)
  & ")");
```

Note that typical Ada code uses a limit of at least 79 columns. Therefore, please don't take the coding style from this course as a reference!

### Note

Each code example from this book has an associated "code block metadata", which contains the name of the "project" and an MD5 hash value. This information is used to identify a single code example.

You can find all code examples in a zip file, which you can [download from the learn website<sup>2</sup>](#). The directory structure in the zip file is based on the code block metadata. For example, if you're searching for a code example with this metadata:

- Project: Courses.Intro\_To\_Ada.Imperative\_Language.Greet
- MD5: cba89a34b87c9dfa71533d982d05e6ab

you will find it in this directory:

<sup>1</sup> <http://creativecommons.org/licenses/by-sa/4.0>

[projects/Courses/Intro\\_To\\_Ada/Imperative\\_Language/Greet/cba89a34b87c9dfa71533d982d05e6ab/](https://learn.adacore.com/zip/learning-ada_code.zip)

In order to use this code example, just follow these steps:

1. Unpack the zip file;
2. Go to target directory;
3. Start GNAT Studio on this directory;
4. Build (or compile) the project;
5. Run the application (if a main procedure is available in the project).

---

<sup>2</sup> [https://learn.adacore.com/zip/learning-ada\\_code.zip](https://learn.adacore.com/zip/learning-ada_code.zip)

## INTRODUCTION

### 1.1 History

In the 1970s the United States Department of Defense (DOD) suffered from an explosion of the number of programming languages, with different projects using different and non-standard dialects or language subsets / supersets. The DOD decided to solve this problem by issuing a request for proposals for a common, modern programming language. The winning proposal was one submitted by Jean Ichbiah from CII Honeywell-Bull.

The first Ada standard was issued in 1983; it was subsequently revised and enhanced in 1995, 2005 and 2012, with each revision bringing useful new features.

This tutorial will focus on Ada 2012 as a whole, rather than teaching different versions of the language.

### 1.2 Ada today

Today, Ada is heavily used in embedded real-time systems, many of which are safety critical. While Ada is and can be used as a general-purpose language, it will really shine in low-level applications:

- Embedded systems with low memory requirements (no garbage collector allowed).
- Direct interfacing with hardware.
- Soft or hard real-time systems.
- Low-level systems programming.

Specific domains seeing Ada usage include Aerospace & Defense, civil aviation, rail, and many others. These applications require a high degree of safety: a software defect is not just an annoyance, but may have severe consequences. Ada provides safety features that detect defects at an early stage — usually at compilation time or using static analysis tools. Ada can also be used to create applications in a variety of other areas, such as:

- Video game programming<sup>3</sup>
- Real-time audio<sup>4</sup>
- Kernel modules<sup>5</sup>

This is a non-comprehensive list that hopefully sheds light on which kind of programming Ada is good at.

---

<sup>3</sup> <https://github.com/AdaDoom3/AdaDoom3>

<sup>4</sup> <http://www.electronicdesign.com/embedded-revolution/assessing-ada-language-audio-applications>

<sup>5</sup> <http://www.nihamkin.com/tag/kernel.html>

In terms of modern languages, the closest in terms of targets and level of abstraction are probably C++<sup>6</sup> and Rust<sup>7</sup>.

### 1.3 Philosophy

Ada's philosophy is different from most other languages. Underlying Ada's design are principles that include the following:

- Readability is more important than conciseness. Syntactically this shows through the fact that keywords are preferred to symbols, that no keyword is an abbreviation, etc.
- Very strong typing. It is very easy to introduce new types in Ada, with the benefit of preventing data usage errors.
  - It is similar to many functional languages in that regard, except that the programmer has to be much more explicit about typing in Ada, because there is almost no type inference.
- Explicit is better than implicit. Although this is a Python<sup>8</sup> commandment, Ada takes it way further than any language we know of:
  - There is mostly no structural typing, and most types need to be explicitly named by the programmer.
  - As previously said, there is mostly no type inference.
  - Semantics are very well defined, and undefined behavior is limited to an absolute minimum.
  - The programmer can generally give a *lot* of information about what their program means to the compiler (and other programmers). This allows the compiler to be extremely helpful (read: strict) with the programmer.

During this course, we will explain the individual language features that are building blocks for that philosophy.

### 1.4 SPARK

While this class is solely about the Ada language, it is worth mentioning that another language, extremely close to and interoperable with Ada, exists: the SPARK language.

SPARK is a subset of Ada, designed so that the code written in SPARK is amenable to automatic proof. This provides a level of assurance with regard to the correctness of your code that is much higher than with a regular programming language.

There is a dedicated course for the SPARK language but keep in mind that every time we speak about the specification power of Ada during this course, it is power that you can leverage in SPARK to help proving the correctness of program properties ranging from absence of run-time errors to compliance with formally specified functional requirements.

---

<sup>6</sup> <https://en.wikipedia.org/wiki/C%2B%2B>

<sup>7</sup> <https://www.rust-lang.org/en-US/>

<sup>8</sup> <https://www.python.org>

## IMPERATIVE LANGUAGE

Ada is a multi-paradigm language with support for object orientation and some elements of functional programming, but its core is a simple, coherent procedural/imperative language akin to C or Pascal.

### In other languages

One important distinction between Ada and a language like C is that statements and expressions are very clearly distinguished. In Ada, if you try to use an expression where a statement is required then your program will fail to compile. This rule supports a useful stylistic principle: expressions are intended to deliver values, not to have side effects. It can also prevent some programming errors, such as mistakenly using the equality operator = instead of the assignment operation := in an assignment statement.

## 2.1 Hello world

Here's a very simple imperative Ada program:

Listing 1: greet.adb

```
1 with Ada.Text_IO;
2
3 procedure Greet is
4 begin
5     -- Print "Hello, World!" to the screen
6     Ada.Text_IO.Put_Line ("Hello, World!");
7 end Greet;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Greet  
MD5: cba89a34b87c9dfa71533d982d05e6ab

### Runtime output

Hello, World!

which we'll assume is in the source file greet.adb.

There are several noteworthy things in the above program:

- A subprogram in Ada can be either a procedure or a function. A procedure, as illustrated above, does not return a value when called.
- **with** is used to reference external modules that are needed in the procedure. This is similar to **import** in various languages or roughly similar to **#include** in C and C++. We'll see later how they work in detail. Here, we are requesting a standard library

module, the `Ada.Text_Io` package, which contains a procedure to print text on the screen: `Put_Line`.

- `Greet` is a procedure, and the main entry point for our first program. Unlike in C or C++, it can be named anything you prefer. The builder will determine the entry point. In our simple example, `gprbuild`, GNAT's builder, will use the file you passed as parameter.
- `Put_Line` is a procedure, just like `Greet`, except it is declared in the `Ada.Text_Io` module. It is the Ada equivalent of C's `printf`.
- Comments start with `--` and go to the end of the line. There is no multi-line comment syntax, that is, it is not possible to start a comment in one line and continue it in the next line. The only way to create multiple lines of comments in Ada is by using `--` on each line. For example:

```
-- We start a comment in this line...
-- and we continue on the second line...
```

### In other languages

Procedures are similar to functions in C or C++ that return `void`. We'll see later how to declare functions in Ada.

Here is a minor variant of the "Hello, World" example:

Listing 2: `greet.adb`

```
1 with Ada.Text_Io; use Ada.Text_Io;
2
3 procedure Greet is
4 begin
5     -- Print "Hello, World!" to the screen
6     Put_Line ("Hello, World!");
7 end Greet;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Greet\_2  
MD5: a58a1193207df44aa6edaa4fe1c14280

### Runtime output

```
Hello, World!
```

This version utilizes an Ada feature known as a `use` clause, which has the form `use package-name`. As illustrated by the call on `Put_Line`, the effect is that entities from the named package can be referenced directly, without the `package-name`. prefix.

## 2.2 Imperative language - If/Then/Else

This section describes Ada's `if` statement and introduces several other fundamental language facilities including integer I/O, data declarations, and subprogram parameter modes.

Ada's `if` statement is pretty unsurprising in form and function:

Listing 3: check\_positive.adb

```

1  with Ada.Text_Io;      use Ada.Text_Io;
2  with Ada.Integer_Text_Io; use Ada.Integer_Text_Io;
3
4  procedure Check_Positive is
5    N : Integer;
6  begin
7    -- Put a String
8    Put ("Enter an integer value: ");
9
10   -- Read in an integer value
11   Get (N);
12
13   if N > 0 then
14     -- Put an Integer
15     Put (N);
16     Put_Line (" is a positive number");
17   end if;
18 end Check_Positive;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Check\_Positive  
MD5: 2e8b4b2f3f258fd9e02c2d65846af101

The **if** statement minimally consists of the reserved word **if**, a condition (which must be a Boolean value), the reserved word **then** and a non-empty sequence of statements (the **then** part) which is executed if the condition evaluates to True, and a terminating **end if**.

This example declares an integer variable N, prompts the user for an integer, checks if the value is positive and, if so, displays the integer's value followed by the string " is a positive number". If the value is not positive, the procedure does not display any output.

The type Integer is a predefined signed type, and its range depends on the computer architecture. On typical current processors Integer is 32-bit signed.

The example illustrates some of the basic functionality for integer input-output. The relevant subprograms are in the predefined package Ada.Integer\_Text\_Io and include the Get procedure (which reads in a number from the keyboard) and the Put procedure (which displays an integer value).

Here's a slight variation on the example, which illustrates an **if** statement with an **else** part:

Listing 4: check\_positive.adb

```

1  with Ada.Text_Io;      use Ada.Text_Io;
2  with Ada.Integer_Text_Io; use Ada.Integer_Text_Io;
3
4  procedure Check_Positive is
5    N : Integer;
6  begin
7    -- Put a String
8    Put ("Enter an integer value: ");
9
10   -- Reads in an integer value
11   Get (N);
12
13   -- Put an Integer
14   Put (N);
15

```

(continues on next page)

(continued from previous page)

```
16  if N > 0 then
17      Put_Line (" is a positive number");
18  else
19      Put_Line (" is not a positive number");
20  end if;
21 end Check_Positive;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Check\_Positive\_2  
MD5: 28fca0d7840d06d478e5933e8182d1db

In this example, if the input value is not positive then the program displays the value followed by the String " is not a positive number".

Our final variation illustrates an **if** statement with **elsif** sections:

Listing 5: check\_direction.adb

```
1  with Ada.Text_Io;          use Ada.Text_Io;
2  with Ada.Integer_Text_Io;  use Ada.Integer_Text_Io;
3
4  procedure Check_Direction is
5      N : Integer;
6  begin
7      Put ("Enter an integer value: ");
8      Get (N);
9      Put (N);
10
11     if N = 0 or N = 360 then
12         Put_Line (" is due north");
13     elsif N in 1 .. 89 then
14         Put_Line (" is in the northeast quadrant");
15     elsif N = 90 then
16         Put_Line (" is due east");
17     elsif N in 91 .. 179 then
18         Put_Line (" is in the southeast quadrant");
19     elsif N = 180 then
20         Put_Line (" is due south");
21     elsif N in 181 .. 269 then
22         Put_Line (" is in the southwest quadrant");
23     elsif N = 270 then
24         Put_Line (" is due west");
25     elsif N in 271 .. 359 then
26         Put_Line (" is in the northwest quadrant");
27     else
28         Put_Line (" is not in the range 0..360");
29     end if;
30 end Check_Direction;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Check\_Direction  
MD5: 7759d30c9bb0bfb88efdf12128f9c382

This example expects the user to input an integer between 0 and 360 inclusive, and displays which quadrant or axis the value corresponds to. The **in** operator in Ada tests whether a scalar value is within a specified range and returns a Boolean result. The effect of the program should be self-explanatory; later we'll see an alternative and more efficient style to accomplish the same effect, through a **case** statement.

Ada's **elsif** keyword differs from C or C++, where nested **else .. if** blocks would be

used instead. And another difference is the presence of the `end if` in Ada, which avoids the problem known as the "dangling else".

## 2.3 Imperative language - Loops

Ada has three ways of specifying loops. They differ from the C / Java / Javascript for-loop, however, with simpler syntax and semantics in line with Ada's philosophy.

### 2.3.1 For loops

The first kind of loop is the `for` loop, which allows iteration through a discrete range.

Listing 6: greet\_5a.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Greet_5a is
4  begin
5      for I in 1 .. 5 loop
6          -- Put_Line is a procedure call
7          Put_Line ("Hello, World!"
8                  & Integer'Image (I));
9          -- ^ Procedure parameter
10     end loop;
11 end Greet_5a;

```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Greet\_5a  
MD5: 7f588b67947126f789333adfaaf1b638

#### Runtime output

```

Hello, World! 1
Hello, World! 2
Hello, World! 3
Hello, World! 4
Hello, World! 5

```

A few things to note:

- `1 .. 5` is a discrete range, from `1` to `5` inclusive.
- The loop parameter `I` (the name is arbitrary) in the body of the loop has a value within this range.
- `I` is local to the loop, so you cannot refer to `I` outside the loop.
- Although the value of `I` is incremented at each iteration, from the program's perspective it is constant. An attempt to modify its value is illegal; the compiler would reject the program.
- `Integer'Image` is a function that takes an Integer and converts it to a `String`. It is an example of a language construct known as an *attribute*, indicated by the `'` syntax, which will be covered in more detail later.
- The `&` symbol is the concatenation operator for String values
- The `end loop` marks the end of the loop

The "step" of the loop is limited to 1 (forward direction) and -1 (backward). To iterate backwards over a range, use the `reverse` keyword:

Listing 7: greet\_5a\_reverse.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Greet_5a_Reverse is
4 begin
5   for I in reverse 1 .. 5 loop
6     Put_Line ("Hello, World!" & Integer'Image (I));
7   end loop;
8 end Greet_5a_Reverse;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Greet\_5a\_Reverse  
MD5: a0d5dcfc471fb1a107477c934fa527c2

### Runtime output

```
Hello, World! 5
Hello, World! 4
Hello, World! 3
Hello, World! 2
Hello, World! 1
```

The bounds of a **for** loop may be computed at run-time; they are evaluated once, before the loop body is executed. If the value of the upper bound is less than the value of the lower bound, then the loop is not executed at all. This is the case also for **reverse** loops. Thus no output is produced in the following example:

Listing 8: greet\_no\_op.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Greet_No_Op is
4 begin
5   for I in reverse 5 .. 1 loop
6     Put_Line ("Hello, World!" & Integer'Image (I));
7   end loop;
8 end Greet_No_Op;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Greet\_No\_Op  
MD5: 5070693fb0324d3e4e43a8c8c4f046e1

### Build output

greet\_no\_op.adb:5:23: warning: loop range is null, loop will not execute [enabled by default]

The **for** loop is more general than what we illustrated here; more on that later.

## 2.3.2 Bare loops

The simplest loop in Ada is the bare loop, which forms the foundation of the other kinds of Ada loops.

Listing 9: greet\_5b.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Greet_5b is
4      -- Variable declaration:
5      I : Integer := 1;
6      -- ^ Type
7      --          ^ Initial value
8  begin
9      loop
10         Put_Line ("Hello, World!"
11             & Integer'Image (I));
12
13         -- Exit statement:
14         exit when I = 5;
15         --          ^ Boolean condition
16
17         -- Assignment:
18         I := I + 1;
19         -- There is no I++ short form to
20         -- increment a variable
21     end loop;
22 end Greet_5b;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Greet\_5b  
MD5: 5b218a64a07f64bd97774b574883c44a

### Runtime output

```

Hello, World! 1
Hello, World! 2
Hello, World! 3
Hello, World! 4
Hello, World! 5

```

This example has the same effect as Greet\_5a shown earlier.

It illustrates several concepts:

- We have declared a variable named I between the **is** and the **begin**. This constitutes a *declarative region*. Ada clearly separates the declarative region from the statement part of a subprogram. A declaration can appear in a declarative region but is not allowed as a statement.
- The bare loop statement is introduced by the keyword **loop** on its own and, like every kind of loop statement, is terminated by the combination of keywords **end loop**. On its own, it is an infinite loop. You can break out of it with an **exit** statement.
- The syntax for assignment is **:=**, and the one for equality is **=**. There is no way to confuse them, because as previously noted, in Ada, statements and expressions are distinct, and expressions are not valid statements.

### 2.3.3 While loops

The last kind of loop in Ada is the **while** loop.

Listing 10: greet\_5c.adb

```
1 with Ada.Text_Io; use Ada.Text_Io;
2
3 procedure Greet_5c is
4     I : Integer := 1;
5 begin
6     -- Condition must be a Boolean value
7     -- (no Integers).
8     -- Operator "<=" returns a Boolean
9     while I <= 5 loop
10         Put_Line ("Hello, World!" & Integer'Image (I));
11
12         I := I + 1;
13     end loop;
14 end Greet_5c;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Greet\_5c  
MD5: 5d1d099477795b226db43736c2810274

### Runtime output

```
Hello, World! 1
Hello, World! 2
Hello, World! 3
Hello, World! 4
Hello, World! 5
```

The condition is evaluated before each iteration. If the result is false, then the loop is terminated.

This program has the same effect as the previous examples.

#### In other languages

Note that Ada has different semantics than C-based languages with respect to the condition in a while loop. In Ada the condition has to be a Boolean value or the compiler will reject the program; the condition is not an integer that is treated as either **True** or **False** depending on whether it is non-zero or zero.

## 2.4 Imperative language - Case statement

Ada's **case** statement is similar to the C and C++ **switch** statement, but with some important differences.

Here's an example, a variation of a program that was shown earlier with an **if** statement:

Listing 11: check\_direction.adb

```
1 with Ada.Text_Io; use Ada.Text_Io;
2 with Ada.Integer_Text_Io; use Ada.Integer_Text_Io;
3
4 procedure Check_Direction is
5     N : Integer;
6 begin
```

(continues on next page)

(continued from previous page)

```

7  loop
8      Put ("Enter an integer value: ");
9      Get (N);
10     Put (N);
11
12     case N is
13         when 0 | 360 =>
14             Put_Line
15             (" is due north");
16         when 1 .. 89 =>
17             Put_Line
18             (" is in the northeast quadrant");
19         when 90 =>
20             Put_Line
21             (" is due east");
22         when 91 .. 179 =>
23             Put_Line
24             (" is in the southeast quadrant");
25         when 180 =>
26             Put_Line
27             (" is due south");
28         when 181 .. 269 =>
29             Put_Line
30             (" is in the southwest quadrant");
31         when 270 =>
32             Put_Line
33             (" is due west");
34         when 271 .. 359 =>
35             Put_Line
36             (" is in the northwest quadrant");
37         when others =>
38             Put_Line
39             (" Au revoir");
40             exit;
41     end case;
42 end loop;
43 end Check_Direction;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Check\_Direction\_2  
MD5: 1c758b76a2c3991cb4e2a0cf5e172ac3

This program repeatedly prompts for an integer value and then, if the value is in the range 0 .. 360, displays the associated quadrant or axis. If the value is an Integer outside this range, the loop (and the program) terminate after outputting a farewell message.

The effect of the case statement is similar to the if statement in an earlier example, but the case statement can be more efficient because it does not involve multiple range tests.

Notable points about Ada's case statement:

- The case expression (here the variable N) must be of a discrete type, i.e. either an integer type or an enumeration type. Discrete types will be covered in more detail later *discrete types* (page 43).
- Every possible value for the case expression needs to be covered by a unique branch of the case statement. This will be checked at compile time.
- A branch can specify a single value, such as 0; a range of values, such as 1 .. 89; or any combination of the two (separated by a |).
- As a special case, an optional final branch can specify **others**, which covers all values

not included in the earlier branches.

- Execution consists of the evaluation of the case expression and then a transfer of control to the statement sequence in the unique branch that covers that value.
- When execution of the statements in the selected branch has completed, control resumes after the **end case**. Unlike C, execution does not fall through to the next branch. So Ada doesn't need (and doesn't have) a **break** statement.

## 2.5 Imperative language - Declarative regions

As mentioned earlier, Ada draws a clear syntactic separation between declarations, which introduce names for entities that will be used in the program, and statements, which perform the processing. The areas in the program where declarations may appear are known as declarative regions.

In any subprogram, the section between the **is** and the **begin** is a declarative region. You can have variables, constants, types, inner subprograms, and other entities there.

We've briefly mentioned variable declarations in previous subsection. Let's look at a simple example, where we declare an integer variable X in the declarative region and perform an initialization and an addition on it:

Listing 12: main.adb

```
1 with Ada.Text_Io; use Ada.Text_Io;
2
3 procedure Main is
4     X : Integer;
5 begin
6     X := 0;
7     Put_Line ("The initial value of X is "
8             & Integer'Image (X));
9
10    Put_Line ("Performing operation on X... ");
11    X := X + 1;
12
13    Put_Line ("The value of X now is "
14             & Integer'Image (X));
15 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Variable\_Declaration  
MD5: cbb08d5e382fbfcc28e986bea80cd253

### Runtime output

```
The initial value of X is 0
Performing operation on X...
The value of X now is 1
```

Let's look at an example of a nested procedure:

Listing 13: main.adb

```
1 with Ada.Text_Io; use Ada.Text_Io;
2
3 procedure Main is
4     procedure Nested is
5 begin
```

(continues on next page)

(continued from previous page)

```

6      Put_Line ("Hello World");
7  end Nested;
8 begin
9  Nested;
10 --  Call to Nested
11 end Main;

```

## Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Nested\_Procedure  
MD5: 2e7fb267e31232196065feb5e35e6ef

## Runtime output

Hello World

A declaration cannot appear as a statement. If you need to declare a local variable amidst the statements, you can introduce a new declarative region with a block statement:

Listing 14: greet.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Greet is
4  begin
5    loop
6      Put_Line ("Please enter your name: ");
7
8      declare
9        Name : String := Get_Line;
10       --          ^ Call to the
11       --          Get_Line function
12      begin
13        exit when Name = "";
14        Put_Line ("Hi " & Name & "!");
15      end;
16
17      -- Name is undefined here
18    end loop;
19
20    Put_Line ("Bye!");
21  end Greet;

```

## Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Greet\_6  
MD5: a9c0c14a1b3e2ebe07cd88f442787e3a

### ⚠ Attention

The Get\_Line function allows you to receive input from the user, and get the result as a string. It is more or less equivalent to the scanf C function.

It returns a **String**, which, as we will see later, is an *Unconstrained array type* (page 74). For now we simply note that, if you wish to declare a **String** variable and do not know its size in advance, then you need to initialize the variable during its declaration.

## 2.6 Imperative language - conditional expressions

Ada 2012 introduced an expression analog for conditional statements (**if** and **case**).

### 2.6.1 If expressions

Here's an alternative version of an example we saw earlier; the **if** statement has been replaced by an **if** expression:

Listing 15: check\_positive.adb

```

1  with Ada.Text_Io;      use Ada.Text_Io;
2  with Ada.Integer_Text_Io; use Ada.Integer_Text_Io;
3
4  procedure Check_Positive is
5    N : Integer;
6  begin
7    Put ("Enter an integer value: ");
8    Get (N);
9    Put (N);
10
11  declare
12    S : constant String :=
13      (if N > 0
14       then " is a positive number"
15       else " is not a positive number");
16  begin
17    Put_Line (S);
18  end;
19 end Check_Positive;

```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Check\_Positive  
MD5: 01f23463b14774f750dbb21f6c65ea09

The **if** expression evaluates to one of the two Strings depending on N, and assigns that value to the local variable S.

Ada's **if** expressions are similar to **if** statements. However, there are a few differences that stem from the fact that it is an expression:

- All branches' expressions must be of the same type
- It *must* be surrounded by parentheses if the surrounding expression does not already contain them
- An **else** branch is mandatory unless the expression following **then** has a Boolean value. In that case an **else** branch is optional and, if not present, defaults to **else True**.

Here's another example:

Listing 16: main.adb

```

1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Main is
4  begin
5    for I in 1 .. 10 loop
6      Put_Line (if I mod 2 = 0
7                  then "Even"

```

(continues on next page)

(continued from previous page)

```

8     else "Odd");
9 end loop;
10 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Even\_Odd  
MD5: c89c3233ab8822c828f7a7bba8fd3f1c

### Runtime output

```

Odd
Even
Odd
Even
Odd
Even
Odd
Even
Odd
Even

```

This program produces 10 lines of output, alternating between "Odd" and "Even".

## 2.6.2 Case expressions

Analogous to **if** expressions, Ada also has **case** expressions. They work just as you would expect.

Listing 17: main.adb

```

1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Main is
4 begin
5   for I in 1 .. 10 loop
6     Put_Line
7     (case I is
8       when 1 | 3 | 5 | 7 | 9 => "Odd",
9       when 2 | 4 | 6 | 8 | 10 => "Even");
10  end loop;
11 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Imperative\_Language.Case\_Expression  
MD5: 6ce40efc987c2665960b1f08d30d780d

### Runtime output

```

Odd
Even
Odd
Even
Odd
Even
Odd
Even
Odd
Even

```

This program has the same effect as the preceding example.

The syntax differs from **case** statements, with branches separated by commas.

## SUBPROGRAMS

### 3.1 Subprograms

So far, we have used procedures, mostly to have a main body of code to execute. Procedures are one kind of *subprogram*.

There are two kinds of subprograms in Ada, *functions* and *procedures*. The distinction between the two is that a function returns a value, and a procedure does not.

This example shows the declaration and definition of a function:

Listing 18: increment.ads

```
1  function Increment (I : Integer) return Integer;
```

Listing 19: increment.adb

```
1  -- We declare (but don't define) a function with
2  -- one parameter, returning an integer value
3
4  function Increment (I : Integer) return Integer is
5  -- We define the Increment function
6  begin
7      return I + 1;
8  end Increment;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Increment  
MD5: 582fe283730a130cec071c455a0ce3d4

Subprograms in Ada can, of course, have parameters. One syntactically important note is that a subprogram which has no parameters does not have a parameter section at all, for example:

```
procedure Proc;
function Func return Integer;
```

Here's another variation on the previous example:

Listing 20: increment\_by.ads

```
1  function Increment_By
2  (I      : Integer := 0;
3  Incr   : Integer := 1) return Integer;
4  --          ^ Default value for parameters
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Increment\_By  
MD5: 5728b915789beee0b5546ea7b36a1cc2

In this example, we see that parameters can have default values. When calling the subprogram, you can then omit parameters if they have a default value. Unlike C/C++, a call to a subprogram without parameters does not include parentheses.

This is the implementation of the function above:

Listing 21: increment\_by.adb

```
1  function Increment_By
2    (I : Integer := 0;
3     Incr : Integer := 1) return Integer is
4 begin
5   return I + Incr;
6 end Increment_By;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Increment\_By  
MD5: 07c85e5c1272ea396bf4dbc0cefcdce7

#### In the GNAT toolchain

The Ada standard doesn't mandate in which file the specification or the implementation of a subprogram must be stored. In other words, the standard doesn't require a specific file structure or specific file name extensions. For example, we could save both the specification and the implementation of the Increment function above in a file called `increment.txt`. (We could even store the entire source code of a system in a single file.) From the standard's perspective, this would be completely acceptable.

The GNAT toolchain, however, requires the following file naming scheme:

- files with the `.ads` extension contain the specification, while
- files with the `.adb` extension contain the implementation.

Therefore, in the GNAT toolchain, the specification of the Increment function must be stored in the `increment.ads` file, while its implementation must be stored in the `increment.adb` file. This rule always applies to packages, which we discuss [later](#) (page 31). (Note, however, that it's possible to circumvent this rule.) For more details, you may refer to the Introduction to GNAT Toolchain course or the [GPRbuild User's Guide](#)<sup>9</sup>.

### 3.1.1 Subprogram calls

We can then call our subprogram this way:

Listing 22: show\_increment.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2  with Increment_By;
3
4  procedure Show_Increment is
5    A, B, C : Integer;
6  begin
7    C := Increment_By;
```

(continues on next page)

<sup>9</sup> [https://docs.adacore.com/gprbuild-docs/html/gprbuild\\_ug.html](https://docs.adacore.com/gprbuild-docs/html/gprbuild_ug.html)

(continued from previous page)

```

8      --          ^ Parameterless call,
9      --          value of I is 0
10     --          and Incr is 1
11
12 Put_Line ("Using defaults for Increment_By is "
13   & Integer'Image (C));
14
15 A := 10;
16 B := 3;
17 C := Increment_By (A, B);
18 --          ^ Regular parameter passing
19
20 Put_Line ("Increment of "
21   & Integer'Image (A)
22   & " with "
23   & Integer'Image (B)
24   & " is "
25   & Integer'Image (C));
26
27 A := 20;
28 B := 5;
29 C := Increment_By (I    => A,
30                     Incr => B);
31 --          ^ Named parameter passing
32
33 Put_Line ("Increment of "
34   & Integer'Image (A)
35   & " with "
36   & Integer'Image (B)
37   & " is "
38   & Integer'Image (C));
39 end Show_Increment;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Increment\_By  
MD5: dcb501c8c6815b03c6841fc8b80d6911

### Runtime output

```

Using defaults for Increment_By is 1
Increment of 10 with 3 is 13
Increment of 20 with 5 is 25

```

Ada allows you to name the parameters when you pass them, whether they have a default or not. There are some rules:

- Positional parameters come first.
- A positional parameter cannot follow a named parameter.

As a convention, people usually name parameters at the call site if the function's corresponding parameters has a default value. However, it is also perfectly acceptable to name every parameter if it makes the code clearer.

## 3.1.2 Nested subprograms

As briefly mentioned earlier, Ada allows you to declare one subprogram inside another.

This is useful for two reasons:

- It lets you organize your programs in a cleaner fashion. If you need a subprogram only as a "helper" for another subprogram, then the principle of localization indicates that the helper subprogram should be declared nested.
- It allows you to share state easily in a controlled fashion, because the nested subprograms have access to the parameters, as well as any local variables, declared in the outer scope.

For the previous example, we can move the duplicated code (call to Put\_Line) to a separate procedure. This is a shortened version with the nested `Display_Result` procedure.

Listing 23: `show_increment.adb`

```
1  with Ada.Text_Io; use Ada.Text_Io;
2  with Increment_By;
3
4  procedure Show_Increment is
5    A, B, C : Integer;
6
7    procedure Display_Result is
8    begin
9      Put_Line ("Increment of "
10        & Integer'Image (A)
11        & " with "
12        & Integer'Image (B)
13        & " is "
14        & Integer'Image (C));
15    end Display_Result;
16
17  begin
18    A := 10;
19    B := 3;
20    C := Increment_By (A, B);
21    Display_Result;
22    A := 20;
23    B := 5;
24    C := Increment_By (A, B);
25    Display_Result;
26  end Show_Increment;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Increment\_By  
MD5: 23ec8ae3080c042123a9e82ee6b3d9e3

### Runtime output

Increment of 10 with 3 is 13  
Increment of 20 with 5 is 25

## 3.1.3 Function calls

An important feature of function calls in Ada is that the return value at a call cannot be ignored; that is, a function call cannot be used as a statement.

If you want to call a function and do not need its result, you will still need to explicitly store it in a local variable.

Listing 24: `quadruple.ads`

```
1  function Quadruple (I : Integer)
2    return Integer;
```

Listing 25: quadruple.adb

```

1  function Quadruple (I : Integer)
2      return Integer is
3
4      function Double (I : Integer)
5          return Integer is
6      begin
7          return I * 2;
8      end Double;
9
10     Res : Integer;
11 begin
12     Double (Double (I));
13     -- ERROR: cannot use call to function
14     -- "Double" as a statement
15
16     Res := Double (Double (I));
17     -- OK: return value of Double is
18     -- assigned to Res
19
20     return Res;
21 end Quadruple;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Quadruple  
MD5: fd525bde1a883d4a3794730695af6469

### Build output

```

quadruple.adb:12:04: error: cannot use call to function "Double" as a statement
quadruple.adb:12:04: error: return value of a function call cannot be ignored
gprbuild: *** compilation phase failed

```

A statement such as `Double (Double (I));` is wrong because we're not assigning the return value to a variable — we can correct this statement by writing `Res := Double (Double (I));`.

### In the GNAT toolchain

In GNAT, with all warnings activated, it becomes even harder to ignore the result of a function, because unused variables will be flagged. For example, this code would not be valid:

```

function Read_Int
  (Stream : Network_Stream;
   Result : out Integer) return Boolean;

procedure Main is
  Stream : Network_Stream := Get_Stream;
  My_Int : Integer;

  -- Warning: in the line below, B is
  -- never read.
  B : Boolean := Read_Int (Stream, My_Int);
begin
  null;
end Main;

```

You then have two solutions to silence this warning:

- Either annotate the variable with `pragma Unreferenced`, e.g.:

```
B : Boolean := Read_Int (Stream, My_Int);  
pragma Unreferenced (B);
```

- Or give the variable a name that contains any of the strings `discard` `dummy` `ignore` `junk` `unused` (case insensitive)

## 3.2 Parameter modes

So far we have seen that Ada is a safety-focused language. There are many ways this is realized, but two important points are:

- Ada makes the user specify as much as possible about the behavior expected for the program, so that the compiler can warn or reject if there is an inconsistency.
- Ada provides a variety of techniques for achieving the generality and flexibility of pointers and dynamic memory management, but without the latter's drawbacks (such as memory leakage and dangling references).

Parameter modes are a feature that helps achieve the two design goals above. A subprogram parameter can be specified with a mode, which is one of the following:

<code>in</code>	Parameter can only be read, not written
<code>out</code>	Parameter can be written to, then read
<code>in out</code>	Parameter can be both read and written

The default mode for parameters is `in`; so far, most of the examples have been using `in` parameters.

### Historically

Functions and procedures were originally more different in philosophy. Before Ada 2012, functions could only take `in` parameters.

## 3.3 Subprogram calls

### 3.3.1 In parameters

The first mode for parameters is the one we have been implicitly using so far. Parameters passed using this mode cannot be modified, so that the following program will cause an error:

Listing 26: swap.adb

```
1 procedure Swap (A, B : Integer) is  
2     Tmp : Integer;  
3 begin  
4     Tmp := A;  
5  
6     -- Error: assignment to "in" mode  
7     -- parameter not allowed  
8     A := B;  
9  
10    -- Error: assignment to "in" mode
```

(continues on next page)

(continued from previous page)

```

11      -- parameter not allowed
12      B := Tmp;
13  end Swap;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Swap  
MD5: 478ac23f878934aae820e4b9c056d939

### Build output

```

swap.adb:8:04: error: assignment to "in" mode parameter not allowed
swap.adb:12:04: error: assignment to "in" mode parameter not allowed
gprbuild: *** compilation phase failed

```

The fact that **in** is the default mode is very important. It means that a parameter will not be modified unless you explicitly specify a mode in which modification is allowed.

### 3.3.2 In out parameters

To correct our code above, we can use an **in out** parameter.

Listing 27: in\_out\_params.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure In_Out_Params is
4      procedure Swap (A, B : in out Integer) is
5          Tmp : Integer;
6      begin
7          Tmp := A;
8          A := B;
9          B := Tmp;
10     end Swap;
11
12     A : Integer := 12;
13     B : Integer := 44;
14 begin
15     Swap (A, B);
16
17     -- Prints 44
18     Put_Line (Integer'Image (A));
19 end In_Out_Params;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.In\_Out\_Params  
MD5: 319358e479449c115cf2b3cbb4ff3a6b

### Runtime output

44

An **in out** parameter will allow read and write access to the object passed as parameter, so in the example above, we can see that A is modified after the call to Swap.



Attention

While `in` `out` parameters look a bit like references in C++, or regular parameters in Java that are passed by-reference, the Ada language standard does not mandate "by reference" passing for `in` `out` parameters except for certain categories of types as will be explained later.

In general, it is better to think of modes as higher level than by-value versus by-reference semantics. For the compiler, it means that an array passed as an `in` parameter might be passed by reference, because it is more efficient (which does not change anything for the user since the parameter is not assignable). However, a parameter of a discrete type will always be passed by copy, regardless of its mode (which is more efficient on most architectures).

### 3.3.3 Out parameters

The `out` mode applies when the subprogram needs to write to a parameter that might be uninitialized at the point of call. Reading the value of an `out` parameter is permitted, but it should only be done after the subprogram has assigned a value to the parameter. Out parameters behave a bit like return values for functions. When the subprogram returns, the actual parameter (a variable) will have the value of the `out` parameter at the point of return.

#### In other languages

Ada doesn't have a tuple construct and does not allow returning multiple values from a subprogram (except by declaring a full-fledged record type). Hence, a way to return multiple values from a subprogram is to use `out` parameters.

For example, a procedure reading integers from the network could have one of the following specifications:

```
procedure Read_Int
  (Stream : Network_Stream;
   Success : out Boolean;
   Result : out Integer);

function Read_Int
  (Stream : Network_Stream;
   Result : out Integer) return Boolean;
```

While reading an `out` variable before writing to it should, ideally, trigger an error, imposing that as a rule would cause either inefficient run-time checks or complex compile-time rules. So from the user's perspective an `out` parameter acts like an uninitialized variable when the subprogram is invoked.

#### In the GNAT toolchain

GNAT will detect simple cases of incorrect use of `out` parameters. For example, the compiler will emit a warning for the following program:

Listing 28: outp.adb

```
1  procedure Outp is
2    procedure Foo (A : out Integer) is
3      B : Integer := A;
4      --          ^ Warning on reference
5      --          to uninitialized A
6    begin
```

```

7      A := B;
8  end Foo;
9 begin
10  null;
11 end Outp;

```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Out\_Params  
MD5: 36bdb4e541297d7fb0b075816cb6e73a

#### Build output

```
outp.adb:3:22: warning: "A" may be referenced before it has a value [enabled by default]
```

### 3.3.4 Forward declaration of subprograms

As we saw earlier, a subprogram can be declared without being fully defined. This is possible in general, and can be useful if you need subprograms to be mutually recursive, as in the example below:

Listing 29: mutually\_recursive\_subprograms.adb

```

1  procedure Mutually_Recursive_Subprograms is
2    procedure Compute_A (V : Natural);
3      -- Forward declaration of Compute_A
4
5    procedure Compute_B (V : Natural) is
6    begin
7      if V > 5 then
8        Compute_A (V - 1);
9        -- Call to Compute_A
10     end if;
11   end Compute_B;
12
13   procedure Compute_A (V : Natural) is
14   begin
15     if V > 2 then
16       Compute_B (V - 1);
17       -- Call to Compute_B
18     end if;
19   end Compute_A;
20 begin
21   Compute_A (15);
22 end Mutually_Recursive_Subprograms;

```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Mutually\_Recursive\_Subprograms  
MD5: 5ee030cdecc6c4aea8916cbb763e8526

### 3.4 Renaming

Subprograms can be renamed by using the **renames** keyword and declaring a new name for a subprogram:

```
procedure New_Proc renames Original_Proc;
```

## Introduction to Ada

This can be useful, for example, to improve the readability of your application when you're using code from external sources that cannot be changed in your system. Let's look at an example:

Listing 30: a\_procedure\_with\_very\_long\_name\_that\_cannot\_be\_changed.ads

```
1 procedure A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed
2   (A_Message : String);
```

Listing 31: a\_procedure\_with\_very\_long\_name\_that\_cannot\_be\_changed.adb

```
1 with Ada.Text_Io; use Ada.Text_Io;
2
3 procedure A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed
4   (A_Message : String) is
5   begin
6     Put_Line (A_Message);
7   end A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Proc\_Renaming  
MD5: 6d4952e9dee8ef69a9e3c3e185c635f1

As the wording in the name of procedure above implies, we cannot change its name. We can, however, rename it to something like Show in our test application and use this shorter name. Note that we also have to declare all parameters of the original subprogram — we may rename them, too, in the declaration. For example:

Listing 32: show\_renaming.adb

```
1 with A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;
2
3 procedure Show_Renaming is
4
5   procedure Show (S : String) renames
6     A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;
7
8   begin
9     Show ("Hello World!");
10  end Show_Renaming;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Proc\_Renaming  
MD5: 5b3b550f8a1cbeb7d9cf3673f6d42b3

### Runtime output

Hello World!

Note that the original name (A\_Procedure\_With\_Very\_Long\_Name\_That\_Cannot\_Be\_Changed) is still visible after the declaration of the Show procedure.

We may also rename subprograms from the standard library. For example, we may rename **Integer'Image** to **Img**:

Listing 33: show\_image\_renaming.adb

```
1 with Ada.Text_Io; use Ada.Text_Io;
2
3 procedure Show_Image_Renaming is
```

(continues on next page)

(continued from previous page)

```

4  function Img (I : Integer) return String
5    renames Integer'Image;
6
7  begin
8    Put_Line (Img (2));
9    Put_Line (Img (3));
10   end Show_Image_Renaming;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Subprograms.Integer\_Image\_Renaming  
 MD5: 9843b9d5967679c4fe8bd83a5213829f

### Runtime output

```

2
3

```

Renaming also allows us to introduce default expressions that were not available in the original declaration. For example, we may specify "Hello World!" as the default for the **String** parameter of the Show procedure:

```

with A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;

procedure Show_Renaming_Defaults is

  procedure Show (S : String := "Hello World!")
    renames
      A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;

begin
  Show;
end Show_Renaming_Defaults;

```



## MODULAR PROGRAMMING

So far, our examples have been simple standalone subprograms. Ada is helpful in that regard, since it allows arbitrary declarations in a declarative part. We were thus able to declare our types and variables in the bodies of main procedures.

However, it is easy to see that this is not going to scale up for real-world applications. We need a better way to structure our programs into modular and distinct units.

Ada encourages the separation of programs into multiple packages and sub-packages, providing many tools to a programmer on a quest for a perfectly organized code-base.

### 4.1 Packages

Here is an example of a package declaration in Ada:

Listing 34: week.ads

```
1 package Week is
2
3     Mon : constant String := "Monday";
4     Tue : constant String := "Tuesday";
5     Wed : constant String := "Wednesday";
6     Thu : constant String := "Thursday";
7     Fri : constant String := "Friday";
8     Sat : constant String := "Saturday";
9     Sun : constant String := "Sunday";
10
11 end Week;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Week  
MD5: 0fa033dc8fe2b9741483de273354e7ee

And here is how you use it:

Listing 35: main.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2 with Week;
3 -- References the Week package, and
4 -- adds a dependency from Main to Week
5
6 procedure Main is
7 begin
8     Put_Line ("First day of the week is "
9               & Week.Mon);
10 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Wek  
MD5: 03e17a75620de6a397b1d3c5a3e22f6a

### Runtime output

First day of the week is Monday

Packages let you make your code modular, separating your programs into semantically significant units. Additionally the separation of a package's specification from its body (which we will see below) can reduce compilation time.

While the `with` clause indicates a dependency, you can see in the example above that you still need to prefix the referencing of entities from the Week package by the name of the package. (If we had included a `use Week` clause, then such a prefix would not have been necessary.)

Accessing entities from a package uses the dot notation, `A.B`, which is the same notation as the one used to access record fields.

A `with` clause can *only* appear in the prelude of a compilation unit (i.e., before the reserved word, such as `procedure`, that marks the beginning of the unit). It is not allowed anywhere else. This rule is only needed for methodological reasons: the person reading your code should be able to see immediately which units the code depends on.

### In other languages

Packages look similar to, but are semantically very different from, header files in C/C++.

- The first and most important distinction is that packages are a language-level mechanism. This is in contrast to a `#include`'d header file, which is a functionality of the C preprocessor.
- An immediate consequence is that the `with` construct is a semantic inclusion mechanism, not a text inclusion mechanism. Hence, when you `with` a package, you are saying to the compiler "I'm depending on this semantic unit", and not "include this bunch of text in place here".
- The effect of a package thus does not vary depending on where it has been `with`ed from. Contrast this with C/C++, where the meaning of the included text depends on the context in which the `#include` appears.

This allows compilation/recompilation to be more efficient. It also allows tools like IDEs to have correct information about the semantics of a program. In turn, this allows better tooling in general, and code that is more analyzable, even by humans.

An important benefit of Ada `with` clauses when compared to `#include` is that it is stateless. The order of `with` and `use` clauses does not matter, and can be changed without side effects.

### In the GNAT toolchain

The Ada language standard does not mandate any particular relationship between source files and packages; for example, in theory you can put all your code in one file, or use your own file naming conventions. In practice, however, an implementation will have specific rules. With GNAT, each top-level compilation unit needs to go into a separate file. In the example above, the Week package will be in an `.ads` file (for Ada specification), and the Main procedure will be in an `.adb` file (for Ada body).

## 4.2 Using a package

As we have seen above, the `with` clause indicates a dependency on another package. However, every reference to an entity coming from the `Week` package had to be prefixed by the full name of the package. It is possible to make every entity of a package visible directly in the current scope, using the `use` clause.

In fact, we have been using the `use` clause since almost the beginning of this tutorial.

Listing 36: main.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2  --          ^ Make every entity of the
3  --          Ada.Text_IO package
4  --          directly visible.
5  with Week;
6
7  procedure Main is
8    use Week;
9    -- Make every entity of the Week
10   -- package directly visible.
11  begin
12    Put_Line ("First day of the week is " & Mon);
13  end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Week  
MD5: ea54077d4ae165b28ae8facfe8ba2db7

### Runtime output

First day of the week is Monday

As you can see in the example above:

- `Put_Line` is a subprogram that comes from the `Ada.Text_IO` package. We can reference it directly because we have `used` the package at the top of the `Main` unit.
- Unlike `with` clauses, a `use` clause can be placed either in the prelude, or in any declarative region. In the latter case the `use` clause will have an effect in its containing lexical scope.

## 4.3 Package body

In the simple example above, the `Week` package only has declarations and no body. That's not a mistake: in a package specification, which is what is illustrated above, you cannot declare bodies. Those have to be in the package body.

Listing 37: operations.ads

```

1  package Operations is
2
3    -- Declaration
4    function Increment_By
5      (I      : Integer;
6       Incr  : Integer := 0) return Integer;
7
8    function Get_Increment_Value return Integer;
9
10   end Operations;

```

Listing 38: operations.adb

```

1 package body Operations is
2
3     Last_Increment : Integer := 1;
4
5     function Increment_By
6         (I      : Integer;
7          Incr : Integer := 0) return Integer is
8
9         begin
10            if Incr /= 0 then
11                Last_Increment := Incr;
12            end if;
13
14            return I + Last_Increment;
15        end Increment_By;
16
17        function Get_Increment_Value return Integer is
18        begin
19            return Last_Increment;
20        end Get_Increment_Value;
21
22    end Operations;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Operations  
MD5: 2adfb64e825605c74fecf6c9d45c8437

Here we can see that the body of the Increment\_By function has to be declared in the body. Coincidentally, introducing a body allows us to put the Last\_Increment variable in the body, and make them inaccessible to the user of the Operations package, providing a first form of encapsulation.

This works because entities declared in the body are *only* visible in the body.

This example shows how Last\_Increment is used indirectly:

Listing 39: main.adb

```

1 with Ada.Text_IO; use Ada.Text_IO;
2 with Operations;
3
4 procedure Main is
5     use Operations;
6
7     I : Integer := 0;
8     R : Integer;
9
10    procedure Display_Update_Values is
11        Incr : constant Integer :=
12            Get_Increment_Value;
13
14        begin
15            Put_Line (Integer'Image (I)
16                      & " incremented by "
17                      & Integer'Image (Incr)
18                      & " is "
19                      & Integer'Image (R));
20            I := R;
21        end Display_Update_Values;
22    begin
23        R := Increment_By (I);

```

(continues on next page)

(continued from previous page)

```

23  Display_Update_Values;
24  R := Increment_By (I);
25  Display_Update_Values;
26
27  R := Increment_By (I, 5);
28  Display_Update_Values;
29  R := Increment_By (I);
30  Display_Update_Values;
31
32  R := Increment_By (I, 10);
33  Display_Update_Values;
34  R := Increment_By (I);
35  Display_Update_Values;
36 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Operations  
MD5: 76190b1261a9652cfb7986ecec191e37

### Runtime output

```

0 incremented by 1 is 1
1 incremented by 1 is 2
2 incremented by 5 is 7
7 incremented by 5 is 12
12 incremented by 10 is 22
22 incremented by 10 is 32

```

## 4.4 Child packages

Packages can be used to create hierarchies. We achieve this by using child packages, which extend the functionality of their parent package. One example of a child package that we've been using so far is the Ada.Text\_Io package. Here, the parent package is called Ada, while the child package is called Text\_Io. In the previous examples, we've been using the Put\_Line procedure from the Text\_Io child package.

#### Important

Ada also supports nested packages. However, since they can be more complicated to use, the recommendation is to use child packages instead. Nested packages will be covered in the advanced course.

Let's begin our discussion on child packages by taking our previous Week package:

Listing 40: week.ads

```

1 package Week is
2
3   Mon : constant String := "Monday";
4   Tue : constant String := "Tuesday";
5   Wed : constant String := "Wednesday";
6   Thu : constant String := "Thursday";
7   Fri : constant String := "Friday";
8   Sat : constant String := "Saturday";
9   Sun : constant String := "Sunday";

```

(continues on next page)

(continued from previous page)

```
10
11 end Week;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Child\_Packages  
MD5: 0fa033dc8fe2b9741483de273354e7ee

If we want to create a child package for Week, we may write:

Listing 41: week-child.ads

```
1 package Week.Child is
2
3     function Get_First_Of_Week return String;
4
5 end Week.Child;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Child\_Packages  
MD5: a7db38e772cf6153b5eb95069517e833

Here, Week is the parent package and Child is the child package. This is the corresponding package body of Week.Child:

Listing 42: week-child.adb

```
1 package body Week.Child is
2
3     function Get_First_Of_Week return String is
4     begin
5         return Mon;
6     end Get_First_Of_Week;
7
8 end Week.Child;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Child\_Packages  
MD5: 04dad82685ad9f0231c3084266b0af83

In the implementation of the Get\_First\_Of\_Week function, we can use the Mon string directly, even though it was declared in the parent package Week. We don't write `with Week` here because all elements from the specification of the Week package — such as Mon, Tue and so on — are visible in the child package Week.Child.

Now that we've completed the implementation of the Week.Child package, we can use elements from this child package in a subprogram by simply writing `with Week.Child`. Similarly, if we want to use these elements directly, we write `use Week.Child` in addition. For example:

Listing 43: main.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2 with Week.Child;  use Week.Child;
3
4 procedure Main is
5 begin
6     Put_Line ("First day of the week is "
```

(continues on next page)

(continued from previous page)

```

7     & Get_First_of_Week);
8 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Child\_Packages  
MD5: e2f5c6ad3a92da4cb04ee7ec12293df4

### Runtime output

First day of the week is Monday

## 4.4.1 Child of a child package

So far, we've seen a two-level package hierarchy. But the hierarchy that we can potentially create isn't limited to that. For instance, we could extend the hierarchy of the previous source code example by declaring a Week.Child.Grandchild package. In this case, Week.Child would be the parent of the Grandchild package. Let's consider this implementation:

Listing 44: week-child-grandchild.ads

```

1 package Week.Child.Grandchild is
2
3     function Get_Second_of_Week return String;
4
5 end Week.Child.Grandchild;

```

Listing 45: week-child-grandchild.adb

```

1 package body Week.Child.Grandchild is
2
3     function Get_Second_of_Week return String is
4     begin
5         return Tue;
6     end Get_Second_of_Week;
7
8 end Week.Child.Grandchild;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Child\_Packages  
MD5: 03ee5932a68212b2e501370212508ab1

We can use this new Grandchild package in our test application in the same way as before: we can reuse the previous test application and adapt the **with** and **use**, and the function call. This is the updated code:

Listing 46: main.adb

```

1 with Ada.Text_IO; use Ada.Text_IO;
2
3 with Week.Child.Grandchild;
4 use Week.Child.Grandchild;
5
6 procedure Main is
7 begin
8     Put_Line ("Second day of the week is "
9               & Get_Second_of_Week);
10 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Child\_Packages  
MD5: 29ee409c8131bd9529c6bf6e366bb390

### Runtime output

Second day of the week is Tuesday

Again, this isn't the limit for the package hierarchy. We could continue to extend the hierarchy of the previous example by implementing a Week.Child.Grandchild.Grand\_grandchild package.

### 4.4.2 Multiple children

So far, we've seen a single child package of a parent package. However, a parent package can also have multiple children. We could extend the example above and implement a Week.Child\_2 package. For example:

Listing 47: week-child\_2.ads

```
1 package Week.Child_2 is
2
3     function Get_Last_Of_Week return String;
4
5 end Week.Child_2;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Child\_Packages  
MD5: bd3f63cacd142d9885600f4000b4573b

Here, Week is still the parent package of the Child package, but it's also the parent of the Child\_2 package. In the same way, Child\_2 is obviously one of the child packages of Week.

This is the corresponding package body of Week.Child\_2:

Listing 48: week-child\_2.adb

```
1 package body Week.Child_2 is
2
3     function Get_Last_Of_Week return String is
4     begin
5         return Sun;
6     end Get_Last_Of_Week;
7
8 end Week.Child_2;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Child\_Packages  
MD5: c2c03e4cb1daff02dd6076c2956ef2aa

We can now reference both children in our test application:

Listing 49: main.adb

```
1 with Ada.Text_Io;  use Ada.Text_Io;
2 with Week.Child;  use Week.Child;
3 with Week.Child_2; use Week.Child_2;
4
5 procedure Main is
```

(continues on next page)

(continued from previous page)

```

6  begin
7    Put_Line ("First day of the week is "
8      & Get_First_Of_Week);
9    Put_Line ("Last day of the week is "
10     & Get_Last_Of_Week);
11 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Child\_Packages  
MD5: 6a91f239fb2a2d8c702409c22467a424

### Runtime output

First day of the week is Monday  
Last day of the week is Sunday

### 4.4.3 Visibility

In the previous section, we've seen that elements declared in a parent package specification are visible in the child package. This is, however, not the case for elements declared in the package body of a parent package.

Let's consider the package Book and its child Additional\_Operations:

Listing 50: book.ads

```

1 package Book is
2
3   Title : constant String := "Visible for my children";
4
5   function Get_Title return String;
6
7   function Get_Author return String;
8
9 end Book;
10

```

Listing 51: book-additional\_operations.ads

```

1 package Book.Additional_Operations is
2
3   function Get_Extended_Title return String;
4
5   function Get_Extended_Author return String;
6
7 end Book.Additional_Operations;
8

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Visibility  
MD5: a0d67cff9aeff288709391d16306df00

This is the body of both packages:

Listing 52: book.adb

```

1 package body Book is
2
3   Author : constant String := "Visible for my children";
4
5 end Book;
6

```

(continues on next page)

(continued from previous page)

```
4      "Author not visible for my children";  
5  
6      function Get_Title return String is  
7      begin  
8          return Title;  
9      end Get_Title;  
10  
11     function Get_Author return String is  
12     begin  
13         return Author;  
14     end Get_Author;  
15  
16 end Book;
```

Listing 53: book-additional\_operations.adb

```
1 package body Book.Additional_Operations is  
2  
3     function Get_Extended_Title return String is  
4     begin  
5         return "Book Title: " & Title;  
6     end Get_Extended_Title;  
7  
8     function Get_Extended_Author return String is  
9     begin  
10        -- "Author" string declared in the body  
11        -- of the Book package is not visible  
12        -- here. Therefore, we cannot write:  
13        --  
14        -- return "Book Author: " & Author;  
15  
16        return "Book Author: Unknown";  
17     end Get_Extended_Author;  
18  
19 end Book.Additional_Operations;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Visibility  
MD5: 68b7490da12bafae0aa6fe0ab76c6b1c

In the implementation of the `Get_Extended_Title`, we're using the `Title` constant from the parent package `Book`. However, as indicated in the comments of the `Get_Extended_Author` function, the `Author` string — which we declared in the body of the `Book` package — isn't visible in the `Book.Additional_Operations` package. Therefore, we cannot use it to implement the `Get_Extended_Author` function.

We can, however, use the `Get_Author` function from `Book` in the implementation of the `Get_Extended_Author` function to retrieve this string. Likewise, we can use this strategy to implement the `Get_Extended_Title` function. This is the adapted code:

Listing 54: book-additional\_operations.adb

```
1 package body Book.Additional_Operations is  
2  
3     function Get_Extended_Title return String is  
4     begin  
5         return "Book Title: " & Get_Title;  
6     end Get_Extended_Title;
```

(continues on next page)

(continued from previous page)

```

8  function Get_Extended_Author return String is
9  begin
10   return "Book Author: " & Get_Author;
11 end Get_Extended_Author;
12
13 end Book.Additional_Operations;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Visibility  
MD5: b00c187cb54d3fcb9574726028c1efc6

This is a simple test application for the packages above:

Listing 55: main.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  with Book.Additional_Operations;
4  use Book.Additional_Operations;
5
6  procedure Main is
7  begin
8   Put_Line (Get_Extended_Title);
9   Put_Line (Get_Extended_Author);
10 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Visibility  
MD5: bdc75987fe61e9401b400f8704890ebe

### Runtime output

Book Title: Visible for my children  
Book Author: Author not visible for my children

By declaring elements in the body of a package, we can implement encapsulation in Ada. Those elements will only be visible in the package body, but nowhere else. This isn't, however, the only way to achieve encapsulation in Ada: we'll discuss other approaches in the *Privacy* (page 111) chapter.

## 4.5 Renaming

Previously, we've mentioned that *subprograms can be renamed* (page 27). We can rename packages, too. Again, we use the **renames** keyword for that. The following example renames the Ada.Text\_IO package as TIO:

Listing 56: main.adb

```

1  with Ada.Text_IO;
2
3  procedure Main is
4   package TIO renames Ada.Text_IO;
5  begin
6   TIO.Put_Line ("Hello");
7  end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Rename\_Text\_IO  
MD5: 33652dd004ef33d95c168ab8893cd412

### Runtime output

Hello

We can use renaming to improve the readability of our code by using shorter package names. In the example above, we write `TI0.Put_Line` instead of the longer version (`Ada.Text_IO.Put_Line`). This approach is especially useful when we don't **use** packages and want to avoid that the code becomes too verbose.

Note we can also rename subprograms and objects inside packages. For instance, we could have just renamed the `Put_Line` procedure in the source code example above:

Listing 57: main.adb

```
1  with Ada.Text_IO;
2
3  procedure Main is
4      procedure Say (Something : String)
5          renames Ada.Text_IO.Put_Line;
6  begin
7      Say ("Hello");
8  end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Modular\_Programming.Rename\_Put\_Line  
MD5: f30174ff29eb01f33bc95f1787f9f1dc

### Runtime output

Hello

In this example, we rename the `Put_Line` procedure to `Say`.

## STRONGLY TYPED LANGUAGE

Ada is a strongly typed language. It is interestingly modern in that respect: strong static typing has been increasing in popularity in programming language design, owing to factors such as the growth of statically typed functional programming, a big push from the research community in the typing domain, and many practical languages with strong type systems.

### 5.1 What is a type?

In statically typed languages, a type is mainly (but not only) a *compile time* construct. It is a construct to enforce invariants about the behavior of a program. Invariants are unchangeable properties that hold for all variables of a given type. Enforcing them ensures, for example, that variables of a data type never have invalid values.

A type is used to reason about the *objects* a program manipulates (an object is a variable or a constant). The aim is to classify objects by what you can accomplish with them (i.e., the operations that are permitted), and this way you can reason about the correctness of the objects' values.

### 5.2 Integers

A nice feature of Ada is that you can define your own integer types, based on the requirements of your program (i.e., the range of values that makes sense). In fact, the definitional mechanism that Ada provides forms the semantic basis for the predefined integer types. There is no "magical" built-in type in that regard, which is unlike most languages, and arguably very elegant.

Listing 58: integer\_type\_example.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Integer_Type_Example is
4   -- Declare a signed integer type,
5   -- and give the bounds
6   type My_Int is range -1 .. 20;
7   --                           ^ High bound
8   --                           ^ Low bound
9
10  -- Like variables, type declarations can
11  -- only appear in declarative regions.
12 begin
13   for I in My_Int loop
14     Put_Line (My_Int'Image (I));
15     --           ^ 'Image attribute
16     --           converts a value
17     --           to a String.
```

(continues on next page)

(continued from previous page)

```
18  end loop;  
19 end Integer_Type_Example;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Integer\_Type\_Example  
MD5: 1d82fa54b604944fdd8652cbf84f4ff2

### Runtime output

```
-1  
0  
1  
2  
3  
4  
5  
6  
7  
8  
9  
10  
11  
12  
13  
14  
15  
16  
17  
18  
19  
20
```

This example illustrates the declaration of a signed integer type, and several things we can do with them.

Every type declaration in Ada starts with the **type** keyword (except for *task types* (page 158)). After the type, we can see a range that looks a lot like the ranges that we use in **for** loops, that defines the low and high bound of the type. Every integer in the inclusive range of the bounds is a valid value for the type.

#### Ada integer types

In Ada, an integer type is not specified in terms of its machine representation, but rather by its range. The compiler will then choose the most appropriate representation.

Another point to note in the above example is the `My_Int'Image (I)` expression. The `Name'Attribute (optional params)` notation is used for what is called an attribute in Ada. An attribute is a built-in operation on a type, a value, or some other program entity. It is accessed by using a ' symbol (the ASCII apostrophe).

Ada has several types available as "built-ins"; **Integer** is one of them. Here is how **Integer** might be defined for a typical processor:

```
type Integer is  
  range -(2 ** 31) .. +(2 ** 31 - 1);
```

`**` is the exponent operator, which means that the first valid value for **Integer** is  $-2^{31}$ , and the last valid value is  $2^{31} - 1$ .

Ada does not mandate the range of the built-in type `Integer`. An implementation for a 16-bit target would likely choose the range  $-2^{15}$  through  $2^{15} - 1$ .

### 5.2.1 Operational semantics

Unlike some other languages, Ada requires that operations on integers should be checked for overflow.

Listing 59: main.adb

```

1  procedure Main is
2    A : Integer := Integer'Last;
3    B : Integer;
4  begin
5    B := A + 5;
6    -- This operation will overflow, eg. it
7    -- will raise an exception at run time.
8  end Main;

```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Overflow\_Check  
MD5: bddd15b394f043442024899d12b982fb

#### Build output

```

main.adb:5:11: warning: value not in range of type "Standard.Integer" [enabled by
  ↪default]
main.adb:5:11: warning: Constraint_Error will be raised at run time [enabled by
  ↪default]

```

#### Runtime output

```
raised CONSTRAINT_ERROR : main.adb:5 overflow check failed
```

There are two types of overflow checks:

- Machine-level overflow, when the result of an operation exceeds the maximum value (or is less than the minimum value) that can be represented in the storage reserved for an object of the type, and
- Type-level overflow, when the result of an operation is outside the range defined for the type.

Mainly for efficiency reasons, while machine-level overflow always results in an exception, type-level overflows will only be checked at specific boundaries, like assignment:

Listing 60: main.adb

```

1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Main is
4    type My_Int is range 1 .. 20;
5    A : My_Int := 12;
6    B : My_Int := 15;
7    M : My_Int := (A + B) / 2;
8    -- No overflow here, overflow checks
9    -- are done at specific boundaries.
10   begin
11     for I in 1 .. M loop
12       Put_Line ("Hello, World!");

```

(continues on next page)

(continued from previous page)

```
13  end loop;
14  -- Loop body executed 13 times
15 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Overflow\_Check\_2  
MD5: d24283cbb42c0be5b5fa215eb16ad2e7

### Runtime output

```
Hello, World!
```

Type-level overflow will only be checked at specific points in the execution. The result, as we see above, is that you might have an operation that overflows in an intermediate computation, but no exception will be raised because the final result does not overflow.

## 5.3 Unsigned types

Ada also features unsigned Integer types. They're called *modular* types in Ada parlance. The reason for this designation is due to their behavior in case of overflow: They simply "wrap around", as if a modulo operation was applied.

For machine sized modular types, for example a modulus of  $2^{32}$ , this mimics the most common implementation behavior of unsigned types. However, an advantage of Ada is that the modulus is more general:

Listing 61: main.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Main is
4      type Mod_Int is mod 2 ** 5;
5      --          ^ Range is 0 .. 31
6
7      A : constant Mod_Int := 20;
8      B : constant Mod_Int := 15;
9
10     M : constant Mod_Int := A + B;
11     -- No overflow here,
12     -- M = (20 + 15) mod 32 = 3
13
14 begin
15     for I in 1 .. M loop
16         Put_Line ("Hello, World!");
17     end loop;
18 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Unsigned\_Types  
MD5: df4efee4eb29e7ea15a0cf961b600dd5

### Runtime output

```
Hello, World!
Hello, World!
Hello, World!
```

Unlike in C/C++, since this wraparound behavior is guaranteed by the Ada specification, you can rely on it to implement portable code. Also, being able to leverage the wrapping on arbitrary bounds is very useful — the modulus does not need to be a power of 2 — to implement certain algorithms and data structures, such as [ring buffers](#)<sup>10</sup>.

## 5.4 Enumerations

Enumeration types are another nicety of Ada's type system. Unlike C's enums, they are *not* integers, and each new enumeration type is incompatible with other enumeration types. Enumeration types are part of the bigger family of discrete types, which makes them usable in certain situations that we will describe later but one context that we have already seen is a case statement.

Listing 62: enumeration\_example.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Enumeration_Example is
4      type Days is (Monday, Tuesday, Wednesday,
5                     Thursday, Friday,
6                     Saturday, Sunday);
7      -- An enumeration type
8  begin
9      for I in Days loop
10         case I is
11             when Saturday .. Sunday =>
12                 Put_Line ("Week end!");
13
14             when Monday .. Friday =>
15                 Put_Line ("Hello on "
16                           & Days'Image (I));
17                 -- 'Image attribute, works on
18                 -- enums too
19         end case;
20     end loop;
21  end Enumeration_Example;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Enumeration\_Example  
MD5: 45d6c83992af4fb6d5015d5f22cb7113

### Runtime output

```
Hello on MONDAY
Hello on TUESDAY
Hello on WEDNESDAY
Hello on THURSDAY
Hello on FRIDAY
```

(continues on next page)

<sup>10</sup> [https://en.wikipedia.org/wiki/Circular\\_buffer](https://en.wikipedia.org/wiki/Circular_buffer)

(continued from previous page)

Week end!  
Week end!

Enumeration types are powerful enough that, unlike in most languages, they're used to define the standard Boolean type:

```
type Boolean is (False, True);
```

As mentioned previously, every "built-in" type in Ada is defined with facilities generally available to the user.

## 5.5 Floating-point types

### 5.5.1 Basic properties

Like most languages, Ada supports floating-point types. The most commonly used floating-point type is **Float**:

Listing 63: floating\_point\_demo.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Floating_Point_Demo is
4   A : constant Float := 2.5;
5 begin
6   Put_Line ("The value of A is "
7             & Float'Image (A));
8 end Floating_Point_Demo;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Floating\_Point\_Demo  
MD5: 06998775497b68b742700138faecbb6a

#### Runtime output

The value of A is 2.50000E+00

The application will display **2.5** as the value of A.

The Ada language does not specify the precision (number of decimal digits in the mantissa) for **Float**; on a typical 32-bit machine the precision will be 6.

All common operations that could be expected for floating-point types are available, including absolute value and exponentiation. For example:

Listing 64: floating\_point\_operations.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Floating_Point_Operations is
4   A : Float := 2.5;
5 begin
6   A := abs (A - 4.5);
7   Put_Line ("The value of A is "
8             & Float'Image (A));
9
10  A := A ** 2 + 1.0;
11  Put_Line ("The value of A is "
```

(continues on next page)

(continued from previous page)

```

12      & Float'Image (A));
13  end Floating_Point_Operations;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Floating\_Point\_Operations  
MD5: c280e0f23e020aaee1a8777e7fb4c242

### Runtime output

```

The value of A is 2.00000E+00
The value of A is 5.00000E+00

```

The value of A is **2.0** after the first operation and **5.0** after the second operation.

In addition to **Float**, an Ada implementation may offer data types with higher precision such as **Long\_Float** and **Long\_Long\_Float**. Like **Float**, the standard does not indicate the exact precision of these types: it only guarantees that the type **Long\_Float**, for example, has at least the precision of **Float**. In order to guarantee that a certain precision requirement is met, we can define custom floating-point types, as we will see in the next section.

## 5.5.2 Precision of floating-point types

Ada allows the user to specify the precision for a floating-point type, expressed in terms of decimal digits. Operations on these custom types will then have at least the specified precision. The syntax for a simple floating-point type declaration is:

```
type T is digits <number_of_decimal_digits>;
```

The compiler will choose a floating-point representation that supports the required precision. For example:

Listing 65: custom\_floating\_types.adb

```

1  with Ada.Text_I0; use Ada.Text_I0;
2
3  procedure Custom_Floating_Types is
4      type T3 is digits 3;
5      type T15 is digits 15;
6      type T18 is digits 18;
7  begin
8      Put_Line ("T3 requires "
9                 & Integer'Image (T3'Size)
10                & " bits");
11      Put_Line ("T15 requires "
12                 & Integer'Image (T15'Size)
13                & " bits");
14      Put_Line ("T18 requires "
15                 & Integer'Image (T18'Size)
16                & " bits");
17  end Custom_Floating_Types;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Custom\_Floating\_Types  
MD5: 3c23738f13e081038996c533da8fb723

### Runtime output

```
T3  requires 32 bits
T15 requires 64 bits
T18 requires 128 bits
```

In this example, the attribute '`Size`' is used to retrieve the number of bits used for the specified data type. As we can see by running this example, the compiler allocates 32 bits for `T3`, 64 bits for `T15` and 128 bits for `T18`. This includes both the mantissa and the exponent.

The number of digits specified in the data type is also used in the format when displaying floating-point variables. For example:

Listing 66: `display_custom_floating_types.adb`

```
1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Display_Custom_Floating_Types is
4      type T3  is digits 3;
5      type T18 is digits 18;
6
7      C1 : constant := 1.0e-4;
8
9      A : constant T3  := 1.0 + C1;
10     B : constant T18 := 1.0 + C1;
11
12 begin
13     Put_Line ("The value of A is "
14                  & T3'Image (A));
15     Put_Line ("The value of B is "
16                  & T18'Image (B));
17 end Display_Custom_Floating_Types;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Display\_Custom\_Floating\_Types  
MD5: 58ec2660388a7f05e139f73e94303cf1

### Runtime output

```
The value of A is 1.00E+00
The value of B is 1.0001000000000000E+00
```

As expected, the application will display the variables according to specified precision (1.00E+00 and 1.0001000000000000E+00).

### 5.5.3 Range of floating-point types

In addition to the precision, a range can also be specified for a floating-point type. The syntax is similar to the one used for integer data types — using the `range` keyword. This simple example creates a new floating-point type based on the type `Float`, for a normalized range between `-1.0` and `1.0`:

Listing 67: `floating_point_range.adb`

```
1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Floating_Point_Range is
4      type T_Norm  is new Float range -1.0 .. 1.0;
5      A : T_Norm;
6
7 begin
8     A := 1.0;
9     Put_Line ("The value of A is "
```

(continues on next page)

(continued from previous page)

```

9      & T_Norm'Image (A));
10 end Floating_Point_Range;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Floating\_Point\_Range  
MD5: b43d596682aa0fa11124a3a3d0596abc

### Runtime output

```
The value of A is 1.00000E+00
```

The application is responsible for ensuring that variables of this type stay within this range; otherwise an exception is raised. In this example, the exception `Constraint_Error` is raised when assigning `2.0` to the variable `A`:

Listing 68: floating\_point\_range\_exception.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Floating_Point_Range_Exception is
4      type T_Norm  is new Float range -1.0 .. 1.0;
5      A : T_Norm;
6  begin
7      A := 2.0;
8      Put_Line ("The value of A is "
9                 & T_Norm'Image (A));
10 end Floating_Point_Range_Exception;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Floating\_Point\_Range\_Exception  
MD5: ecda66589ba28e453956dca159ea5f0d

### Build output

```

floating_point_range_exception.adb:7:09: warning: value not in range of type "T_Norm" defined at line 4 [enabled by default]
floating_point_range_exception.adb:7:09: warning: Constraint_Error will be raised at run time [enabled by default]

```

### Runtime output

```
raised CONSTRAINT_ERROR : floating_point_range_exception.adb:7 range check failed
```

Ranges can also be specified for custom floating-point types. For example:

Listing 69: custom\_range\_types.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2  with Ada.Numerics; use Ada.Numerics;
3
4  procedure Custom_Range_Types is
5      type T6_Inv_Trig is
6          digits 6 range -Pi / 2.0 .. Pi / 2.0;
7  begin
8      null;
9  end Custom_Range_Types;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Custom\_Range\_Types  
MD5: 7b62abc869290a30e351163f670059e0

In this example, we are defining a type called T6\_Inv\_Trig, which has a range from  $-\pi / 2$  to  $\pi / 2$  with a minimum precision of 6 digits. (Pi is defined in the predefined package Ada.Numerics.)

## 5.6 Strong typing

As noted earlier, Ada is strongly typed. As a result, different types of the same family are incompatible with each other; a value of one type cannot be assigned to a variable from the other type. For example:

Listing 70: illegal\_example.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Illegal_Example is
4      -- Declare two different floating point types
5      type Meters is new Float;
6      type Miles  is new Float;
7
8      Dist_Imperial : Miles;
9
10     -- Declare a constant
11     Dist_Metric : constant Meters := 1000.0;
12 begin
13     -- Not correct: types mismatch
14     Dist_Imperial := Dist_Metric * 621.371e-6;
15     Put_Line (Miles'Image (Dist_Imperial));
16 end Illegal_Example;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Imperial\_Metric\_Error  
MD5: e28e341c5eda9b3b4cef691fa24b7f7e

### Build output

```
illegal_example.adb:14:33: error: expected type "Miles" defined at line 6
illegal_example.adb:14:33: error: found type "Meters" defined at line 5
gprbuild: *** compilation phase failed
```

A consequence of these rules is that, in the general case, a "mixed mode" expression like `2 * 3.0` will trigger a compilation error. In a language like C or Python, such expressions are made valid by implicit conversions. In Ada, such conversions must be made explicit:

Listing 71: conv.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2  procedure Conv is
3      type Meters is new Float;
4      type Miles  is new Float;
5      Dist_Imperial : Miles;
6      Dist_Metric : constant Meters := 1000.0;
7 begin
8     Dist_Imperial :=
9         Miles (Dist_Metric) * 621.371e-6;
10    -- ^^^^^^
```

(continues on next page)

(continued from previous page)

```

11  --      Type conversion, from Meters to Miles
12  --      Now the code is correct
13
14  Put_Line (Miles'Image (Dist_Imperial));
15 end Conv;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Imperial\_Metric  
MD5: e455641e86227e80e5f920b5af6315d4

### Runtime output

6.21371E-01

Of course, we probably do not want to write the conversion code every time we convert from meters to miles. The idiomatic Ada way in that case would be to introduce conversion functions along with the types.

Listing 72: conv.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Conv is
4      type Meters is new Float;
5      type Miles  is new Float;
6
7      -- Function declaration, like procedure
8      -- but returns a value.
9      function To_Miles (M : Meters) return Miles is
10         -- ^ Return type
11     begin
12         return Miles (M) * 621.371e-6;
13     end To_Miles;
14
15     Dist_Imperial : Miles;
16     Dist_Metric   : constant Meters := 1000.0;
17 begin
18     Dist_Imperial := To_Miles (Dist_Metric);
19     Put_Line (Miles'Image (Dist_Imperial));
20 end Conv;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Imperial\_Metric\_Func  
MD5: 661737fa9f130ac3070210bbf6f08214

### Runtime output

6.21371E-01

If you write a lot of numeric code, having to explicitly provide such conversions might seem painful at first. However, this approach brings some advantages. Notably, you can rely on the absence of implicit conversions, which will in turn prevent some subtle errors.

#### In other languages

In C, for example, the rules for implicit conversions may not always be completely obvious. In Ada, however, the code will always do exactly what it seems to do. For example:

```
int a = 3, b = 2;
float f = a / b;
```

This code will compile fine, but the result of `f` will be 1.0 instead of 1.5, because the compiler will generate an integer division (three divided by two) that results in one. The software developer must be aware of data conversion issues and use an appropriate casting:

```
int a = 3, b = 2;
float f = (float)a / b;
```

In the corrected example, the compiler will convert both variables to their corresponding floating-point representation before performing the division. This will produce the expected result.

This example is very simple, and experienced C developers will probably notice and correct it before it creates bigger problems. However, in more complex applications where the type declaration is not always visible — e.g. when referring to elements of a `struct` — this situation might not always be evident and quickly lead to software defects that can be harder to find.

The Ada compiler, in contrast, will always reject code that mixes floating-point and integer variables without explicit conversion. The following Ada code, based on the erroneous example in C, will not compile:

Listing 73: main.adb

```
1  procedure Main is
2    A : Integer := 3;
3    B : Integer := 2;
4    F : Float;
5  begin
6    F := A / B;
7  end Main;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Implicit\_Cast  
MD5: 38a8fcc6608c22e22940052ab8dd62f4

#### Build output

```
main.adb:6:11: error: expected type "Standard.Float"
main.adb:6:11: error: found type "Standard.Integer"
gprbuild: *** compilation phase failed
```

The offending line must be changed to `F := Float (A) / Float (B);` in order to be accepted by the compiler.

You can use Ada's strong typing to help enforce invariants in your code, as in the example above: Since Miles and Meters are two different types, you cannot mistakenly convert an instance of one to an instance of the other.

## 5.7 Derived types

In Ada you can create new types based on existing ones. This is very useful: you get a type that has the same properties as some existing type but is treated as a distinct type in the interest of strong typing.

Listing 74: main.adb

```

1  procedure Main is
2    -- ID card number type,
3    -- incompatible with Integer.
4    type Social_Security_Number is new Integer
5      range 0 .. 999_99_9999;
6      -- ^ Since a SSN has 9 digits
7      -- max., and cannot be
8      -- negative, we enforce
9      -- a validity constraint.
10
11   SSN : Social_Security_Number := 555_55_5555;
12   -- ^ You can put underscores as
13   -- formatting in any number.
14
15   I : Integer;
16
17   -- The value -1 below will cause a
18   -- runtime error and a compile time
19   -- warning with GNAT.
20   Invalid : Social_Security_Number := -1;
21
22 begin
23   -- Illegal, they have different types:
24   I := SSN;
25
26   -- Likewise illegal:
27   SSN := I;
28
29   -- OK with explicit conversion:
30   I := Integer (SSN);
31
32   -- Likewise OK:
33   SSN := Social_Security_Number (I);
34 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Derived\_Types  
MD5: 63445601ddb5e52dceab095d3305623a

### Build output

```

main.adb:21:40: warning: value not in range of type "Social_Security_Number" ↴
  ↴defined at line 4 [enabled by default]
main.adb:21:40: warning: Constraint_Error will be raised at run time [enabled by ↴
  ↴default]
main.adb:24:09: error: expected type "Standard.Integer"
main.adb:24:09: error: found type "Social_Security_Number" defined at line 4
main.adb:27:11: error: expected type "Social_Security_Number" defined at line 4
main.adb:27:11: error: found type "Standard.Integer"
gprbuild: *** compilation phase failed

```

The type `Social_Security` is said to be a *derived type*; its *parent type* is `Integer`.

As illustrated in this example, you can refine the valid range when defining a derived scalar type (such as integer, floating-point and enumeration).

The syntax for enumerations uses the `range <range>` syntax:

Listing 75: greet.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Greet is
4     type Days is (Monday, Tuesday, Wednesday,
5                    Thursday, Friday,
6                    Saturday, Sunday);
7
8     type Weekend_Days is new
9         Days range Saturday .. Sunday;
10        -- New type, where only Saturday and Sunday
11        -- are valid literals.
12 begin
13     null;
14 end Greet;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Days  
MD5: 853b5c1576961c7c20d4306275122364

## 5.8 Subtypes

As we are starting to see, types may be used in Ada to enforce constraints on the valid range of values. However, we sometimes want to enforce constraints on some values while staying within a single type. This is where subtypes come into play. A subtype does not introduce a new type.

Listing 76: greet.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Greet is
4     type Days is (Monday, Tuesday, Wednesday,
5                    Thursday, Friday,
6                    Saturday, Sunday);
7
8     -- Declaration of a subtype
9     subtype Weekend_Days is
10        Days range Saturday .. Sunday;
11        -- ^ Constraint of the subtype
12
13     M : Days := Sunday;
14
15     S : Weekend_Days := M;
16     -- No error here, Days and Weekend_Days
17     -- are of the same type.
18 begin
19     for I in Days loop
20         case I is
21             -- Just like a type, a subtype can
22             -- be used as a range
23             when Weekend_Days =>
24                 Put_Line ("Week end!");
25             when others =>
26                 Put_Line ("Hello on "
27                           & Days'Image (I));
28         end case;

```

(continues on next page)

(continued from previous page)

```
29      end loop;
30  end Greet;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Days\_Subtype  
MD5: 8ee7127d152a8b2c9d0ac74d05fc2fc2

### Runtime output

```
Hello on MONDAY
Hello on TUESDAY
Hello on WEDNESDAY
Hello on THURSDAY
Hello on FRIDAY
Week end!
Week end!
```

Several subtypes are predefined in the standard package in Ada, and are automatically available to you:

```
subtype Natural is Integer range 0 .. Integer'Last;
subtype Positive is Integer range 1 .. Integer'Last;
```

While subtypes of a type are statically compatible with each other, constraints are enforced at run time: if you violate a subtype constraint, an exception will be raised.

Listing 77: greet.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Greet is
4    type Days is (Monday, Tuesday, Wednesday,
5                  Thursday, Friday,
6                  Saturday, Sunday);
7
8    subtype Weekend_Days is
9      Days range Saturday .. Sunday;
10
11   Day      : Days := Saturday;
12   Weekend : Weekend_Days;
13 begin
14   Weekend := Day;
15   --      ^ Correct: Same type, subtype
16   --      constraints are respected
17   Weekend := Monday;
18   --      ^ Wrong value for the subtype
19   --      Compiles, but exception at runtime
20 end Greet;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Days\_Subtype\_Error  
MD5: 84d42d276d26544f35edab5870459378

### Build output

```
greet.adb:17:15: warning: value not in range of type "Weekend_Days" defined at line 8 [enabled by default]
greet.adb:17:15: warning: Constraint_Error will be raised at run time [enabled by default]
```

### Runtime output

```
raised CONSTRAINT_ERROR : greet.adb:17 range check failed
```

### 5.8.1 Subtypes as type aliases

Previously, we've seen that we can create new types by declaring e.g. `type Miles is new Float`. We could also create type aliases, which generate alternative names — *aliases* — for known types. Note that type aliases are sometimes called *type synonyms*.

We achieve this in Ada by using subtypes without new constraints. In this case, however, we don't get all of the benefits of Ada's strong type checking. Let's rewrite an example using type aliases:

Listing 78: undetected\_imperial\_metric\_error.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Undetected_Imperial_Metric_Error is
4      -- Declare two type aliases
5      subtype Meters is Float;
6      subtype Miles is Float;
7
8      Dist_Imperial : Miles;
9
10     -- Declare a constant
11     Dist_Metric : constant Meters := 100.0;
12
13 begin
14     -- No conversion to Miles type required:
15     Dist_Imperial := (Dist_Metric * 1609.0)
16                  / 1000.0;
17
18     -- Not correct, but undetected:
19     Dist_Imperial := Dist_Metric;
20
21     Put_Line (Miles'Image (Dist_Imperial));
22 end Undetected_Imperial_Metric_Error;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Strongly\_Typed\_Language.Undetected\_Imperial\_Metric\_Error  
MD5: cdb8f949c69f3c480502b859dac298ee

### Runtime output

```
1.00000E+02
```

In the example above, the fact that both `Meters` and `Miles` are subtypes of `Float` allows us to mix variables of both types without type conversion. This, however, can lead to all sorts of programming mistakes that we'd like to avoid, as we can see in the undetected error highlighted in the code above. In that example, the error in the assignment of a value in meters to a variable meant to store values in miles remains undetected because both `Meters` and `Miles` are subtypes of `Float`. Therefore, the recommendation is to use strong typing — via `type X is new Y` — for cases such as the one above.

There are, however, many situations where type aliases are useful. For example, in an application that uses floating-point types in multiple contexts, we could use type aliases to indicate additional meaning to the types or to avoid long variable names. For example, instead of writing:

```
Paid_Amount, Due_Amount : Float;
```

We could write:

```
subtype Amount is Float;
```

```
Paid, Due : Amount;
```

### In other languages

In C, for example, we can use a **typedef** declaration to create a type alias. For example:

```
typedef float meters;
```

This corresponds to the declaration that we've seen above using subtypes. Other programming languages include this concept in similar ways. For example:

- C++: `using meters = float;`
- Swift: `typealias Meters = Double`
- Kotlin: `typealias Meters = Double`
- Haskell: `type Meters = Float`

Note, however, that subtypes in Ada correspond to type aliases if, and only if, they don't have new constraints. Thus, if we add a new constraint to a subtype declaration, we don't have a type alias anymore. For example, the following declaration *can't* be considered a type alias of **Float**:

```
subtype Meters is Float range 0.0 .. 1_000_000.0;
```

Let's look at another example:

```
subtype Degree_Celsius is Float;
```

```
subtype Liquid_Water_Temperature is
  Degree_Celsius range 0.0 .. 100.0;
```

```
subtype Running_Water_Temperature is
  Liquid_Water_Temperature;
```

In this example, `Liquid_Water_Temperature` isn't an alias of `Degree_Celsius`, since it adds a new constraint that wasn't part of the declaration of the `Degree_Celsius`. However, we do have two type aliases here:

- `Degree_Celsius` is an alias of **Float**;
- `Running_Water_Temperature` is an alias of `Liquid_Water_Temperature`, even if `Liquid_Water_Temperature` itself has a constrained range.



## RECORDS

So far, all the types we have encountered have values that are not decomposable: each instance represents a single piece of data. Now we are going to see our first class of composite types: records.

Records allow composing a value out of instances of other types. Each of those instances will be given a name. The pair consisting of a name and an instance of a specific type is called a field, or a component.

### 6.1 Record type declaration

Here is an example of a simple record declaration:

```
type Date is record
  -- The following declarations are
  -- components of the record
  Day : Integer range 1 .. 31;
  Month : Months;
  -- You can add custom constraints
  -- on fields
  Year : Integer range 1 .. 3000;
end record;
```

Fields look a lot like variable declarations, except that they are inside of a record definition. And as with variable declarations, you can specify additional constraints when supplying the subtype of the field.

```
type Date is record
  Day : Integer range 1 .. 31;
  Month : Months := January;
  -- This component has a default value
  Year : Integer range 1 .. 3000 := 2012;
  --
  -- Default value
end record;
```

Record components can have default values. When a variable having the record type is declared, a field with a default initialization will be automatically set to this value. The value can be any expression of the component type, and may be run-time computable.

In the remaining sections of this chapter, we see how to use record types. In addition to that, we discuss more about records in [another chapter](#) (page 97).

## 6.2 Aggregates

```
-- Positional components
Ada_Birthday    : Date := (10, December, 1815);

-- Named components
Leap_Day_2020   : Date := (Day    => 29,
                           Month  => February,
                           Year   => 2020);
-- ^ By name
```

Records have a convenient notation for expressing values, illustrated above. This notation is called aggregate notation, and the literals are called aggregates. They can be used in a variety of contexts that we will see throughout the course, one of which is to initialize records.

An aggregate is a list of values separated by commas and enclosed in parentheses. It is allowed in any context where a value of the record is expected.

Values for the components can be specified positionally, as in Ada\_Birthday example, or by name, as in Leap\_Day\_2020. A mixture of positional and named values is permitted, but you cannot use a positional notation after a named one.

## 6.3 Component selection

To access components of a record instance, you use an operation that is called component selection. This is achieved by using the dot notation. For example, if we declare a variable Some\_Day of the Date record type mentioned above, we can access the Year component by writing Some\_Day.Year.

Let's look at an example:

Listing 79: record\_selection.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Record_Selection is
4
5    type Months is
6      (January, February, March, April,
7       May, June, July, August, September,
8       October, November, December);
9
10   type Date is record
11     Day   : Integer range 1 .. 31;
12     Month : Months;
13     Year  : Integer range 1 .. 3000 := 2032;
14   end record;
15
16   procedure Display_Date (D : Date) is
17   begin
18     Put_Line ("Day:" & Integer'Image (D.Day)
19               & ", Month: "
20               & Months'Image (D.Month)
21               & ", Year: "
22               & Integer'Image (D.Year));
23   end Display_Date;
24
25   Some_Day : Date := (1, January, 2000);
26
```

(continues on next page)

(continued from previous page)

```

27 begin
28   Display_Date (Some_Day);
29
30   Put_Line ("Changing year... ");
31   Some_Day.Year := 2001;
32
33   Display_Date (Some_Day);
34 end Record_Selection;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Records.Record\_Selection  
MD5: 79602cf4d011ba7423d07772b13e2b5a

### Runtime output

```

Day: 1, Month: JANUARY, Year: 2000
Changing year...
Day: 1, Month: JANUARY, Year: 2001

```

As you can see in this example, we can use the dot notation in the expression `D.Year` or `Some_Day.Year` to access the information stored in that component, as well as to modify this information in assignments. To be more specific, when we use `D.Year` in the call to `Put_Line`, we're retrieving the information stored in that component. When we write `Some_Day.Year := 2001`, we're overwriting the information that was previously stored in the `Year` component of `Some_Day`.

## 6.4 Renaming

In previous chapters, we've discussed *subprogram* (page 27) and *package* (page 41) renaming. We can rename record components as well. Instead of writing the full component selection using the dot notation, we can declare an alias that allows us to access the same component. This is useful to simplify the implementation of a subprogram, for example.

We can rename record components by using the `renames` keyword in a variable declaration. For example:

```

Some_Day : Date;
Y         : Integer renames Some_Day.Year;

```

Here, `Y` is an alias, so that every time we use `Y`, we are really using the `Year` component of `Some_Day`.

Let's look at a complete example:

Listing 80: dates.ads

```

1 package Dates is
2
3   type Months is
4     (January, February, March, April,
5      May, June, July, August, September,
6      October, November, December);
7
8   type Date is record
9     Day   : Integer range 1 .. 31;
10    Month : Months;
11    Year  : Integer range 1 .. 3000 := 2032;
12  end record;

```

(continues on next page)

(continued from previous page)

```
13  procedure Increase_Month
14  (Some_Day : in out Date);
15
16  procedure Display_Month
17  (Some_Day : Date);
18
19
20 end Dates;
```

Listing 81: dates.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  package body Dates is
4
5    procedure Increase_Month
6    (Some_Day : in out Date)
7    is
8      -- Renaming components from
9      -- the Date record
10     M : Months renames Some_Day.Month;
11     Y : Integer renames Some_Day.Year;
12
13     -- Renaming function (for Months
14     -- enumeration)
15     function Next (M : Months)
16       return Months
17     renames Months'Succ;
18
19 begin
20   if M = December then
21     M := January;
22     Y := Y + 1;
23   else
24     M := Next (M);
25   end if;
26 end Increase_Month;
27
28  procedure Display_Month
29  (Some_Day : Date)
30
31  is
32    -- Renaming components from
33    -- the Date record
34    M : Months renames Some_Day.Month;
35    Y : Integer renames Some_Day.Year;
36
37 begin
38   Put_Line ("Month: "
39             & Months'Image (M)
40             & ", Year: "
41             & Integer'Image (Y));
42 end Display_Month;
43
44 end Dates;
```

Listing 82: main.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2  with Dates;      use Dates;
3
4  procedure Main is
5    D : Date := (1, January, 2000);
```

(continues on next page)

(continued from previous page)

```
6  begin
7      Display_Month (D);
8
9      Put_Line ("Increasing month...");
10     Increase_Month (D);
11
12     Display_Month (D);
13 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Record\_Component\_Renaming  
MD5: 905390bd02b8417039052218800975a3

### Runtime output

```
Month: JANUARY, Year: 2000
Increasing month...
Month: FEBRUARY, Year: 2000
```

We apply renaming to two components of the Date record in the implementation of the Increase\_Month procedure. Then, instead of directly using Some\_Day.Month and Some\_Day.Year in the next operations, we simply use the renamed versions M and Y.

Note that, in the example above, we also rename Months' `Succ` — which is the function that gives us the next month — to `Next`.



## ARRAYS

Arrays provide another fundamental family of composite types in Ada.

### 7.1 Array type declaration

Arrays in Ada are used to define contiguous collections of elements that can be selected by indexing. Here's a simple example:

Listing 83: greet.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Greet is
4      type My_Int is range 0 .. 1000;
5      type Index is range 1 .. 5;
6
7      type My_Int_Array is
8          array (Index) of My_Int;
9          --          ^ Type of elements
10         --          ^ Bounds of the array
11     Arr : My_Int_Array := (2, 3, 5, 7, 11);
12         --          ^ Array literal
13         --          (aggregate)
14
15     V : My_Int;
16 begin
17     for I in Index loop
18         V := Arr (I);
19         --         ^ Take the Ith element
20         Put (My_Int'Image (V));
21     end loop;
22     New_Line;
23 end Greet;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Greet  
MD5: ffdd2ba2322b0946dfcac3a55bce5270

#### Runtime output

```
2 3 5 7 11
```

The first point to note is that we specify the index type for the array, rather than its size. Here we declared an integer type named `Index` ranging from `1` to `5`, so each array instance will have 5 elements, with the initial element at index 1 and the last element at index 5.

Although this example used an integer type for the index, Ada is more general: any discrete type is permitted to index an array, including [Enum types](#) (page 47). We will soon see what that means.

Another point to note is that querying an element of the array at a given index uses the same syntax as for function calls: that is, the array object followed by the index in parentheses.

Thus when you see an expression such as `A (B)`, whether it is a function call or an array subscript depends on what `A` refers to.

Finally, notice how we initialize the array with the `(2, 3, 5, 7, 11)` expression. This is another kind of aggregate in Ada, and is in a sense a literal expression for an array, in the same way that `3` is a literal expression for an integer. The notation is very powerful, with a number of properties that we will introduce later. A detailed overview appears in the notation of [aggregate types](#) (page 85).

Unrelated to arrays, the example also illustrated two procedures from `Ada.Text_Io`:

- `Put`, which displays a string without a terminating end of line
- `New_Line`, which outputs an end of line

Let's now delve into what it means to be able to use any discrete type to index into the array.

### In other languages

Semantically, an array object in Ada is the entire data structure, and not simply a handle or pointer. Unlike C and C++, there is no implicit equivalence between an array and a pointer to its initial element.

Listing 84: `array_bounds_example.adb`

```
1 with Ada.Text_Io; use Ada.Text_Io;
2
3 procedure Array_Bounds_Example is
4   type My_Int is range 0 .. 1000;
5
6   type Index is range 11 .. 15;
7   --           ^ Low bound can
8   --           be any value
9
10  type My_Int_Array is
11    array (Index) of My_Int;
12
13  Tab : constant My_Int_Array :=
14    (2, 3, 5, 7, 11);
15
16 begin
17   for I in Index loop
18     Put (My_Int'Image (Tab (I)));
19   end loop;
20   New_Line;
21 end Array_Bounds_Example;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Array\_Bounds\_Example  
MD5: e5fe9e7b83055f3ae23dd890e29c22de

### Runtime output

2 3 5 7 11

One effect is that the bounds of an array can be any values. In the first example we constructed an array type whose first index is 1, but in the example above we declare an array type whose first index is 11.

That's perfectly fine in Ada, and moreover since we use the index type as a range to iterate over the array indices, the code using the array does not need to change.

That leads us to an important consequence with regard to code dealing with arrays. Since the bounds can vary, you should not assume / hard-code specific bounds when iterating / using arrays. That means the code above is good, because it uses the index type, but a for loop as shown below is bad practice even though it works correctly:

```
for I in 11 .. 15 loop
    Tab (I) := Tab (I) * 2;
end loop;
```

Since you can use any discrete type to index an array, enumeration types are permitted.

Listing 85: month\_example.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Month_Example is
4      type Month_Duration is range 1 .. 31;
5      type Month is (Jan, Feb, Mar, Apr,
6                      May, Jun, Jul, Aug,
7                      Sep, Oct, Nov, Dec);
8
9      type My_Int_Array is
10         array (Month) of Month_Duration;
11         --      ^ Can use an enumeration type
12         --      as the index
13
14      Tab : constant My_Int_Array :=
15          --      ^ constant is like a variable but
16          --      cannot be modified
17          (31, 28, 31, 30, 31, 30,
18          31, 31, 30, 31, 30, 31);
19          -- Maps months to number of days
20          -- (ignoring leap years)
21
22      Feb_Days : Month_Duration := Tab (Feb);
23          -- Number of days in February
24 begin
25     for M in Month loop
26         Put_Line
27             (Month'Image (M) & " has "
28             & Month_Duration'Image (Tab (M))
29             & " days.");
30             --      ^ Concatenation operator
31     end loop;
32 end Month_Example;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Month\_Example  
MD5: 420bb8faa36d0efd3d071c76c2033d21

### Runtime output

```
JAN has 31 days.
FEB has 28 days.
MAR has 31 days.
```

(continues on next page)

(continued from previous page)

```
APR has 30 days.  
MAY has 31 days.  
JUN has 30 days.  
JUL has 31 days.  
AUG has 31 days.  
SEP has 30 days.  
OCT has 31 days.  
NOV has 30 days.  
DEC has 31 days.
```

In the example above, we are:

- Creating an array type mapping months to month durations in days.
- Creating an array, and instantiating it with an aggregate mapping months to their actual durations in days.
- Iterating over the array, printing out the months, and the number of days for each.

Being able to use enumeration values as indices is very helpful in creating mappings such as shown above one, and is an often used feature in Ada.

## 7.2 Indexing

We have already seen the syntax for selecting elements of an array. There are however a few more points to note.

First, as is true in general in Ada, the indexing operation is strongly typed. If you use a value of the wrong type to index the array, you will get a compile-time error.

Listing 86: greet.adb

```
1  with Ada.Text_Io; use Ada.Text_Io;  
2  
3  procedure Greet is  
4      type My_Int is range 0 .. 1000;  
5  
6      type My_Index is range 1 .. 5;  
7      type Your_Index is range 1 .. 5;  
8  
9      type My_Int_Array is  
10         array (My_Index) of My_Int;  
11  
12     Tab : My_Int_Array := (2, 3, 5, 7, 11);  
13  begin  
14      for I in Your_Index loop  
15          Put (My_Int'Image (Tab (I)));  
16          --                                         ^ Compile time error  
17      end loop;  
18      New_Line;  
19  end Greet;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Greet\_2  
MD5: 54543017e4ec69d24bf9e43d507b50e6

### Build output

```

greet.adb:15:31: error: expected type "My_Index" defined at line 6
greet.adb:15:31: error: found type "Your_Index" defined at line 7
gprbuild: *** compilation phase failed

```

Second, arrays in Ada are bounds checked. This means that if you try to access an element outside of the bounds of the array, you will get a run-time error instead of accessing random memory as in unsafe languages.

Listing 87: greet.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Greet is
4      type My_Int is range 0 .. 1000;
5      type Index  is range 1 .. 5;
6
7      type My_Int_Array is
8          array (Index) of My_Int;
9
10     Tab : My_Int_Array := (2, 3, 5, 7, 11);
11 begin
12     for I in Index range 2 .. 6 loop
13         Put (My_Int'Image (Tab (I)));
14         --                                     ^ Will raise an
15         --                                     exception when
16         --                                     I = 6
17     end loop;
18     New_Line;
19 end Greet;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Greet\_3  
MD5: 0102674d089be838f1dfbf0791d99fce

### Build output

```

greet.adb:12:30: warning: static value out of range of type "Index" defined at line 5 [enabled by default]
greet.adb:12:30: warning: Constraint_Error will be raised at run time [enabled by default]
greet.adb:12:30: warning: suspicious loop bound out of range of loop subtype [enabled by default]
greet.adb:12:30: warning: loop executes zero times or raises Constraint_Error [enabled by default]

```

### Runtime output

```
raised CONSTRAINT_ERROR : greet.adb:12 range check failed
```

## 7.3 Simpler array declarations

In the previous examples, we have always explicitly created an index type for the array. While this can be useful for typing and readability purposes, sometimes you simply want to express a range of values. Ada allows you to do that, too.

Listing 88: simple\_array\_bounds.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Simple_Array_Bounds is
4     type My_Int is range 0 .. 1000;
5
6     type My_Int_Array is
7         array (1 .. 5) of My_Int;
8         --          ^ Subtype of Integer
9
10    Tab : constant My_Int_Array :=
11        (2, 3, 5, 7, 11);
12
13 begin
14     for I in 1 .. 5 loop
15         --          ^ Subtype of Integer
16         Put (My_Int'Image (Tab (I)));
17     end loop;
18     New_Line;
19 end Simple_Array_Bounds;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Simple\_Array\_Bounds  
MD5: c337a7fe0dacc5f60f7b234aa96d39

### Runtime output

```
2 3 5 7 11
```

This example defines the range of the array via the range syntax, which specifies an anonymous subtype of Integer and uses it to index the array.

This means that the type of the index is **Integer**. Similarly, when you use an anonymous range in a for loop as in the example above, the type of the iteration variable is also **Integer**, so you can use I to index Tab.

You can also use a named subtype for the bounds for an array.

## 7.4 Range attribute

We noted earlier that hard coding bounds when iterating over an array is a bad idea, and showed how to use the array's index type/subtype to iterate over its range in a **for** loop. That raises the question of how to write an iteration when the array has an anonymous range for its bounds, since there is no name to refer to the range. Ada solves that via several attributes of array objects:

Listing 89: range\_example.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Range_Example is
4     type My_Int is range 0 .. 1000;
5
6     type My_Int_Array is
7         array (1 .. 5) of My_Int;
8
9     Tab : constant My_Int_Array :=
10        (2, 3, 5, 7, 11);
11
12 begin
```

(continues on next page)

(continued from previous page)

```

12  for I in Tab'Range loop
13      -- ^ Gets the range of Tab
14      Put (My_Int'Image (Tab (I)));
15  end loop;
16  New_Line;
17 end Range_Example;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Range\_Example  
MD5: 8b0d7bf346cb59999dfd12dbaaf3e2a6

### Runtime output

```
2 3 5 7 11
```

If you want more fine grained control, you can use the separate attributes '`First`' and '`Last`'.

Listing 90: `array_attributes_example.adb`

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Array_Attributes_Example is
4      type My_Int is range 0 .. 1000;
5
6      type My_Int_Array is
7          array (1 .. 5) of My_Int;
8
9      Tab : My_Int_Array := (2, 3, 5, 7, 11);
10
11 begin
12     for I in Tab'First .. Tab'Last - 1 loop
13         -- ^ Iterate on every index
14         -- except the last
15         Put (My_Int'Image (Tab (I)));
16     end loop;
17     New_Line;
18 end Array_Attributes_Example;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Array\_Attributes\_Example  
MD5: 95cc407c8aadd936e050fe3505e8fb46

### Runtime output

```
2 3 5 7
```

The '`Range`', '`First`' and '`Last`' attributes in these examples could also have been applied to the array type name, and not just the array instances.

Although not illustrated in the above examples, another useful attribute for an array instance `A` is `A'Length`, which is the number of elements that `A` contains.

It is legal and sometimes useful to have a "null array", which contains no elements. To get this effect, define an index range whose upper bound is less than the lower bound.

## 7.5 Unconstrained arrays

Let's now consider one of the most powerful aspects of Ada's array facility.

Every array type we have defined so far has a fixed size: every instance of this type will have the same bounds and therefore the same number of elements and the same size.

However, Ada also allows you to declare array types whose bounds are not fixed: in that case, the bounds will need to be provided when creating instances of the type.

Listing 91: unconstrained\_array\_example.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Unconstrained_Array_Example is
4      type Days is (Monday, Tuesday, Wednesday,
5                     Thursday, Friday,
6                     Saturday, Sunday);
7
8      type Workload_Type is
9          array (Days range <>) of Natural;
10         -- Indefinite array type
11         -- ^ Bounds are of type Days,
12         -- but not known
13
14      Workload : constant
15          Workload_Type (Monday .. Friday) :=
16              -- ^ Specify the bounds
17              -- when declaring
18              (Friday => 7, others => 8);
19              -- ^ Default value
20              -- ^ Specify element by name of index
21 begin
22     for I in Workload'Range loop
23         Put_Line (Integer'Image (Workload (I)));
24     end loop;
25 end Unconstrained_Array_Example;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Unconstrained\_Array\_Example  
MD5: c84910e9b424cfabbbbe018ba0a6de59

### Runtime output

```

8
8
8
8
7

```

The fact that the bounds of the array are not known is indicated by the `Days range <>` syntax. Given a discrete type `Discrete_Type`, if we use `Discrete_Type` for the index in an array type then `Discrete_Type` serves as the type of the index and comprises the range of index values for each array instance.

If we define the index as `Discrete_Type range <>` then `Discrete_Type` serves as the type of the index, but different array instances may have different bounds from this type.

An array type that is defined with the `Discrete_Type range <>` syntax for its index is referred to as an unconstrained array type, and, as illustrated above, the bounds need to be provided when an instance is created.

The above example also shows other forms of the aggregate syntax. You can specify associations by name, by giving the value of the index on the left side of an arrow association. `1 => 2` thus means "assign value 2 to the element at index 1 in my array". `others => 8` means "assign value 8 to every element that wasn't previously assigned in this aggregate".

### ⚠ Attention

The so-called "box" notation (`<>`) is commonly used as a wildcard or placeholder in Ada. You will often see it when the meaning is "what is expected here can be anything".

### ⓘ In other languages

While unconstrained arrays in Ada might seem similar to variable length arrays in C, they are in reality much more powerful, because they're truly first-class values in the language. You can pass them as parameters to subprograms or return them from functions, and they implicitly contain their bounds as part of their value. This means that it is useless to pass the bounds or length of an array explicitly along with the array, because they are accessible via the `'First`, `'Last`, `'Range` and `'Length` attributes explained earlier.

Although different instances of the same unconstrained array type can have different bounds, a specific instance has the same bounds throughout its lifetime. This allows Ada to implement unconstrained arrays efficiently; instances can be stored on the stack and do not require heap allocation as in languages like Java.

## 7.6 Predefined array type: String

A recurring theme in our introduction to Ada types has been the way important built-in types like `Boolean` or `Integer` are defined through the same facilities that are available to the user. This is also true for strings: The `String` type in Ada is a simple array.

Here is how the string type is defined in Ada:

```
type String is
  array (Positive range <>) of Character;
```

The only built-in feature Ada adds to make strings more ergonomic is custom literals, as we can see in the example below.

### 💡 Hint

String literals are a syntactic sugar for aggregates, so that in the following example, A and B have the same value.

Listing 92: string\_literals.ads

```
1 package String_Literals is
2   -- Those two declarations are equivalent
3   A : String (1 .. 11) := "Hello World";
4   B : String (1 .. 11) :=
5     ('H', 'e', 'l', 'l', 'o', ' ', '
6     'W', 'o', 'r', 'l', 'd');
7 end String_Literals;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.String\_Literals  
MD5: 8e5871c8ead4ff8da643539857e23b30

Listing 93: greet.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Greet is
4     Message : String (1 .. 11) := "dlrow olleH";
5     --          ^ Pre-defined array type.
6     --          Component type is Character
7 begin
8     for I in reverse Message'Range loop
9         --          ^ Iterate in reverse order
10        Put (Message (I));
11    end loop;
12    New_Line;
13 end Greet;
```

However, specifying the bounds of the object explicitly is a bit of a hassle; you have to manually count the number of characters in the literal. Fortunately, Ada gives you an easier way.

You can omit the bounds when creating an instance of an unconstrained array type if you supply an initialization, since the bounds can be deduced from the initialization expression.

Listing 94: greet.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Greet is
4     Message : constant String := "dlrow olleH";
5     --          ^ Bounds are automatically
6     --          computed from
7     --          initialization value
8 begin
9     for I in reverse Message'Range loop
10        Put (Message (I));
11    end loop;
12    New_Line;
13 end Greet;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Greet\_5  
MD5: 21448a1007a07ec9d434880628625c3f

### Runtime output

Hello World

Listing 95: main.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Main is
4     type Integer_Array is
5         array (Natural range <>) of Integer;
6
7     My_Array : constant Integer_Array :=
```

(continues on next page)

(continued from previous page)

```

8      (1, 2, 3, 4);
9      ^^^^^^^^^^^^^^
10     -- Bounds are automatically
11     -- computed from
12     -- initialization value
13 begin
14   null;
15 end Main;

```

### ⚠ Attention

As you can see above, the standard **String** type in Ada is an array. As such, it shares the advantages and drawbacks of arrays: a **String** value is stack allocated, it is accessed efficiently, and its bounds are immutable.

If you want something akin to C++'s `std::string`, you can use *Unbounded Strings* (page 245) from Ada's standard library. This type is more like a mutable, automatically managed string buffer to which you can add content.

## 7.7 Restrictions

A very important point about arrays: bounds *have* to be known when instances are created. It is for example illegal to do the following.

```

declare
  A : String;
begin
  A := "World";
end;

```

Also, while you of course can change the values of elements in an array, you cannot change the array's bounds (and therefore its size) after it has been initialized. So this is also illegal:

```

declare
  A : String := "Hello";
begin
  A := "World";    -- OK: Same size
  A := "Hello World"; -- Not OK: Different size
end;

```

Also, while you can expect a warning for this kind of error in very simple cases like this one, it is impossible for a compiler to know in the general case if you are assigning a value of the correct length, so this violation will generally result in a run-time error.

### ℹ Attention

While we will learn more about this later, it is important to know that arrays are not the only types whose instances might be of unknown size at compile-time.

Such objects are said to be of an *indefinite subtype*, which means that the subtype size is not known at compile time, but is dynamically computed (at run time).

Listing 96: indefinite\_subtypes.adb

```
1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Indefinite_Subtypes is
4      function Get_Number return Integer is
5      begin
6          return Integer'Value (Get_Line);
7      end Get_Number;
8
9      A : String := "Hello";
10     -- Indefinite subtype
11
12     B : String (1 .. 5) := "Hello";
13     -- Definite subtype
14
15     C : String (1 .. Get_Number);
16     -- Indefinite subtype
17     -- (Get_Number's value is computed at
18     -- run-time)
19
20 begin
21     null;
22 end Indefinite_Subtypes;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Indefinite\_Subtypes  
MD5: a24235838511a94879f74757421a28f0

Here, the 'Value attribute converts the string to an integer.

## 7.8 Returning unconstrained arrays

The return type of a function can be any type; a function can return a value whose size is unknown at compile time. Likewise, the parameters can be of any type.

For example, this is a function that returns an unconstrained **String**:

Listing 97: main.adb

```
1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Main is
4
5      type Days is (Monday, Tuesday, Wednesday,
6                     Thursday, Friday,
7                     Saturday, Sunday);
8
9      function Get_Day_Name (Day : Days := Monday)
10                     return String is
11
12 begin
13     return
14         (case Day is
15             when Monday    => "Monday",
16             when Tuesday   => "Tuesday",
17             when Wednesday => "Wednesday",
18             when Thursday  => "Thursday",
19             when Friday    => "Friday",
20             when Saturday  => "Saturday",
21             when Sunday    => "Sunday");
```

(continues on next page)

(continued from previous page)

```

22
23 begin
24   Put_Line ("First day is "
25             & Get_Day_Name (Days'First));
26 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Day\_Name\_1  
MD5: 0b7c567c723ded52d8e95c4ef46bcecc

### Runtime output

First day is Monday

(This example is for illustrative purposes only. There is a built-in mechanism, the '`Image`' attribute for scalar types, that returns the name (as a `String`) of any element of an enumeration type. For example `Days'Image(Monday)` is "`"MONDAY"`".)

#### In other languages

Returning variable size objects in languages lacking a garbage collector is a bit complicated implementation-wise, which is why C and C++ don't allow it, preferring to depend on explicit dynamic allocation / free from the user.

The problem is that explicit storage management is unsafe as soon as you want to collect unused memory. Ada's ability to return variable size objects will remove one use case for dynamic allocation, and hence, remove one potential source of bugs from your programs.

Rust follows the C/C++ model, but with safe pointer semantics. However, dynamic allocation is still used. Ada can benefit from a possible performance edge because it can use any model.

## 7.9 Declaring arrays (2)

While we can have array types whose size and bounds are determined at run time, the array's component type needs to be of a definite and constrained type.

Thus, if you need to declare, for example, an array of strings, the `String` subtype used as component will need to have a fixed size.

Listing 98: `show_days.adb`

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Days is
4    type Days is (Monday, Tuesday, Wednesday,
5                  Thursday, Friday,
6                  Saturday, Sunday);
7
8    subtype Day_Name is String (1 .. 2);
9    -- Subtype of string with known size
10
11   type Days_Name_Type is
12     array (Days) of Day_Name;
13     -- ^ Type of the index
14     --          ^ Type of the element.
15     --          Must be definite

```

(continues on next page)

(continued from previous page)

```
16  Names : constant Days_Name_Type :=  
17    ("Mo", "Tu", "We", "Th", "Fr", "Sa", "Su");  
18    -- Initial value given by aggregate  
19  
20 begin  
21  for I in Names'Range loop  
22    Put_Line (Names (I));  
23  end loop;  
24 end Show_Days;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Day\_Name\_2  
MD5: bc66303091c084f66abde72ae59f55a9

### Runtime output

```
Mo  
Tu  
We  
Th  
Fr  
Sa  
Su
```

## 7.10 Array slices

One last feature of Ada arrays that we're going to cover is array slices. It is possible to take and use a slice of an array (a contiguous sequence of elements) as a name or a value.

Listing 99: main.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;  
2  
3  procedure Main is  
4    Buf : String := "Hello ...";  
5  
6    Full_Name : String := "John Smith";  
7  begin  
8    Buf (7 .. 9) := "Bob";  
9    -- Careful! This works because the string  
10   -- on the right side is the same length as  
11   -- the replaced slice!  
12  
13   -- Prints "Hello Bob"  
14   Put_Line (Buf);  
15  
16   -- Prints "Hi John"  
17   Put_Line ("Hi " & Full_Name (1 .. 4));  
18 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Slices  
MD5: cdf582c6c9089658236f5c79b7be4c3f

### Runtime output

```
Hello Bob
Hi John
```

As we can see above, you can use a slice on the left side of an assignment, to replace only part of an array.

A slice of an array is of the same type as the array, but has a different subtype, constrained by the bounds of the slice.

### ⚠️ Attention

Ada has [multidimensional arrays](#)<sup>11</sup>, which are not covered in this course. Slices will only work on one dimensional arrays.

## 7.11 Renaming

So far, we've seen that the following elements can be renamed: [subprograms](#) (page 27), [packages](#) (page 41), and [record components](#) (page 63). We can also rename objects by using the `renames` keyword. This allows for creating alternative names for these objects. Let's look at an example:

Listing 100: measurements.ads

```
1 package Measurements is
2
3     subtype Degree_Celsius is Float;
4
5     Current_Temperature : Degree_Celsius;
6
7 end Measurements;
```

Listing 101: main.adb

```
1 with Ada.Text_IO;  use Ada.Text_IO;
2 with Measurements;
3
4 procedure Main is
5     subtype Degrees is
6         Measurements.Degree_Celsius;
7
8     T : Degrees
9         renames Measurements.Current_Temperature;
10
11 begin
12     T := 5.0;
13
14     Put_Line (Degrees'Image (T));
15     Put_Line (Degrees'Image
16             (Measurements.Current_Temperature));
17
18     T := T + 2.5;
19
20     Put_Line (Degrees'Image (T));
21     Put_Line (Degrees'Image
22             (Measurements.Current_Temperature));
23 end Main;
```

### Code block metadata

<sup>11</sup> <http://www.adu-auth.org/standards/12rm/html/RM-3-6.html>

Project: Courses.Intro\_To\_Ada.Arrays.Variable\_Renaming  
MD5: 4426aeaa364cb5cf10ff40e1bccb9757

### Runtime output

```
5.00000E+00
5.00000E+00
7.50000E+00
7.50000E+00
```

In the example above, we declare a variable T by renaming the Current\_Temperature object from the Measurements package. As you can see by running this example, both Current\_Temperature and its alternative name T have the same values:

- first, they show the value 5.0
- after the addition, they show the value 7.5.

This is because they are essentially referring to the same object, but with two different names.

Note that, in the example above, we're using Degrees as an alias of Degree\_Celsius. We discussed this method *earlier in the course* (page 58).

Renaming can be useful for improving the readability of more complicated array indexing. Instead of explicitly using indices every time we're accessing certain positions of the array, we can create shorter names for these positions by renaming them. Let's look at the following example:

Listing 102: colors.ads

```
1 package Colors is
2
3     type Color is (Black,
4                     Red,
5                     Green,
6                     Blue,
7                     White);
8
9     type Color_Array is
10        array (Positive range <>) of Color;
11
12    procedure Reverse_It (X : in out Color_Array);
13
14 end Colors;
```

Listing 103: colors.adb

```
1 package body Colors is
2
3     procedure Reverse_It (X : in out Color_Array)
4     is
5     begin
6         for I in X'First ..
7             (X'Last + X'First) / 2
8         loop
9             declare
10                 Tmp      : Color;
11                 X_Left   : Color
12                     renames X (I);
13                 X_Right  : Color
14                     renames X (X'Last + X'First - I);
15             begin
```

(continues on next page)

(continued from previous page)

```

16      Tmp      := X_Left;
17      X_Left   := X_Right;
18      X_Right  := Tmp;
19      end;
20      end loop;
21  end Reverse_It;
22
23 end Colors;

```

Listing 104: test\_reverse\_colors.adb

```

1  with Ada.Text_Io; use Ada.Text_Io;
2
3  with Colors; use Colors;
4
5  procedure Test_Reverse_Colors is
6
7      My_Colors : Color_Array (1 .. 5) :=
8          (Black, Red, Green, Blue, White);
9
10 begin
11     for C of My_Colors loop
12         Put_Line ("My_Color: "
13             & Color'Image (C));
14     end loop;
15
16     New_Line;
17     Put_Line ("Reversing My_Color...");
18     New_Line;
19     Reverse_It (My_Colors);
20
21     for C of My_Colors loop
22         Put_Line ("My_Color: "
23             & Color'Image (C));
24     end loop;
25
26 end Test_Reverse_Colors;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Arrays.Reverse\_Colors  
MD5: cd9fd7f64d1ec8967e340d57fd7afc0a

### Runtime output

```

My_Color: BLACK
My_Color: RED
My_Color: GREEN
My_Color: BLUE
My_Color: WHITE

Reversing My_Color...

My_Color: WHITE
My_Color: BLUE
My_Color: GREEN
My_Color: RED
My_Color: BLACK

```

In the example above, package Colors implements the procedure Reverse\_It by declaring new names for two positions of the array. The actual implementation becomes easy to read:

```
begin
  Tmp      := X_Left;
  X_Left   := X_Right;
  X_Right := Tmp;
end;
```

Compare this to the alternative version without renaming:

```
begin
  Tmp          := X (I);
  X (I)        := X (X'Last +
                      X'First - I);
  X (X'Last + X'First - I) := Tmp;
end;
```

## MORE ABOUT TYPES

### 8.1 Aggregates: A primer

So far, we have talked about aggregates quite a bit and have seen a number of examples. Now we will revisit this feature in some more detail.

An Ada aggregate is, in effect, a literal value for a composite type. It's a very powerful notation that helps you to avoid writing procedural code for the initialization of your data structures in many cases.

A basic rule when writing aggregates is that *every component* of the array or record has to be specified, even components that have a default value.

This means that the following code is incorrect:

Listing 105: incorrect.ads

```
1 package Incorrect is
2     type Point is record
3         X, Y : Integer := 0;
4     end record;
5
6     Origin : Point := (X => 0);
7 end Incorrect;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Types.Incorrect\_Aggregate  
MD5: 80a3475dece1c42cfb67b1d57b5bd464

#### Build output

```
incorrect.ads:6:22: error: no value supplied for component "Y"
gprbuild: *** compilation phase failed
```

There are a few shortcuts that you can use to make the notation more convenient:

- To specify the default value for a component, you can use the `<>` notation.
- You can use the `|` symbol to give several components the same value.
- You can use the `others` choice to refer to every component that has not yet been specified, provided all those fields have the same type.
- You can use the range notation `..` to refer to specify a contiguous sequence of indices in an array.

However, note that as soon as you used a named association, all subsequent components likewise need to be specified with named associations.

Listing 106: points.ads

```
1 package Points is
2     type Point is record
3         X, Y : Integer := 0;
4     end record;
5
6     type Point_Array is
7         array (Positive range <>) of Point;
8
9         -- use the default values
10        Origin : Point := (X | Y => <>);
11
12        -- likewise, use the defaults
13        Origin_2 : Point := (others => <>);
14
15        Points_1 : Point_Array := ((1, 2), (3, 4));
16        Points_2 : Point_Array := (1          => (1, 2),
17                                2          => (3, 4),
18                                3 .. 20 => <>);
19    end Points;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Types.Points  
MD5: 48ea183a42f203325ed6190fdbd8493d9

## 8.2 Overloading and qualified expressions

Ada has a general concept of name overloading, which we saw earlier in the section on *enumeration types* (page 47).

Let's take a simple example: it is possible in Ada to have functions that have the same name, but different types for their parameters.

Listing 107: pkg.ads

```
1 package Pkg is
2     function F (A : Integer) return Integer;
3     function F (A : Character) return Integer;
4 end Pkg;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Types.Overloading  
MD5: defae85228ee183b536af395d077e71e

This is a common concept in programming languages, called [overloading](#)<sup>12</sup>, or name overloading.

One of the novel aspects of Ada's overloading facility is the ability to resolve overloading based on the return type of a function.

Listing 108: pkg.ads

```
1 package Pkg is
2     type SSID is new Integer;
```

(continues on next page)

<sup>12</sup> [https://en.wikipedia.org/wiki/Function\\_overloading](https://en.wikipedia.org/wiki/Function_overloading)

(continued from previous page)

```

4  function Convert (Self : SSID)
5      return Integer;
6  function Convert (Self : SSID)
7      return String;
8 end Pkg;

```

Listing 109: main.adb

```

1  with Ada.Text_Io; use Ada.Text_Io;
2  with Pkg;          use Pkg;
3
4  procedure Main is
5      S : String := Convert (123_145_299);
6      --          ^ Valid, will choose the
7      --          proper Convert
8  begin
9      Put_Line (S);
10 end Main;

```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.More\_About\_Types.Overloading  
MD5: aa556b55ee89f9c5f8f7e138d84c27b8

**⚠ Attention**

Note that overload resolution based on the type is allowed for both functions and enumeration literals in Ada - which is why you can have multiple enumeration literals with the same name. Semantically, an enumeration literal is treated like a function that has no parameters.

However, sometimes an ambiguity makes it impossible to resolve which declaration of an overloaded name a given occurrence of the name refers to. This is where a qualified expression becomes useful.

Listing 110: pkg.ads

```

1  package Pkg is
2      type SSID is new Integer;
3
4      function Convert (Self : SSID)
5          return Integer;
6      function Convert (Self : SSID)
7          return String;
8      function Convert (Self : Integer)
9          return String;
10 end Pkg;

```

Listing 111: main.adb

```

1  with Ada.Text_Io; use Ada.Text_Io;
2  with Pkg;          use Pkg;
3
4  procedure Main is
5      S : String := Convert (123_145_299);
6      --          ^ Invalid, which convert
7      --          should we call?
8

```

(continues on next page)

(continued from previous page)

```
9  S2 : String := Convert (SSID'(123_145_299));
10 --                                     ^ We specify that the
11 --                                     type of the
12 --                                     expression is SSID.
13
14 -- We could also have declared a temporary
15
16 I : SSID := 123_145_299;
17
18 S3 : String := Convert (I);
19 begin
20   Put_Line (S);
21 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Types.Overloading\_Error  
MD5: 722660d8b692cde65a1c2b7800dd78c4

Syntactically the target of a qualified expression can be either any expression in parentheses, or an aggregate:

Listing 112: qual\_expr.ads

```
1  package Qual_Expr is
2    type Point is record
3      A, B : Integer;
4    end record;
5
6    P : Point := Point'(12, 15);
7
8    A : Integer := Integer'(12);
9  end Qual_Expr;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Types.Qual\_Expr  
MD5: e71523eb441a28a4f6549d5f0418620a

This illustrates that qualified expressions are a convenient (and sometimes necessary) way for the programmer to make the type of an expression explicit, for the compiler of course, but also for other programmers.

#### ⚠ Attention

While they look and feel similar, type conversions and qualified expressions are *not* the same.

A qualified expression specifies the exact type that the target expression will be resolved to, whereas a type conversion will try to convert the target and issue a run-time error if the target value cannot be so converted.

Note that you can use a qualified expression to convert from one subtype to another, with an exception raised if a constraint is violated.

```
X : Integer := Natural'(1);
```

## 8.3 Character types

As noted earlier, each enumeration type is distinct and incompatible with every other enumeration type. However, what we did not mention previously is that character literals are permitted as enumeration literals. This means that in addition to the language's strongly typed character types, user-defined character types are also permitted:

Listing 113: character\_example.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Character_Example is
4      type My_Char is ('a', 'b', 'c');
5      -- Our custom character type, an
6      -- enumeration type with 3 valid values.
7
8      C : Character;
9      -- ^ Built-in character type
10     -- (it's an enumeration type)
11
12     M : My_Char;
13 begin
14     C := '?';
15     -- ^ Character literal
16     -- (enumeration literal)
17
18     M := 'a';
19
20     C := 65;
21     -- ^ Invalid: 65 is not a
22     -- Character value
23
24     C := Character'Val (65);
25     -- Assign the character at
26     -- position 65 in the
27     -- enumeration (which is 'A')
28
29     M := C;
30     -- ^ Invalid: C is of type Character,
31     -- and M is a My_Char
32
33     M := 'd';
34     -- ^ Invalid: 'd' is not a valid
35     -- literal for type My_Char
36 end Character_Example;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Types.Character\_Example  
MD5: e4c5a07dbe8745749056f8c110d69fa3

### Build output

```

character_example.adb:20:09: error: expected type "Standard.Character"
character_example.adb:20:09: error: found type universal integer
character_example.adb:29:09: error: expected type "My_Char" defined at line 4
character_example.adb:29:09: error: found type "Standard.Character"
character_example.adb:33:09: error: character not defined for type "My_Char" ↴
    ↴ defined at line 4
gprbuild: *** compilation phase failed

```

In this example, we're using characters in the definition of My\_Char.



## ACCESS TYPES (POINTERS)

### 9.1 Overview

Pointers are a potentially dangerous construct, which conflicts with Ada's underlying philosophy.

There are two ways in which Ada helps shield programmers from the dangers of pointers:

1. One approach, which we have already seen, is to provide alternative features so that the programmer does not need to use pointers. Parameter modes, arrays, and varying size types are all constructs that can replace typical pointer usages in C.
2. Second, Ada has made pointers as safe and restricted as possible, but allows "escape hatches" when the programmer explicitly requests them and presumably will be exercising such features with appropriate care.

Here is how you declare a simple pointer type, or access type, in Ada:

Listing 114: dates.ads

```
1 package Dates is
2   type Months is
3     (January, February, March, April,
4      May, June, July, August, September,
5      October, November, December);
6
7   type Date is record
8     Day   : Integer range 1 .. 31;
9     Month : Months;
10    Year  : Integer;
11  end record;
12 end Dates;
```

Listing 115: access\_types.ads

```
1 with Dates; use Dates;
2
3 package Access_Types is
4   -- Declare an access type
5   type Date_Acc is access Date;
6   --           ^ "Designated type"
7   --           ^ Date_Acc values
8   --           point to Date
9   --           objects
10
11  D : Date_Acc := null;
12  --           ^ Literal for
13  --           "access to nothing"
```

(continues on next page)

(continued from previous page)

```
14  -- ^ Access to date
15 end Access_Types;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Access\_Types.Access\_Types  
MD5: d3421918c48c221836bd03b9e68fb5

This illustrates how to:

- Declare an access type whose values point to ("designate") objects from a specific type
- Declare a variable (access value) from this access type
- Give it a value of **null**

In line with Ada's strong typing philosophy, if you declare a second access type whose designated type is Date, the two access types will be incompatible with each other:

Listing 116: access\_types.ads

```
1 with Dates; use Dates;
2
3 package Access_Types is
4   -- Declare an access type
5   type Date_Acc is access Date;
6   type Date_Acc_2 is access Date;
7
8   D : Date_Acc := null;
9   D2 : Date_Acc_2 := D;
10  --          ^ Invalid! Different types
11 end Access_Types;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Access\_Types.Access\_Types  
MD5: af0dff5a26cb16f0fe15c84286557a44

### Build output

```
access_types.ads:9:24: error: expected type "Date_Acc_2" defined at line 6
access_types.ads:9:24: error: found type "Date_Acc" defined at line 5
gprbuild: *** compilation phase failed
```

### In other languages

In most other languages, pointer types are structurally, not nominally typed, like they are in Ada, which means that two pointer types will be the same as long as they share the same target type and accessibility rules.

Not so in Ada, which takes some time getting used to. A seemingly simple problem is, if you want to have a canonical access to a type, where should it be declared? A commonly used pattern is that if you need an access type to a specific type you "own", you will declare it along with the type:

```
package Access_Types is
  type Point is record
    X, Y : Natural;
  end record;

  type Point_Access is access Point;
end Access_Types;
```

## 9.2 Allocation (by type)

Once we have declared an access type, we need a way to give variables of the types a meaningful value! You can allocate a value of an access type with the `new` keyword in Ada.

Listing 117: access\_types.ads

```

1  with Dates; use Dates;
2
3  package Access_Types is
4      type Date_Acc is access Date;
5
6      D : Date_Acc := new Date;
7      --          ^ Allocate a new Date record
8  end Access_Types;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Access\_Types.Access\_Types  
MD5: e0be95b966e4aebaaf25db646d60c35c

If the type you want to allocate needs constraints, you can put them in the subtype indication, just as you would do in a variable declaration:

Listing 118: access\_types.ads

```

1  with Dates; use Dates;
2
3  package Access_Types is
4      type String_Acc is access String;
5      --
6      -- Access to unconstrained array type
7      Msg : String_Acc;
8      --          ^ Default value is null
9
10     Buffer : String_Acc :=
11         new String (1 .. 10);
12         --          ^ Constraint required
13 end Access_Types;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Access\_Types.Access\_Types  
MD5: 83cf7a1074ff1b739658508098aa8208

In some cases, though, allocating just by specifying the type is not ideal, so Ada also allows you to initialize along with the allocation. This is done via the qualified expression syntax:

Listing 119: access\_types.ads

```

1  with Dates; use Dates;
2
3  package Access_Types is
4      type Date_Acc is access Date;
5      type String_Acc is access String;
6
7      D : Date_Acc := new Date'(30, November, 2011);
8

```

(continues on next page)

(continued from previous page)

```
9  Msg : String_Acc := new String'("Hello");
10 end Access_Types;
```

### 9.3 Dereferencing

The last important piece of Ada's access type facility is how to get from an access value to the object that is pointed to, that is, how to dereference the pointer. Dereferencing a pointer uses the `.all` syntax in Ada, but is often not needed — in many cases, the access value will be implicitly dereferenced for you:

Listing 120: access\_types.ads

```
1  with Dates; use Dates;
2
3  package Access_Types is
4    type Date_Acc is access Date;
5
6    D      : Date_Acc := new Date'(30, November, 2011);
7
8    Today : Date := D.all;                                -- Access value dereference
9
10   J      : Integer := D.Day;                            -- Implicit dereference
11   --          for record and array
12   --          components
13   --          Equivalent to D.all.day
14
15
16 end Access_Types;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Access\_Types.Access\_Types  
MD5: 5cd1c259da04010b0dc1b43e9bd93b55

### 9.4 Other features

As you might know if you have used pointers in C or C++, we are still missing features that are considered fundamental to the use of pointers, such as:

- Pointer arithmetic (being able to increment or decrement a pointer in order to point to the next or previous object)
- Manual deallocation - what is called `free` or `delete` in C. This is a potentially unsafe operation. To keep within the realm of safe Ada, you need to never deallocate manually.

Those features exist in Ada, but are only available through specific standard library APIs.

#### ⚠ Attention

The guideline in Ada is that most of the time you can avoid manual allocation, and you should.

There are many ways to avoid manual allocation, some of which have been covered (such as parameter modes). The language also provides library abstractions to avoid pointers:

1. One is the use of *containers* (page 201). Containers help users avoid pointers, because container memory is automatically managed.
2. A container to note in this context is the *Indefinite holder*<sup>13</sup>. This container allows you to store a value of an indefinite type such as String.
3. GNATCOLL has a library for smart pointers, called *RefCount*<sup>14</sup>. Those pointers' memory is automatically managed, so that when an allocated object has no more references to it, the memory is automatically deallocated.

## 9.5 Mutually recursive types

The linked list is a common idiom in data structures; in Ada this would be most naturally defined through two types, a record type and an access type, that are mutually dependent. To declare mutually dependent types, you can use an incomplete type declaration:

Listing 121: simple\_list.ads

```

1 package Simple_List is
2   type Node;
3   -- This is an incomplete type declaration,
4   -- which is completed in the same
5   -- declarative region.
6
7   type Node_Acc is access Node;
8
9   type Node is record
10    Content   : Natural;
11    Prev, Next : Node_Acc;
12  end record;
13 end Simple_List;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Access\_Types.Simple\_List  
MD5: 4929b89c1fc913da635fa02e48248271

In this example, the *Node* and *Node\_Acc* types are mutually dependent.

<sup>13</sup> <http://www.ada-auth.org/standards/12rat/html/Rat12-8-5.html>

<sup>14</sup> <https://github.com/AdaCore/gnatcoll-core/blob/master/src/gnatcoll-refcount.ads>



## MORE ABOUT RECORDS

### 10.1 Dynamically sized record types

We have previously seen *some simple examples of record types* (page 61). Let's now look at some of the more advanced properties of this fundamental language feature.

One point to note is that object size for a record type does not need to be known at compile time. This is illustrated in the example below:

Listing 122: runtime\_length.ads

```
1 package Runtime_Length is
2     function Compute_Max_Len return Natural;
3 end Runtime_Length;
```

Listing 123: var\_size\_record.ads

```
1 with Runtime_Length; use Runtime_Length;
2
3 package Var_Size_Record is
4     Max_Len : constant Natural := Compute_Max_Len;
5     --          ^ Not known at compile time
6
7     type Items_Array is
8         array (Positive range <>) of Integer;
9
10    type Growable_Stack is record
11        Items : Items_Array (1 .. Max_Len);
12        Len   : Natural;
13    end record;
14    -- Growable_Stack is a definite type, but
15    -- size is not known at compile time.
16
17    G : Growable_Stack;
18 end Var_Size_Record;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Records.Var\_Size\_Record  
MD5: 6fb0b3f2b685a72ec694640ce378f77c

It is completely fine to determine the size of your records at run time, but note that all objects of this type will have the same size.

## 10.2 Records with discriminant

In the example above, the size of the `Items` field is determined once, at run-time, but every `Growable_Stack` instance will be exactly the same size. But maybe that's not what you want to do. We saw that arrays in general offer this flexibility: for an unconstrained array type, different objects can have different sizes.

You can get analogous functionality for records, too, using a special kind of field that is called a discriminant:

Listing 124: var\_size\_record\_2.ads

```

1  package Var_Size_Record_2 is
2    type Items_Array is
3      array (Positive range <>) of Integer;
4
5    type Growable_Stack (Max_Len : Natural) is
6      record
7        -- ^ Discriminant. Cannot be
8        -- modified once
9        -- initialized.
10       Items : Items_Array (1 .. Max_Len);
11       Len   : Natural := 0;
12     end record;
13     -- Growable_Stack is an indefinite type
14     -- (like an array)
15 end Var_Size_Record_2;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Records.Var\_Size\_Record\_2  
MD5: 0c2ffe41b7553984e1ef48a50386559f

Discriminants, in their simple forms, are constant: You cannot modify them once you have initialized the object. This intuitively makes sense since they determine the size of the object.

Also, they make a type indefinite: Whether or not the discriminant is used to specify the size of an object, a type with a discriminant will be indefinite if the discriminant is not declared with an initialization:

Listing 125: test\_discriminants.ads

```

1  package Test_Discriminants is
2    type Point (X, Y : Natural) is record
3      null;
4    end record;
5
6    P : Point;
7    -- ERROR: Point is indefinite, so you
8    -- need to specify the discriminants
9    -- or give a default value
10
11   P2 : Point (1, 2);
12   P3 : Point := (1, 2);
13   -- Those two declarations are equivalent.
14
15 end Test_Discriminants;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Records.Test\_Discriminants  
MD5: c3ec81ccae0d4144fe952ad99482be81

## Build output

```
test_discriminants.ads:6:08: error: unconstrained subtype not allowed (need initialization)
test_discriminants.ads:6:08: error: provide initial value or explicit discriminant values
test_discriminants.ads:6:08: error: or give default discriminant values for type "Point"
gprbuild: *** compilation phase failed
```

This also means that, in the example above, you cannot declare an array of Point values, because the size of a Point is not known.

As mentioned in the example above, we could provide a default value for the discriminants, so that we could legally declare Point values without specifying the discriminants. For the example above, this is how it would look:

Listing 126: test\_discriminants.ads

```
1 package Test_Discriminants is
2   type Point (X, Y : Natural := 0) is record
3     null;
4   end record;
5
6   P : Point;
7   -- We can now simply declare a "Point"
8   -- without further ado. In this case,
9   -- we're using the default values (0)
10  -- for X and Y.
11
12  P2 : Point (1, 2);
13  P3 : Point := (1, 2);
14  -- We can still specify discriminants.
15
16 end Test_Discriminants;
```

## Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Records.Test\_Discriminants  
MD5: 259f6cdf7fa857cc006dac6d1daedd73

Also note that, even though the Point type now has default discriminants, we can still specify discriminants, as we're doing in the declarations of P2 and P3.

In most other respects discriminants behave like regular fields: You have to specify their values in aggregates, as seen above, and you can access their values via the dot notation.

Listing 127: main.adb

```
1 with Ada.Text_IO;      use Ada.Text_IO;
2
3 with Var_Size_Record_2; use Var_Size_Record_2;
4
5 procedure Main is
6   procedure Print_Stack (G : Growable_Stack) is
7   begin
8     Put("<Stack, items: [");
9     for I in G.Items'Range loop
10      exit when I > G.Len;
```

(continues on next page)

(continued from previous page)

```
11      Put (" " & Integer'Image (G.Items (I)));
12  end loop;
13  Put_Line ("]>");
14 end Print_Stack;
15
16 S : Growable_Stack :=
17   (Max_Len => 128,
18    Items   => (1, 2, 3, 4, others => <>),
19    Len      => 4);
20 begin
21   Print_Stack (S);
22 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Records.Var\_Size\_Record\_2  
MD5: 4e8c102cd93dc5d8aa1b402589c5239b

### Runtime output

```
<Stack, items: [ 1 2 3 4]>
```

#### Note

In the examples above, we used a discriminant to determine the size of an array, but it is not limited to that, and could be used, for example, to determine the size of a nested discriminated record.

## 10.3 Variant records

The examples of discriminants thus far have illustrated the declaration of records of varying size, by having components whose size depends on the discriminant.

However, discriminants can also be used to obtain the functionality of what are sometimes called "variant records": records that can contain different sets of fields.

Listing 128: variant\_record.ads

```
1 package Variant_Record is
2   -- Forward declaration of Expr
3   type Expr;
4
5   -- Access to a Expr
6   type Expr_Access is access Expr;
7
8   type Expr_Kind_Type is (Bin_Op_Plus,
9                           Bin_Op_Minus,
10                          Num);
11  -- A regular enumeration type
12
13  type Expr (Kind : Expr_Kind_Type) is record
14    -- ^ The discriminant is an
15    -- enumeration value
16    case Kind is
17      when Bin_Op_Plus | Bin_Op_Minus =>
18        Left, Right : Expr_Access;
19      when Num =>
```

(continues on next page)

(continued from previous page)

```

20      Val : Integer;
21  end case;
22  -- Variant part. Only one, at the end of
23  -- the record definition, but can be
24  -- nested
25  end record;
26 end Variant_Record;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Records.Variant\_Record  
MD5: af9c1edca3ed6b2d938249c7258806b1

The fields that are in a **when** branch will be only available when the value of the discriminant is covered by the branch. In the example above, you will only be able to access the fields **Left** and **Right** when the **Kind** is **Bin\_Op\_Plus** or **Bin\_Op\_Minus**.

If you try to access a field that is not valid for your record, a **Constraint\_Error** will be raised.

Listing 129: main.adb

```

1 with Variant_Record; use Variant_Record;
2
3 procedure Main is
4   E : Expr := (Num, 12);
5 begin
6   E.Left := new Expr'(Num, 15);
7   -- Will compile but fail at runtime
8 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Records.Variant\_Record  
MD5: d157d5f96db0825b9376ba7fca9613ed

### Build output

```

main.adb:6:05: warning: component not present in subtype of "Expr" defined at line ↴
  ↴4 [enabled by default]
main.adb:6:05: warning: Constraint_Error will be raised at run time [enabled by ↴
  ↴default]

```

### Runtime output

```
raised CONSTRAINT_ERROR : main.adb:6 discriminant check failed
```

Here is how you could write an evaluator for expressions:

Listing 130: main.adb

```

1 with Ada.Text_IO;      use Ada.Text_IO;
2
3 with Variant_Record; use Variant_Record;
4
5 procedure Main is
6   function Eval_Expr (E : Expr) return Integer is
7     (case E.Kind is
8      when Bin_Op_Plus  =>
9        Eval_Expr (E.Left.all)

```

(continues on next page)

(continued from previous page)

```
10      + Eval_Expr (E.Right.all),
11  when Bin_Op_Minus =>
12      Eval_Expr (E.Left.all)
13      - Eval_Expr (E.Right.all),
14  when Num => E.Val);
15
16  E : Expr := (Bin_Op_Plus,
17      new Expr'(Bin_Op_Minus,
18          new Expr'(Num, 12),
19          new Expr'(Num, 15)),
20      new Expr'(Num, 3));
21 begin
22     Put_Line (Integer'Image (Eval_Expr (E)));
23 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.More\_About\_Records.Variant\_Record  
MD5: 807dbb921b44b3eaeaf1baf6ffelafaa

### Runtime output

0

#### In other languages

Ada's variant records are very similar to Sum types in functional languages such as OCaml or Haskell. A major difference is that the discriminant is a separate field in Ada, whereas the 'tag' of a Sum type is kind of built in, and only accessible with pattern matching.

There are other differences (you can have several discriminants in a variant record in Ada). Nevertheless, they allow the same kind of type modeling as sum types in functional languages.

Compared to C/C++ unions, Ada variant records are more powerful in what they allow, and are also checked at run time, which makes them safer.

## FIXED-POINT TYPES

In this chapter, we discuss fixed-point types, which can be classified in two categories: *decimal fixed-point types* (page 103) and *ordinary (binary) fixed-point types* (page 107). Afterward a brief overview of each category, we discuss some *differences between fixed-point and floating-point types* (page 108).

### 11.1 Decimal fixed-point types

We have already seen how to specify *floating-point types* (page 48). However, in some applications floating-point is not appropriate since, for example, the roundoff error from binary arithmetic may be unacceptable or perhaps the hardware does not support floating-point instructions. Ada provides a category of types, the decimal fixed-point types, that allows the programmer to specify the required decimal precision (number of digits) as well as the scaling factor (a power of ten) and, optionally, a range. In effect the values will be represented as integers implicitly scaled by the specified power of 10. This is useful, for example, for financial applications.

The syntax for a simple decimal fixed-point type is

```
type <type-name> is
  delta <delta-value> digits <digits-value>;
```

In this case, the **delta** and the **digits** will be used by the compiler to derive a range.

#### 11.1.1 Decimal delta

The delta determines the required decimal precision for the type. For example, if we want to be able to use two digits after the decimal point, we would write **delta 10.0 \*\* (-2)** — which is equivalent to **delta 0.01**. (You can use any of those definitions: both **delta 10.0 \*\* (-2)** and **delta 0.01** are correct.)

Listing 131: decimal\_fixed\_point\_types.adb

```
1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Decimal_Fixed_Point_Types is
4    type Decimal is
5      delta 10.0 ** (-1) digits 3;
6
7    -- Alternatively:
8    -- type Decimal is
9    --   delta 0.1 digits 3;
10 begin
11   Put_Line
12     ("The decimal precision of Decimal is "
13      & Decimal'Delta'Image);
14 end Decimal_Fixed_Point_Types;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Fixed\_Point\_Types.Decimal\_Fixed\_Point\_Types  
MD5: bcb69d188151413ee4f4a413d85bb0de

### Runtime output

The decimal precision of Decimal is 1.0000000000000000E-01

In this example, we declare the `Decimal` type, which has a decimal precision of 0.1. We use the '`Delta`' attribute to show the decimal precision of the type.

## 11.1.2 Decimal digits

Unsurprisingly, the `digits` part of the type declaration determines the number of digits that a type is able to represent. For example, by writing `digits 3`, we're able to represent values with three digits ranging from -999 to 999. For example:

Listing 132: decimal\_fixed\_point\_types.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Decimal_Fixed_Point_Types is
4     type Decimal is
5         delta 10.0 ** (0) digits 3;
6 begin
7     Put_Line ("The minimum value of Decimal is "
8             & Decimal'First'Image);
9     Put_Line ("The maximum value of Decimal is "
10            & Decimal'Last'Image);
11 end Decimal_Fixed_Point_Types;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Fixed\_Point\_Types.Decimal\_Fixed\_Point\_Types  
MD5: 8b130008590ca44d7ff588aef1ad3270

### Runtime output

The minimum value of Decimal is -999.0  
The maximum value of Decimal is 999.0

In this example, we declare the `Decimal` type, which has a range from -999 to 999. We use the '`First`' and '`Last`' attributes to show the first and last value of the range, respectively.

### For further reading...

When running the application above, we see that the first and last values are -999.0 and 999.0, respectively — i.e. values with the decimal point. Strictly speaking, however, the actual first and last values are -999 and 999 because we selected a delta of 1.0. The decimal point (.0) we see in the application output (in the values -999.0 and 999.0) is only there to indicate that this is not an integer value — however, it doesn't indicate an extended decimal precision at all.

Listing 133: decimal\_fixed\_point\_types.adb

```

1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Decimal_Fixed_Point_Types is
4    type Decimal is
5      delta 10.0 ** (0) digits 3;
6
7    D : Decimal := 0.1;
8      ^^^
9    --   --  ERROR: value cannot be represented
10   --      by Decimal type.
11 begin
12   Put_Line (D'Image);
13 end Decimal_Fixed_Point_Types;

```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.Fixed\_Point\_Types.Decimal\_Fixed\_Point\_Type\_  
 ↳Precision\_Error  
 MD5: d169d7796b61392baada9795ee8b8187

**Build output**

decimal\_fixed\_point\_types.adb:7:19: error: value has extraneous low order digits  
 gprbuild: \*\*\* compilation phase failed

Assigning the value 0.1 to D is wrong because the Decimal type cannot represent this value.

### For further reading...

The `Decimal` type above is similar to — but far from being equivalent to — the following floating-point type declaration:

Listing 134: `decimal_fixed_point_types.adb`

```
1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Decimal_Fixed_Point_Types is
4      type Float_999 is
5          digits 3
6          range -999.0 .. 999.0;
7  begin
8      Put_Line ("The minimum value of Float_999 is "
9                 & Float_999'First'Image);
10     Put_Line ("The maximum value of Float_999 is "
11                 & Float_999'Last'Image);
12 end Decimal_Fixed_Point_Types;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Fixed\_Point\_Types.Floating\_Point\_Range\_Equivalent  
MD5: 49a6316e1c6846da75d5d9862b1f7ef8

### Runtime output

```
The minimum value of Float_999 is -9.99000E+02
The maximum value of Float_999 is 9.99000E+02
```

The `Float_999` type from this example has (roughly) the same range as the `Decimal` type that we declared in the previous example: -999 to 999. However, there are substantial *differences between fixed-point and floating-point types* (page 108), so we cannot say that these type declarations are equivalent.

### 11.1.3 Decimal delta and digits

By combining those three digits (i.e. `digits 3`) with a decimal precision of two digits after the decimal point (`delta 10.0 ** (-2)`), we get a range from -9.99 to 9.99. For example:

Listing 135: `decimal_fixed_point_types.adb`

```
1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Decimal_Fixed_Point_Types is
4      type Decimal is
5          delta 10.0 ** (-2) digits 3;
6  begin
7      Put_Line ("The minimum value of Decimal is "
8                 & Decimal'First'Image);
9      Put_Line ("The maximum value of Decimal is "
10                 & Decimal'Last'Image);
11 end Decimal_Fixed_Point_Types;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Fixed\_Point\_Types.Decimal\_Fixed\_Point\_Types  
MD5: 2f55fc3d0f7f77c5f17f750680933422

### Runtime output

```
The minimum value of Decimal is -9.99
The maximum value of Decimal is 9.99
```

In this example, we declare the `Decimal` type, which has a range from -9.99 to 9.99 (as expected).

### 11.1.4 Requirements for the `delta`

Note that the `delta` expression for decimal fixed-point types must be a power of 10. Using a different value for the power leads to compilation errors. For example:

Listing 136: `decimal_fixed_point_type_error.ads`

```
1 package Decimal_Fixed_Point_Type_Error is
2
3     type Decimal_Error_1 is
4         delta 2.0 ** (-1) digits 3;
5         ^^^^^^^^^^
6
7         -- ERROR: not power of ten
8
9     type Decimal_Error_2 is
10        delta 0.125 digits 3;
11        ^^^
12
13        -- ERROR: not power of ten
14
15 end Decimal_Fixed_Point_Type_Error;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Fixed\_Point\_Types.Decimal\_Fixed\_Point\_Type\_Error  
MD5: f4541eb0256b3770c1b8a9fd4a6c02fe

#### Build output

```
decimal_fixed_point_type_error.ads:4:06: error: delta expression must be a power
  ↵ of 10
decimal_fixed_point_type_error.ads:9:06: error: delta expression must be a power
  ↵ of 10
gprbuild: *** compilation phase failed
```

In this example, the type declarations (of `Decimal_Error_1` and `Decimal_Error_2`) are wrong because the `delta` expression is not a power of 10.

## 11.2 Ordinary fixed-point types

Ordinary fixed-point types are similar to decimal fixed-point types in that the values are, in effect, scaled integers. The difference between them is in the scale factor: for a decimal fixed-point type, the scaling, given explicitly by the type's `delta`, is always a power of ten. In contrast, for an ordinary fixed-point type, the `delta` isn't limited to power of 10 values, but it can have any arbitrary base.

#### For further reading...

When representing ordinary fixed-point types on the machine, the compiler selects a scaling factor derived from the value of `delta` specified in the type declaration. This compiler-selected scaling factor is, by default, a power of two — even if the value provided for the `delta` isn't a power of two. Therefore, ordinary fixed-point types are sometimes called binary fixed-point types.

The syntax for an ordinary fixed-point type is

```
type <type-name> is
  delta <delta-value>
  range <lower-bound> .. <upper-bound>;
```

For example, we can define an ordinary fixed-point type `T_Inv_Trig` for inverse trigonometric calculations:

Listing 137: `custom_fixed_point_range.adb`

```
1  with Ada.Text_IO;  use Ada.Text_IO;
2  with Ada.Numerics; use Ada.Numerics;
3
4  procedure Custom_Fixed_Point_Range is
5    type T_Inv_Trig is
6      delta 0.0005
7      range -Pi / 2.0 .. Pi / 2.0;
8  begin
9    Put_Line ("Delta      value of T_Inv_Trig: "
10              & T_Inv_Trig'Image
11                  (T_Inv_Trig'Delta));
12   Put_Line ("Minimum  value of T_Inv_Trig: "
13              & T_Inv_Trig'Image
14                  (T_Inv_Trig'First));
15   Put_Line ("Maximum  value of T_Inv_Trig: "
16              & T_Inv_Trig'Image
17                  (T_Inv_Trig'Last));
18 end Custom_Fixed_Point_Range;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Fixed\_Point\_Types.Custom\_Fixed\_Point\_Range  
MD5: 41b251e49ab887496ecd491431b7ffb0

### Runtime output

```
Delta      value of T_Inv_Trig:  0.0005
Minimum  value of T_Inv_Trig: -1.5708
Maximum  value of T_Inv_Trig:  1.5708
```

In this example, we are defining the `T_Inv_Trig` type with a range from  $-\pi/2$  to  $\pi/2$ , and a delta of 0.0005. Note that, in this case, the delta is neither a power of ten nor a power of two. (In fact, this value corresponds to `2000.0 ** (-1)`.)

## 11.3 Fixed-point vs. floating-point types

The main difference between fixed-point and floating-point types is that fixed-point types don't have an exponent. This has an impact on calculations using small values: while they might still be representable with floating-point types, those small values might simply *disappear* (i.e. become zero) in the fixed-point representation. Let's see an example where we compare the decimal type `Decimal` to the floating-point type `Float_32`:

Listing 138: `decimal_vs_floating_point_types.adb`

```
1  with Ada.Text_IO;  use Ada.Text_IO;
2
3  procedure Decimal_Vs_Floating_Point_Types is
4    type Decimal is
5      delta 10.0 ** (-2) digits 9;
```

(continues on next page)

(continued from previous page)

```

6      type Float_32 is
7        digits 6
8        range -9999999.99 .. 9999999.99;
9
10     D : Decimal := 0.01;
11     F : Float_32 := 0.01;
12
13 begin
14   Put_Line ("D = " &
15             D'Image);
16   Put_Line ("F = " &
17             F'Image);
18
19   D := D / 2.0;
20   -- ~~~~~
21   -- Value becomes zero.
22
23   F := F / 2.0;
24   -- ~~~~~
25   -- Exponent is used to
26   -- represent smaller
27   -- value.
28
29   Put_Line ("D = " &
30             D'Image);
31   Put_Line ("F = " &
32             F'Image);
33 end Decimal_Vs_Floating_Point_Types;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Fixed\_Point\_Types.Decimal\_Vs\_Floating\_Point\_Types  
MD5: dbbe2642fb64ae54c7634274e8ff6c79

### Runtime output

```

D = 0.01
F = 1.00000E-02
D = 0.00
F = 5.00000E-03

```

Both types in this example have roughly the same size and range. However, the result of the divide-by-two operation isn't the same: because of the exponent, F has the expected value (0.005) after the operation, while the value of D is zero. The reason is that the resulting value 0.005 cannot be represented by the decimal precision of the Decimal type. In the case of F, however, the value can be represented due to a simple change in the exponent.

This lack of precision we just described might seem like a drawback for fixed-point types. However, depending on the algorithm and its field of application, this is the exact behavior that we might be looking for. As mentioned in the beginning of this chapter, financial applications benefit from decimal types, while using floating-point type for these applications can lead to unpredictable (or undesirable) behavior.

Another major difference concerns the way fixed-point operations translate into machine operations. In most cases, operations on fixed-point types are modeled in a processor by using integer registers and instructions. Essentially, the compiler maps fixed-point types to integer types, but it uses slightly different numeric rules. This fact can be an advantage for specific embedded applications where a floating-point unit might be either non-existent or its usage might have a higher associated cost in terms of CPU cycles or power consumption. Therefore, for these specific applications, using fixed-point types could be considered as an alternative.



## PRIVACY

One of the main principles of modular programming, as well as object oriented programming, is [encapsulation<sup>15</sup>](#).

Encapsulation, briefly, is the concept that the implementer of a piece of software will distinguish between the code's public interface and its private implementation.

This is not only applicable to software libraries but wherever abstraction is used.

In Ada, the granularity of encapsulation is a bit different from most object-oriented languages, because privacy is generally specified at the package level.

### 12.1 Basic encapsulation

Listing 139: encapsulate.ads

```
1 package Encapsulate is
2   procedure Hello;
3
4   private
5
6   procedure Hello2;
7   -- Not visible from external units
8 end Encapsulate;
```

Listing 140: encapsulate.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 package body Encapsulate is
4
5   procedure Hello is
6   begin
7     Put_Line ("Hello");
8   end Hello;
9
10  procedure Hello2 is
11  begin
12    Put_Line ("Hello #2");
13  end Hello2;
14
15 end Encapsulate;
```

---

<sup>15</sup> [https://en.wikipedia.org/wiki/Encapsulation\\_\(computer\\_programming\)](https://en.wikipedia.org/wiki/Encapsulation_(computer_programming))

Listing 141: main.adb

```
1 with Encapsulate;
2
3 procedure Main is
4 begin
5     Encapsulate.Hello;
6     Encapsulate.Hello2;
7     -- Invalid: Hello2 is not visible
8 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Privacy.Encapsulate  
MD5: cf56ee89481962d1e0a6d1e9ad888362

### Build output

```
main.adb:6:15: error: "Hello2" is not a visible entity of "Encapsulate"
gprbuild: *** compilation phase failed
```

## 12.2 Abstract data types

With this high-level granularity, it might not seem obvious how to hide the implementation details of a type. Here is how it can be done in Ada:

Listing 142: stacks.ads

```
1 package Stacks is
2     type Stack is private;
3     -- Declare a private type: You cannot depend
4     -- on its implementation. You can only assign
5     -- and test for equality.
6
7     procedure Push (S    : in out Stack;
8                      Val :        Integer);
9     procedure Pop  (S    : in out Stack;
10                      Val :        out Integer);
11
12     private
13
14         subtype Stack_Index is
15             Natural range 1 .. 10;
16
17         type Content_Type is
18             array (Stack_Index) of Natural;
19
20         type Stack is record
21             Top      : Stack_Index;
22             Content : Content_Type;
23         end record;
24     end Stacks;
```

Listing 143: stacks.adb

```
1 package body Stacks is
2
3     procedure Push (S    : in out Stack;
4                      Val :        Integer) is
5 begin
```

(continues on next page)

(continued from previous page)

```

6      -- Missing implementation!
7      null;
8  end Push;

9
10 procedure Pop (S  : in out Stack;
11                  Val :      out Integer) is
12 begin
13      -- Dummy implementation!
14      Val := 0;
15  end Pop;
16
17 end Stacks;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Privacy.Stacks  
MD5: 364df7c6806af4a1bc957c2c2d53b2cc

In the above example, we define a stack type in the public part (known as the *visible part* of the package spec in Ada), but the exact representation of that type is private.

Then, in the private part, we define the representation of that type. We can also declare other types that will be used as *helpers* for our main public type. This is useful since declaring helper types is common in Ada.

A few words about terminology:

- The Stack type as viewed from the public part is called the *partial view* of the type. This is what clients have access to.
- The Stack type as viewed from the private part or the body of the package is called the *full view* of the type. This is what implementers have access to.

From the point of view of the client (the *with'ing* unit), only the public (visible) part is important, and the private part could as well not exist. It makes it very easy to read linearly the part of the package that is important for you.

```

-- No need to read the private part to use the package
package Stacks is
  type Stack is private;

  procedure Push (S  : in out Stack;
                 Val :      Integer);
  procedure Pop (S  : in out Stack;
                 Val :      out Integer);
private
  ...
end Stacks;

```

Here is how the Stacks package would be used:

```

-- Example of use
with Stacks; use Stacks;

procedure Test_Stack is
  S : Stack;
  Res : Integer;
begin
  Push (S, 5);
  Push (S, 7);
  Pop (S, Res);
end Test_Stack;

```

## 12.3 Limited types

Ada's *limited* type facility allows you to declare a type for which assignment and comparison operations are not automatically provided.

Listing 144: stacks.ads

```

1  package Stacks is
2    type Stack is limited private;
3    -- Limited type. Cannot assign nor compare.
4
5    procedure Push (S  : in out Stack;
6                    Val :           Integer);
7    procedure Pop  (S  : in out Stack;
8                    Val :           out Integer);
9
10   private
11     subtype Stack_Index is
12       Natural range 1 .. 10;
13
14     type Content_Type is
15       array (Stack_Index) of Natural;
16
17     type Stack is limited record
18       Top      : Stack_Index;
19       Content : Content_Type;
20     end record;
21   end Stacks;

```

Listing 145: stacks.adb

```

1  package body Stacks is
2
3    procedure Push (S  : in out Stack;
4                    Val :           Integer) is
5    begin
6      -- Missing implementation!
7      null;
8    end Push;
9
10   procedure Pop  (S  : in out Stack;
11                  Val :           out Integer) is
12   begin
13     -- Dummy implementation!
14     Val := 0;
15   end Pop;
16
17 end Stacks;

```

Listing 146: main.adb

```

1  with Stacks; use Stacks;
2
3  procedure Main is
4      S, S2 : Stack;
5  begin
6      S := S2;
7      -- Illegal: S is limited.
8  end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Privacy.Limited\_Stacks  
MD5: 811343b46f20ac6af5e1bf26561f8d8d

### Build output

```
main.adb:6:04: error: left hand of assignment must not be limited type
gprbuild: *** compilation phase failed
```

This is useful because, for example, for some data types the built-in assignment operation might be incorrect (for example when a deep copy is required).

Ada does allow you to overload the comparison operators = and /= for limited types (and to override the built-in declarations for non-limited types).

Ada also allows you to implement special semantics for assignment via [controlled types](#)<sup>16</sup>. However, in some cases assignment is simply inappropriate; one example is the [File\\_Type](#) from the Ada.Text\_Io package, which is declared as a limited type and thus attempts to assign one file to another would be detected as illegal.

## 12.4 Child packages & privacy

We've seen previously (in the [child packages section](#) (page 35)) that packages can have child packages. Privacy plays an important role in child packages. This section discusses some of the privacy rules that apply to child packages.

Although the private part of a package P is meant to encapsulate information, certain parts of a child package P.C can have access to this private part of P. In those cases, information from the private part of P can then be used as if it were declared in the public part of its specification. To be more specific, the body of P.C and the private part of the specification of P.C have access to the private part of P. However, the public part of the specification of P.C only has access to the public part of P's specification. The following table summarizes this:

Part of a child package	Access to the private part of its parent's specification
Specification: public part	
Specification: private part	✓
Body	✓

The rest of this section shows examples of how this access to private information actually works for child packages.

Let's first look at an example where the body of a child package P.C has access to the private part of the specification of its parent P. We've seen, in a previous source-code example,

<sup>16</sup> <http://www.adu-auth.org/standards/12rm/html/RM-7-6.html>

that the Hello2 procedure declared in the private part of the Encapsulate package cannot be used in the Main procedure, since it's not visible there. This limitation doesn't apply, however, for parts of the child packages of the Encapsulate package. In fact, the body of its child package Encapsulate.Child has access to the Hello2 procedure and can call it there, as you can see in the implementation of the Hello3 procedure of the Child package:

Listing 147: encapsulate.ads

```
1 package Encapsulate is
2     procedure Hello;
3
4     private
5
6         procedure Hello2;
7         -- Not visible from external units
8         -- But visible in child packages
9     end Encapsulate;
```

Listing 148: encapsulate.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 package body Encapsulate is
4
5     procedure Hello is
6     begin
7         Put_Line ("Hello");
8     end Hello;
9
10    procedure Hello2 is
11    begin
12        Put_Line ("Hello #2");
13    end Hello2;
14
15 end Encapsulate;
```

Listing 149: encapsulate-child.ads

```
1 package Encapsulate.Child is
2
3     procedure Hello3;
4
5 end Encapsulate.Child;
```

Listing 150: encapsulate-child.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 package body Encapsulate.Child is
4
5     procedure Hello3 is
6     begin
7         -- Using private procedure Hello2
8         -- from the parent package
9         Hello2;
10        Put_Line ("Hello #3");
11    end Hello3;
12
13 end Encapsulate.Child;
```

Listing 151: main.adb

```

1  with Encapsulate.Child;
2
3  procedure Main is
4  begin
5      Encapsulate.Child.Hello3;
6  end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Privacy.Encapsulate\_Child  
MD5: 1533f43eee8f8b4d14c9b2101f42f13a

### Runtime output

```
Hello #2
Hello #3
```

The same mechanism applies to types declared in the private part of a parent package. For instance, the body of a child package can access components of a record declared in the private part of its parent package. Let's look at an example:

Listing 152: my\_types.ads

```

1  package My_Types is
2
3      type Priv_Rec is private;
4
5  private
6
7      type Priv_Rec is record
8          Number : Integer := 42;
9      end record;
10
11 end My_Types;

```

Listing 153: my\_types-ops.ads

```

1  package My_Types.Ops is
2
3      procedure Display (E : Priv_Rec);
4
5  end My_Types.Ops;

```

Listing 154: my\_types-ops.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  package body My_Types.Ops is
4
5      procedure Display (E : Priv_Rec) is
6      begin
7          Put_Line ("Priv_Rec.Number: "
8                  & Integer'Image (E.Number));
9      end Display;
10
11 end My_Types.Ops;

```

Listing 155: main.adb

```
1  with Ada.Text_IO;  use Ada.Text_IO;
2
3  with My_Types;      use My_Types;
4  with My_Types.Ops;  use My_Types.Ops;
5
6  procedure Main is
7    E : Priv_Rec;
8  begin
9    Put_Line ("Presenting information:");
10
11   -- The following code would trigger a
12   -- compilation error here:
13
14   -- Put_Line ("Priv_Rec.Number: "
15   --            & Integer'Image (E.Number));
16
17   Display (E);
18 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Privacy.Private\_Type\_Child  
MD5: 9960611460bc1190b30949eca08fc02b

### Runtime output

```
Presenting information:
Priv_Rec.Number: 42
```

In this example, we don't have access to the `Number` component of the record type `Priv_Rec` in the `Main` procedure. You can see this in the call to `Put_Line` that has been commented-out in the implementation of `Main`. Trying to access the `Number` component there would trigger a compilation error. But we do have access to this component in the body of the `My_Types.Ops` package, since it's a child package of the `My_Types` package. Therefore, `Ops`'s body has access to the declaration of the `Priv_Rec` type — which is in the private part of its parent, the `My_Types` package. For this reason, the same call to `Put_Line` that would trigger a compilation error in the `Main` procedure works fine in the `Display` procedure of the `My_Types.Ops` package.

This kind of privacy rules for child packages allows for extending the functionality of a parent package and, at the same time, retain its encapsulation.

As we mentioned previously, in addition to the package body, the private part of the specification of a child package `P.C` also has access to the private part of the specification of its parent `P`. Let's look at an example where we declare an object of private type `Priv_Rec` in the private part of the child package `My_Types.Child` and initialize the `Number` component of the `Priv_Rec` record directly:

```
package My_Types.Child is

private

  E : Priv_Rec := (Number => 99);

end My_Types.Ops;
```

As expected, we wouldn't be able to initialize this component if we moved this declaration to the public (visible) part of the same child package:

```
package My_Types.Child is
  E : Priv_Rec := (Number => 99);
end My_Types.Ops;
```

The declaration above triggers a compilation error, since type `Priv_Rec` is private. Because the public part of `My_Types.Child` is also visible outside the child package, Ada cannot allow accessing private information in this part of the specification.



## GENERICs

### 13.1 Introduction

Generics are used for metaprogramming in Ada. They are useful for abstract algorithms that share common properties with each other.

Either a subprogram or a package can be generic. A generic is declared by using the keyword **generic**. For example:

Listing 156: operator.ads

```
1 generic
2   type T is private;
3   -- Declaration of formal types and objects
4   -- Below, we could use one of the following:
5   -- <procedure | function | package>
6   procedure Operator (Dummy : in out T);
```

Listing 157: operator.adb

```
1 procedure Operator (Dummy : in out T) is
2 begin
3   null;
4 end Operator;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Show\_Simple\_Generic  
MD5: 1321d437043dafdb725fad416e654318

### 13.2 Formal type declaration

Formal types are abstractions of a specific type. For example, we may want to create an algorithm that works on any integer type, or even on any type at all, whether a numeric type or not. The following example declares a formal type T for the Set procedure.

Listing 158: set.ads

```
1 generic
2   type T is private;
3   -- T is a formal type that indicates that
4   -- any type can be used, possibly a numeric
5   -- type or possibly even a record type.
6   procedure Set (Dummy : T);
```

Listing 159: set.adb

```
1 procedure Set (Dummy : T) is
2 begin
3     null;
4 end Set;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Show\_Formal\_Type\_Declaration  
MD5: 668156f66b2479c4932d18b5ad35deba

The declaration of T as **private** indicates that you can map any definite type to it. But you can also restrict the declaration to allow only some types to be mapped to that formal type. Here are some examples:

Formal Type	Format
Any type	<b>type T is private;</b>
Any discrete type	<b>type T is (&lt;&gt;);</b>
Any floating-point type	<b>type T is digits &lt;&gt;;</b>

## 13.3 Formal object declaration

Formal objects are similar to subprogram parameters. They can reference formal types declared in the formal specification. For example:

Listing 160: set.ads

```
1 generic
2     type T is private;
3     X : in out T;
4     -- X can be used in the Set procedure
5 procedure Set (E : T);
```

Listing 161: set.adb

```
1 procedure Set (E : T) is
2     pragma Unreferenced (E, X);
3 begin
4     null;
5 end Set;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Show\_Formal\_Object\_Declaration  
MD5: 1b88bc0e5b8f48a35394966e6af07ac0

Formal objects can be either input parameters or specified using the **in out** mode.

## 13.4 Generic body definition

We don't repeat the **generic** keyword for the body declaration of a generic subprogram or package. Instead, we start with the actual declaration and use the generic types and objects we declared. For example:

Listing 162: set.ads

```

1 generic
2   type T is private;
3   X : in out T;
4 procedure Set (E : T);

```

Listing 163: set.adb

```

1 procedure Set (E : T) is
2   -- Body definition: "generic" keyword
3   -- is not used
4 begin
5   X := E;
6 end Set;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Show\_Generic\_Body\_Definition  
MD5: de611ef77b528543fd6bad82c53857f7

## 13.5 Generic instantiation

Generic subprograms or packages can't be used directly. Instead, they need to be instantiated, which we do using the `new` keyword, as shown in the following example:

Listing 164: set.ads

```

1 generic
2   type T is private;
3   X : in out T;
4   -- X can be used in the Set procedure
5 procedure Set (E : T);

```

Listing 165: set.adb

```

1 procedure Set (E : T) is
2 begin
3   X := E;
4 end Set;

```

Listing 166: show\_generic\_instantiation.adb

```

1 with Ada.Text_IO; use Ada.Text_IO;
2 with Set;
3
4 procedure Show_Generic_Instantiation is
5
6   Main    : Integer := 0;
7   Current : Integer;
8
9   procedure Set_Main is new Set (T => Integer,
10                                X => Main);
11   -- Here, we map the formal parameters to
12   -- actual types and objects.
13   --
14   -- The same approach can be used to
15   -- instantiate functions or packages, e.g.:

```

(continues on next page)

(continued from previous page)

```
16  -- 
17  --  function Get_Main is new ...
18  --  package Integer_Queue is new ...
19
20 begin
21   Current := 10;
22
23   Set_Main (Current);
24   Put_Line ("Value of Main is "
25             & Integer'Image (Main));
26 end Show_Generic_Instantiation;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Show\_Generic\_Instantiation  
MD5: 13dc0692252496d954240952561e1c05

### Runtime output

Value of Main is 10

In the example above, we instantiate the procedure Set by mapping the formal parameters T and X to actual existing elements, in this case the **Integer** type and the Main variable.

## 13.6 Generic packages

The previous examples focused on generic subprograms. In this section, we look at generic packages. The syntax is similar to that used for generic subprograms: we start with the **generic** keyword and continue with formal declarations. The only difference is that **package** is specified instead of a subprogram keyword.

Here's an example:

Listing 167: element.ads

```
1 generic
2   type T is private;
3 package Element is
4
5   procedure Set (E : T);
6   procedure Reset;
7   function Get return T;
8   function Is_Valid return Boolean;
9
10  Invalid_Element : exception;
11
12 private
13   Value : T;
14   Valid : Boolean := False;
15 end Element;
```

Listing 168: element.adb

```
1 package body Element is
2
3   procedure Set (E : T) is
4   begin
5     Value := E;
6     Valid := True;
```

(continues on next page)

(continued from previous page)

```

7  end Set;

8

9  procedure Reset is
10 begin
11     Valid := False;
12 end Reset;

13

14  function Get return T is
15 begin
16     if not Valid then
17         raise Invalid_Element;
18     end if;
19     return Value;
20 end Get;

21

22  function Is_Valid return Boolean is (Valid);
23 end Element;

```

Listing 169: show\_generic\_package.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2  with Element;
3
4  procedure Show_Generic_Package is
5
6      package I is new Element (T => Integer);
7
8      procedure Display_Initialized is
9      begin
10         if I.Is_Valid then
11             Put_Line ("Value is initialized");
12         else
13             Put_Line ("Value is not initialized");
14         end if;
15     end Display_Initialized;
16
17 begin
18     Display_Initialized;
19
20     Put_Line ("Initializing...");
21     I.Set (5);
22     Display_Initialized;
23     Put_Line ("Value is now set to "
24             & Integer'Image (I.Get));
25
26     Put_Line ("Resetting...");
27     I.Reset;
28     Display_Initialized;
29
30 end Show_Generic_Package;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Show\_Generic\_Package  
MD5: c5278a06c6d06f1f37353ee0ca6686ec

### Runtime output

```

Value is not initialized
Initializing...
Value is initialized

```

(continues on next page)

(continued from previous page)

```
Value is now set to 5
Resetting...
Value is not initialized
```

In the example above, we created a simple container named `Element`, with just one single element. This container tracks whether the element has been initialized or not.

After writing the package definition, we create the instance `I` of the `Element`. We use the instance by calling the package subprograms (Set, Reset, and Get).

## 13.7 Formal subprograms

In addition to formal types and objects, we can also declare formal subprograms or packages. This course only describes formal subprograms; formal packages are discussed in the advanced course.

We use the `with` keyword to declare a formal subprogram. In the example below, we declare a formal function (`Comparison`) to be used by the generic procedure `Check`.

Listing 170: check.ads

```
1 generic
2   Description : String;
3   type T is private;
4   with function Comparison (X, Y : T)
5           return Boolean;
6 procedure Check (X, Y : T);
```

Listing 171: check.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Check (X, Y : T) is
4   Result : Boolean;
5 begin
6   Result := Comparison (X, Y);
7   if Result then
8     Put_Line
9       ("Comparison (" & Description & ") between arguments is OK!");
10  else
11    Put_Line
12      ("Comparison (" & Description & ") between arguments is not OK!");
13  end if;
14 end Check;
```

Listing 172: show\_formal\_subprogram.adb

```
1 with Check;
2
3 procedure Show_Formal_Subprogram is
4
5   A, B : Integer;
6
7   procedure Check_Is_Equal is new
8     Check (Description => "equality",
```

(continues on next page)

(continued from previous page)

```

9      T      => Integer,
10     Comparison  => Standard."=");
11     -- Here, we are mapping the standard
12     -- equality operator for Integer types to
13     -- the Comparison formal function
14 begin
15   A := 0;
16   B := 1;
17   Check_Is_Equal (A, B);
18 end Show_Formal_Subprogram;

```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.Generics.Show\_Formal\_Subprogram  
MD5: 1c463a47e9ce56b5afbc1da6acd116d

**Runtime output**

Comparison (equality) between arguments is not OK!

## 13.8 Example: I/O instances

Ada offers generic I/O packages that can be instantiated for standard and derived types. One example is the generic `Float_Io` package, which provides procedures such as `Put` and `Get`. In fact, `Float_Text_Io` — available from the standard library — is an instance of the `Float_Io` package, and it's defined as:

```

with Ada.Text_Io;

package Ada.Float_Text_Io is new Ada.Text_Io.Float_Io (Float);

```

You can use it directly with any object of floating-point type. For example:

Listing 173: `show_float_text_io.adb`

```

1 with Ada.Float_Text_Io;
2
3 procedure Show_Float_Text_Io is
4   X : constant Float := 2.5;
5
6   use Ada.Float_Text_Io;
7 begin
8   Put (X);
9 end Show_Float_Text_Io;

```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.Generics.Show\_Float\_Text\_Io  
MD5: 7cc9b547ef301a2071e9fb65caa4631b

**Runtime output**

2.50000E+00

Instantiating generic I/O packages can be useful for derived types. For example, let's create a new type `Price` that must be displayed with two decimal digits after the point, and no exponent.

Listing 174: show\_float\_io\_inst.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Float_IO_Inst is
4
5    type Price is digits 3;
6
7    package Price_IO is new
8      Ada.Text_IO.Float_IO (Price);
9
10   P : Price;
11
12  begin
13    -- Set to zero => don't display exponent
14    Price_IO.Default_Exp := 0;
15
16    P := 2.5;
17    Price_IO.Put (P);
18    New_Line;
19
20    P := 5.75;
21    Price_IO.Put (P);
22    New_Line;
23  end Show_Float_IO_Inst;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Show\_Float\_IO\_Inst  
MD5: 583c761421d7fdb812dd2a183b676bae

### Runtime output

```
2.50
5.75
```

By adjusting `Default_Exp` from the `Price_IO` instance to *remove* the exponent, we can control how variables of `Price` type are displayed. Just as a side note, we could also have written:

```
-- [...]
type Price is new Float;
package Price_IO is new
  Ada.Text_IO.Float_IO (Price);
begin
  Price_IO.Default_Aft := 2;
  Price_IO.Default_Exp := 0;
```

In this case, we're adjusting `Default_Aft`, too, to get two decimal digits after the point when calling `Put`.

In addition to the generic `Float_IO` package, the following generic packages are available from `Ada.Text_IO`:

- `Enumeration_IO` for enumeration types;
- `Integer_IO` for integer types;
- `Modular_IO` for modular types;
- `Fixed_IO` for fixed-point types;

- `Decimal_I0` for decimal types.

In fact, we could rewrite the example above using decimal types:

Listing 175: `show_decimal_io_inst.adb`

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show.Decimal_I0_Inst is
4
5    type Price is delta 10.0 ** (-2) digits 12;
6
7    package Price_I0 is new
8      Ada.Text_IO.Decimal_I0 (Price);
9
10   P : Price;
11 begin
12   Price_I0.Default_Exp := 0;
13
14   P := 2.5;
15   Price_I0.Put (P);
16   New_Line;
17
18   P := 5.75;
19   Price_I0.Put (P);
20   New_Line;
21 end Show.Decimal_I0_Inst;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Show.Decimal\_I0\_Inst  
MD5: f413570759dcb32cc166078b3ceelal6

### Runtime output

2.50  
5.75

## 13.9 Example: ADTs

An important application of generics is to model abstract data types (ADTs). In fact, Ada includes a library with numerous ADTs using generics: `Ada.Containers` (described in the [containers section](#) (page 201)).

A typical example of an ADT is a stack:

Listing 176: `stacks.ads`

```

1  generic
2    Max : Positive;
3    type T is private;
4  package Stacks is
5
6    type Stack is limited private;
7
8    Stack_Underflow, Stack_Overflow : exception;
9
10   function Is_Empty (S : Stack) return Boolean;
11
12   function Pop (S : in out Stack) return T;
13

```

(continues on next page)

(continued from previous page)

```

14  procedure Push (S : in out Stack;
15      V :          T);
16
17  private
18
19  type Stack_Array is
20      array (Natural range <>) of T;
21
22  Min : constant := 1;
23
24  type Stack is record
25      Container : Stack_Array (Min .. Max);
26      Top        : Natural := Min - 1;
27  end record;
28
29 end Stacks;

```

Listing 177: stacks.adb

```

1  package body Stacks is
2
3      function Is_Empty (S : Stack) return Boolean is
4          (S.Top < S.Container'First);
5
6      function Is_Full (S : Stack) return Boolean is
7          (S.Top >= S.Container'Last);
8
9      function Pop (S : in out Stack) return T is
10     begin
11         if Is_Empty (S) then
12             raise Stack_Underflow;
13         else
14             return X : T do
15                 X      := S.Container (S.Top);
16                 S.Top := S.Top - 1;
17             end return;
18         end if;
19     end Pop;
20
21     procedure Push (S : in out Stack;
22                     V :          T) is
23     begin
24         if Is_Full (S) then
25             raise Stack_Overflow;
26         else
27             S.Top          := S.Top + 1;
28             S.Container (S.Top) := V;
29         end if;
30     end Push;
31
32 end Stacks;

```

Listing 178: show\_stack.adb

```

1  with Ada.Text_Io; use Ada.Text_Io;
2  with Stacks;
3
4  procedure Show_Stack is
5
6      package Integer_Stacks is new

```

(continues on next page)

(continued from previous page)

```

7  Stacks (Max => 10,
8    T => Integer);
9  use Integer_Stacks;
10
11  Values : Integer_Stacks.Stack;
12
13 begin
14  Push (Values, 10);
15  Push (Values, 20);
16
17  Put_Line ("Last value was "
18    & Integer'Image (Pop (Values)));
19 end Show_Stack;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Show\_Stack  
MD5: ee112d395552c1a02d211b9e5425dc71

### Runtime output

Last value was 20

In this example, we first create a generic stack package (**Stacks**) and then instantiate it to create a stack of up to 10 integer values.

## 13.10 Example: Swap

Let's look at a simple procedure that swaps variables of type **Color**:

Listing 179: colors.ads

```

1  package Colors is
2    type Color is (Black, Red, Green,
3      Blue, White);
4
5    procedure Swap_Colors (X, Y : in out Color);
6  end Colors;

```

Listing 180: colors.adb

```

1  package body Colors is
2
3    procedure Swap_Colors (X, Y : in out Color) is
4      Tmp : constant Color := X;
5    begin
6      X := Y;
7      Y := Tmp;
8    end Swap_Colors;
9
10 end Colors;

```

Listing 181: test\_non\_generic\_swap\_colors.adb

```

1  with Ada.Text_Io; use Ada.Text_Io;
2  with Colors;      use Colors;
3
4  procedure Test_Non_Generic_Swap_Colors is

```

(continues on next page)

(continued from previous page)

```
5  A, B, C : Color;
6  begin
7      A := Blue;
8      B := White;
9      C := Red;
10
11     Put_Line ("Value of A is "
12                  & Color'Image (A));
13     Put_Line ("Value of B is "
14                  & Color'Image (B));
15     Put_Line ("Value of C is "
16                  & Color'Image (C));
17
18     New_Line;
19     Put_Line ("Swapping A and C... ");
20     New_Line;
21     Swap_Colors (A, C);
22
23     Put_Line ("Value of A is "
24                  & Color'Image (A));
25     Put_Line ("Value of B is "
26                  & Color'Image (B));
27     Put_Line ("Value of C is "
28                  & Color'Image (C));
29 end Test_Non_Generic_Swap_Colors;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Test\_Non\_Generic\_Swap\_Colors  
MD5: 4d1cf826a1676c3750a8aabd484ac71f

### Runtime output

```
Value of A is BLUE
Value of B is WHITE
Value of C is RED

Swapping A and C...

Value of A is RED
Value of B is WHITE
Value of C is BLUE
```

In this example, `Swap_Colors` can only be used for the `Color` type. However, this algorithm can theoretically be used for any type, whether an enumeration type or a complex record type with many elements. The algorithm itself is the same: it's only the type that differs. If, for example, we want to swap variables of `Integer` type, we don't want to duplicate the implementation. Therefore, such an algorithm is a perfect candidate for abstraction using generics.

In the example below, we create a generic version of `Swap_Colors` and name it `Generic_Swap`. This generic version can operate on any type due to the declaration of formal type `T`.

Listing 182: generic\_swap.ads

```
1  generic
2      type T is private;
3  procedure Generic_Swap (X, Y : in out T);
```

Listing 183: generic\_swap.adb

```

1  procedure Generic_Swap (X, Y : in out T) is
2    Tmp : constant T := X;
3  begin
4    X := Y;
5    Y := Tmp;
6  end Generic_Swap;

```

Listing 184: colors.ads

```

1  with Generic_Swap;
2
3  package Colors is
4
5    type Color is (Black, Red, Green,
6                  Blue, White);
7
8    procedure Swap_Colors is new
9      Generic_Swap (T => Color);
10
11 end Colors;

```

Listing 185: test\_swap\_colors.adb

```

1  with Ada.Text_IO;  use Ada.Text_IO;
2  with Colors;       use Colors;
3
4  procedure Test_Swap_Colors is
5    A, B, C : Color;
6  begin
7    A := Blue;
8    B := White;
9    C := Red;
10
11   Put_Line ("Value of A is "
12             & Color'Image (A));
13   Put_Line ("Value of B is "
14             & Color'Image (B));
15   Put_Line ("Value of C is "
16             & Color'Image (C));
17
18   New_Line;
19   Put_Line ("Swapping A and C... ");
20   New_Line;
21   Swap_Colors (A, C);
22
23   Put_Line ("Value of A is "
24             & Color'Image (A));
25   Put_Line ("Value of B is "
26             & Color'Image (B));
27   Put_Line ("Value of C is "
28             & Color'Image (C));
29 end Test_Swap_Colors;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Test\_Swap\_Colors  
MD5: a5d94a40bd9d1c6736cc873f8b58e867

### Runtime output

```
Value of A is BLUE
Value of B is WHITE
Value of C is RED
```

```
Swapping A and C...
```

```
Value of A is RED
Value of B is WHITE
Value of C is BLUE
```

As we can see in the example, we can create the same `Swap_Colors` procedure as we had in the non-generic version of the algorithm by declaring it as an instance of the generic `Generic_Swap` procedure. We specify that the generic `T` type will be mapped to the `Color` type by passing it as an argument to the `Generic_Swap` instantiation.

### 13.11 Example: Reversing

The previous example, with an algorithm to swap two values, is one of the simplest examples of using generics. Next we study an algorithm for reversing elements of an array. First, let's start with a non-generic version of the algorithm, one that works specifically for the `Color` type:

Listing 186: colors.ads

```
1 package Colors is
2
3     type Color is (Black, Red, Green,
4                     Blue, White);
5
6     type Color_Array is
7         array (Integer range <>) of Color;
8
9     procedure Reverse_It (X : in out Color_Array);
10
11 end Colors;
```

Listing 187: colors.adb

```
1 package body Colors is
2
3     procedure Reverse_It (X : in out Color_Array)
4     is
5     begin
6         for I in X'First .. (X'Last + X'First) / 2 loop
7             declare
8                 Tmp      : Color;
9                 X_Left  : Color
10                renames X (I);
11                 X_Right : Color
12                renames X (X'Last + X'First - I);
13             begin
14                 Tmp      := X_Left;
15                 X_Left  := X_Right;
16                 X_Right := Tmp;
17             end;
18         end loop;
19     end Reverse_It;
20
21 end Colors;
```

Listing 188: test\_non\_generic\_reverse\_colors.adb

```

1  with Ada.Text_Io; use Ada.Text_Io;
2  with Colors;      use Colors;
3
4  procedure Test_Non_Generic_Reverse_Colors is
5
6      My_Colors : Color_Array (1 .. 5) :=
7          (Black, Red, Green, Blue, White);
8
9  begin
10     for C of My_Colors loop
11         Put_Line ("My_Color: " & Color'Image (C));
12     end loop;
13
14     New_Line;
15     Put_Line ("Reversing My_Color...");
16     New_Line;
17     Reverse_It (My_Colors);
18
19     for C of My_Colors loop
20         Put_Line ("My_Color: " & Color'Image (C));
21     end loop;
22
23 end Test_Non_Generic_Reverse_Colors;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Test\_Non\_Generic\_Reverse\_Colors  
MD5: 9b3a489d0bc0ecd79de6ba99fd7cd44f

### Runtime output

```

My_Color: BLACK
My_Color: RED
My_Color: GREEN
My_Color: BLUE
My_Color: WHITE

Reversing My_Color...

My_Color: WHITE
My_Color: BLUE
My_Color: GREEN
My_Color: RED
My_Color: BLACK

```

The procedure `Reverse_It` takes an array of colors, starts by swapping the first and last elements of the array, and continues doing that with successive elements until it reaches the middle of array. At that point, the entire array has been reversed, as we see from the output of the test program.

To abstract this procedure, we declare formal types for three components of the algorithm:

- the elements of the array (`Color` type in the example)
- the range used for the array (`Integer` range in the example)
- the actual array type (`Color_Array` type in the example)

This is a generic version of the algorithm:

Listing 189: generic\_reverse.ads

```
1 generic
2   type T is private;
3   type Index is range <>;
4   type Array_T is
5     array (Index range <>) of T;
6 procedure Generic_Reverse (X : in out Array_T);
```

Listing 190: generic\_reverse.adb

```
1 procedure Generic_Reverse (X : in out Array_T) is
2 begin
3   for I in X'First .. (X'Last + X'First) / 2 loop
4     declare
5       Tmp      : T;
6       X_Left  : T
7         renames X (I);
8       X_Right : T
9         renames X (X'Last + X'First - I);
10    begin
11      Tmp      := X_Left;
12      X_Left  := X_Right;
13      X_Right := Tmp;
14    end;
15  end loop;
16 end Generic_Reverse;
```

Listing 191: colors.ads

```
1 with Generic_Reverse;
2
3 package Colors is
4
5   type Color is (Black, Red, Green,
6                 Blue, White);
7
8   type Color_Array is
9     array (Integer range <>) of Color;
10
11  procedure Reverse_It is new
12    Generic_Reverse (T      => Color,
13                     Index  => Integer,
14                     Array_T => Color_Array);
15
16 end Colors;
```

Listing 192: test\_reverse\_colors.adb

```
1 with Ada.Text_Io; use Ada.Text_Io;
2 with Colors;      use Colors;
3
4 procedure Test_Reverse_Colors is
5
6   My_Colors : Color_Array (1 .. 5) :=
7     (Black, Red, Green, Blue, White);
8
9 begin
10   for C of My_Colors loop
```

(continues on next page)

(continued from previous page)

```

11  Put_Line ("My_Color: "
12    & Color'Image (C));
13  end loop;
14
15  New_Line;
16  Put_Line ("Reversing My_Color...");
17  New_Line;
18  Reverse_It (My_Colors);
19
20  for C of My_Colors loop
21    Put_Line ("My_Color: "
22      & Color'Image (C));
23  end loop;
24
25 end Test_Reverse_Colors;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Test\_Reverse\_Colors  
MD5: 9ef175c517d7574b4b65b24ba0027f1f

### Runtime output

```

My_Color: BLACK
My_Color: RED
My_Color: GREEN
My_Color: BLUE
My_Color: WHITE

Reversing My_Color...

My_Color: WHITE
My_Color: BLUE
My_Color: GREEN
My_Color: RED
My_Color: BLACK

```

As mentioned above, we're abstracting three components of the algorithm:

- the T type abstracts the elements of the array
- the Index type abstracts the range used for the array
- the Array\_T type abstracts the array type and uses the formal declarations of the T and Index types.

## 13.12 Example: Test application

In the previous example we've focused only on abstracting the reversing algorithm itself. However, we could have decided to also abstract our small test application. This could be useful if we, for example, decide to test other procedures that change elements of an array.

In order to do this, we again have to choose the elements to abstract. We therefore declare the following formal parameters:

- S: the string containing the array name
- a function Image that converts an element of type T to a string
- a procedure Test that performs some operation on the array

Note that Image and Test are examples of formal subprograms and S is an example of a formal object.

## Introduction to Ada

Here is a version of the test application making use of the generic `Perform_Test` procedure:

Listing 193: `generic_reverse.ads`

```
1 generic
2   type T is private;
3   type Index is range <>;
4   type Array_T is
5     array (Index range <>) of T;
6   procedure Generic_Reverse (X : in out Array_T);
```

Listing 194: `generic_reverse.adb`

```
1 procedure Generic_Reverse (X : in out Array_T) is
2 begin
3   for I in X'First .. (X'Last + X'First) / 2 loop
4     declare
5       Tmp      : T;
6       X_Left  : T
7       renames X (I);
8       X_Right : T
9       renames X (X'Last + X'First - I);
10 begin
11   Tmp      := X_Left;
12   X_Left  := X_Right;
13   X_Right := Tmp;
14 end;
15 end loop;
16 end Generic_Reverse;
```

Listing 195: `perform_test.ads`

```
1 generic
2   type T is private;
3   type Index is range <>;
4   type Array_T is
5     array (Index range <>) of T;
6   S : String;
7   with function Image (E : T)
8     return String is <>;
9   with procedure Test (X : in out Array_T);
10  procedure Perform_Test (X : in out Array_T);
```

Listing 196: `perform_test.adb`

```
1 with Ada.Text_Io;  use Ada.Text_Io;
2
3 procedure Perform_Test (X : in out Array_T) is
4 begin
5   for C of X loop
6     Put_Line (S & ": " & Image (C));
7   end loop;
8
9   New_Line;
10  Put_Line ("Testing " & S & "...");
11  New_Line;
12  Test (X);
13
14  for C of X loop
15    Put_Line (S & ": " & Image (C));
```

(continues on next page)

(continued from previous page)

```

16    end loop;
17 end Perform_Test;

```

Listing 197: colors.ads

```

1 with Generic_Reverse;
2
3 package Colors is
4
5   type Color is (Black, Red, Green,
6                  Blue, White);
7
8   type Color_Array is
9     array (Integer range <>) of Color;
10
11  procedure Reverse_It is new
12    Generic_Reverse (T      => Color,
13                     Index  => Integer,
14                     Array_T => Color_Array);
15
16 end Colors;

```

Listing 198: test\_reverse\_colors.adb

```

1 with Colors;      use Colors;
2 with Perform_Test;
3
4 procedure Test_Reverse_Colors is
5
6   procedure Perform_Test_Reverse_It is new
7     Perform_Test (T      => Color,
8                   Index  => Integer,
9                   Array_T => Color_Array,
10                  S      => "My_Color",
11                  Image  => Color'Image,
12                  Test   => Reverse_It);
13
14   My_Colors : Color_Array (1 .. 5) :=
15     (Black, Red, Green, Blue, White);
16
17 begin
18   Perform_Test_Reverse_It (My_Colors);
19 end Test_Reverse_Colors;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Generics.Test\_Reverse\_Colors\_2  
MD5: 04640309f4f7e9f8bcff137d1a6f8733

### Runtime output

```

My_Color: BLACK
My_Color: RED
My_Color: GREEN
My_Color: BLUE
My_Color: WHITE

Testing My_Color...

My_Color: WHITE

```

(continues on next page)

(continued from previous page)

```
My_Color: BLUE
My_Color: GREEN
My_Color: RED
My_Color: BLACK
```

In this example, we create the procedure `Perform_Test_Reverse_It` as an instance of the generic procedure (`Perform_Test`). Note that:

- For the formal `Image` function, we use the '`Image` attribute of the `Color` type
- For the formal `Test` procedure, we reference the `Reverse_Array` procedure from the package.

## EXCEPTIONS

Ada uses exceptions for error handling. Unlike many other languages, Ada speaks about *raising*, not *throwing*, an exception and *handling*, not *catching*, an exception.

### 14.1 Exception declaration

Ada exceptions are not types, but instead objects, which may be peculiar to you if you're used to the way Java or Python support exceptions. Here's how you declare an exception:

Listing 199: exceptions.ads

```
1 package Exceptions is
2     My_Except : exception;
3     -- Like an object. *NOT* a type !
4 end Exceptions;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Exceptions.Show\_Exception  
MD5: 6201faeca9b029c790023856d2c8c419

Even though they're objects, you're going to use each declared exception object as a "kind" or "family" of exceptions. Ada does not require that a subprogram declare every exception it can potentially raise.

### 14.2 Raising an exception

To raise an exception of our newly declared exception kind, do the following:

Listing 200: main.adb

```
1 with Exceptions; use Exceptions;
2
3 procedure Main is
4 begin
5     raise My_Except;
6     -- Execution of current control flow
7     -- abandoned; an exception of kind
8     -- "My_Except" will bubble up until it
9     -- is caught.
10 end Main;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Exceptions.Show\_Exception  
MD5: 24b40ae1509722adf51c3dd0d3ea4fbe

### Runtime output

```
raised EXCEPTIONS.MY_EXCEPT : main.adb:5
```

Here, the My\_Except exception is raised. We can also specify a message:

Listing 201: main.adb

```
1 with Exceptions; use Exceptions;
2
3 procedure Main is
4 begin
5   raise My_Except with "My exception message";
6   -- Execution of current control flow
7   -- abandoned; an exception of kind
8   -- "My_Except" with associated string will
9   -- bubble up until it is caught.
10 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Exceptions.Show\_Exception  
MD5: 279299c9703c3ed4e51fdd7c3a5e1392

### Runtime output

```
raised EXCEPTIONS.MY_EXCEPT : My exception message
```

In this case, we see an additional message when the exception is displayed.

## 14.3 Handling an exception

Next, we address how to handle exceptions that were raised by us or libraries that we call. The neat thing in Ada is that you can add an exception handler to any statement block as follows:

Listing 202: open\_file.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2 with Ada.Exceptions; use Ada.Exceptions;
3
4 procedure Open_File is
5   File : File_Type;
6 begin
7   -- Block (sequence of statements)
8 begin
9   Open (File, In_File, "input.txt");
10 exception
11   when E : Name_Error =>
12   -- ^ Exception to be handled
13   Put ("Cannot open input file : ");
14   Put_Line (Exception_Message (E));
15   raise;
16   -- Reraise current occurrence
17 end;
18 end Open_File;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Exceptions.Show\_Exception\_Handling  
MD5: 4ea1d5da684a6d7d7ee32908810e9c8f

### Runtime output

```
Cannot open input file : input.txt: No such file or directory
raised ADA.IO_EXCEPTIONS.NAME_ERROR : input.txt: No such file or directory
```

In the example above, we're using the `Exception_Message` function from the `Ada.Exceptions` package. This function returns the message associated with the exception as a string.

You don't need to introduce a block just to handle an exception: you can add it to the statements block of your current subprogram:

Listing 203: open\_file.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2  with Ada.Exceptions; use Ada.Exceptions;
3
4  procedure Open_File is
5    File : File_Type;
6  begin
7    Open (File, In_File, "input.txt");
8    -- Exception block can be added to any block
9  exception
10    when Name_Error =>
11      Put ("Cannot open input file");
12  end Open_File;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Exceptions.Show\_Exception\_Message  
MD5: 838e87ae416b3a717901cdc00eb71b40

### Runtime output

```
Cannot open input file
```

#### Attention

Exception handlers have an important restriction that you need to be careful about: Exceptions raised in the declarative section are not caught by the handlers of that block. So for example, in the following code, the exception will not be caught.

Listing 204: be\_careful.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2  with Ada.Exceptions; use Ada.Exceptions;
3
4  procedure Be_Careful is
5    function Dangerous return Integer is
6    begin
7      raise Constraint_Error;
8      return 42;
9    end Dangerous;
10
11 begin
12  declare
13    A : Integer := Dangerous;
```

```
14  begin
15      Put_Line (Integer'Image (A));
16  exception
17      when Constraint_Error =>
18          Put_Line ("error!");
19  end;
20 end Be_Careful;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Exceptions.Be\_Careful  
MD5: 6ea8a214bbbaca09d7444136d069e782

### Runtime output

```
raised CONSTRAINT_ERROR : be_careful.adb:7 explicit raise
```

This is also the case for the top-level exception block that is part of the current subprogram.

## 14.4 Predefined exceptions

Ada has a very small number of predefined exceptions:

- `Constraint_Error` is the main one you might see. It's raised:
  - When bounds don't match or, in general, any violation of constraints.
  - In case of overflow
  - In case of null dereferences
  - In case of division by 0
- `Program_Error` might appear, but probably less often. It's raised in more arcane situations, such as for order of elaboration issues and some cases of detectable erroneous execution.
- `Storage_Error` will happen because of memory issues, such as:
  - Not enough memory (allocator)
  - Not enough stack
- `Tasking_Error` will happen with task related errors, such as any error happening during task activation.

You should not reuse predefined exceptions. If you do then, it won't be obvious when one is raised that it is because something went wrong in a built-in language operation.

## TASKING

Tasks and protected objects allow the implementation of concurrency in Ada. The following sections explain these concepts in more detail.

### 15.1 Tasks

A task can be thought as an application that runs *concurrently* with the main application. In other programming languages, a task might be called a [thread<sup>17</sup>](#), and tasking might be called [multithreading<sup>18</sup>](#).

Tasks may synchronize with the main application but may also process information completely independently from the main application. Here we show how this is accomplished.

#### 15.1.1 Simple task

Tasks are declared using the keyword **task**. The task implementation is specified in a **task body** block. For example:

Listing 205: show\_simple\_task.adb

```
1 with Ada.Text_Io; use Ada.Text_Io;
2
3 procedure Show_Simple_Task is
4     task T;
5
6     task body T is
7         begin
8             Put_Line ("In task T");
9         end T;
10    begin
11        Put_Line ("In main");
12    end Show_Simple_Task;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Simple\_Task  
MD5: b17d9b35b4b2b53bc59776749e1be219

#### Runtime output

```
In task T
In main
```

<sup>17</sup> [https://en.wikipedia.org/wiki/Thread\\_\(computing\)](https://en.wikipedia.org/wiki/Thread_(computing))

<sup>18</sup> [https://en.wikipedia.org/wiki/Thread\\_\(computing\)#Multithreading](https://en.wikipedia.org/wiki/Thread_(computing)#Multithreading)

Here, we're declaring and implementing the task T. As soon as the main application starts, task T starts automatically — it's not necessary to manually start this task. By running the application above, we can see that both calls to Put\_Line are performed.

Note that:

- The main application is itself a task (the main or “environment” task).
  - In this example, the subprogram Show\_Simple\_Task is the main task of the application.
- Task T is a subtask.
  - Each subtask has a master, which represents the program construct in which the subtask is declared. In this case, the main subprogram Show\_Simple\_Task is T's master.
  - The master construct is executed by some enclosing task, which we will refer to as the “master task” of the subtask.
- The number of tasks is not limited to one: we could include a task T2 in the example above.
  - This task also starts automatically and runs *concurrently* with both task T and the main task. For example:

Listing 206: show\_simple\_tasks.adb

```
1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Show_Simple_Tasks is
4      task T;
5      task T2;
6
7      task body T is
8          begin
9              Put_Line ("In task T");
10             end T;
11
12      task body T2 is
13          begin
14              Put_Line ("In task T2");
15             end T2;
16
17      begin
18          Put_Line ("In main");
19      end Show_Simple_Tasks;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Multiple\_Simple\_Task  
MD5: 5e24b797e742bec306ad498f4f40d2b4

### Runtime output

```
In task T
In task T2
In main
```

## 15.1.2 Simple synchronization

As we've just seen, as soon as the master construct reaches its “begin”, its subtasks also start automatically. The master continues its processing until it has nothing more to do. At that point, however, it will not terminate. Instead, the master waits until its subtasks have

finished before it allows itself to complete. In other words, this waiting process provides synchronization between the master task and its subtasks. After this synchronization, the master construct will complete. For example:

Listing 207: show\_simple\_sync.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Simple_Sync is
4      task T;
5      task body T is
6          begin
7              for I in 1 .. 10 loop
8                  Put_Line ("hello");
9              end loop;
10             end T;
11     begin
12         null;
13         -- Will wait here until all tasks
14         -- have terminated
15     end Show_Simple_Sync;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Simple\_Sync  
MD5: 84afce465854f99f8cbe0b57714d8a5f

### Runtime output

```

hello

```

The same mechanism is used for other subprograms that contain subtasks: the subprogram execution will wait for its subtasks to finish. So this mechanism is not limited to the main subprogram and also applies to any subprogram called by the main subprogram, directly or indirectly.

Synchronization also occurs if we move the task to a separate package. In the example below, we declare a task T in the package Simple\_Sync\_Pkg.

Listing 208: simple\_sync\_pkg.ads

```

1  package Simple_Sync_Pkg is
2      task T;
3  end Simple_Sync_Pkg;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Simple\_Sync\_Pkg  
MD5: 2f9be044d04994240970f150e2293d5e

This is the corresponding package body:

Listing 209: simple\_sync\_pkg.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 package body Simple_Sync_Pkg is
4     task body T is
5         begin
6             for I in 1 .. 10 loop
7                 Put_Line ("hello");
8             end loop;
9         end T;
10    end Simple_Sync_Pkg;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Simple\_Sync\_Pkg  
MD5: b668451e4fb10e802f619889bcd743ff

Because the package is **with**'ed by the main procedure, the task T defined in the package will become a subtask of the main task. For example:

Listing 210: test\_simple\_sync\_pkg.adb

```
1 with Simple_Sync_Pkg;
2
3 procedure Test_Simple_Sync_Pkg is
4 begin
5     null;
6     -- Will wait here until all tasks
7     -- have terminated
8 end Test_Simple_Sync_Pkg;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Simple\_Sync\_Pkg  
MD5: e51565b91767ce198496ef3e9c582ac8

### Runtime output

```
hello
```

As soon as the main subprogram returns, the main task synchronizes with any subtasks spawned by packages T from Simple\_Sync\_Pkg before finally terminating.

### 15.1.3 Delay

We can introduce a delay by using the keyword **delay**. This puts the current task to sleep for the length of time (in seconds) specified in the delay statement. For example:

Listing 211: show\_delay.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Delay is
4
5      task T;
6
7      task body T is
8          begin
9              for I in 1 .. 5 loop
10                  Put_Line ("hello from task T");
11                  delay 1.0;
12                  -- ^ Wait 1.0 seconds
13              end loop;
14          end T;
15      begin
16          delay 1.5;
17          Put_Line ("hello from main");
18      end Show_Delay;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Delay  
MD5: 4a6e8039744301a128e8fb2dd27902a5

### Runtime output

```

hello from task T
hello from task T
hello from main
hello from task T
hello from task T
hello from task T

```

In this example, we're making the task T wait one second after each time it displays the "hello" message. In addition, the main task is waiting 1.5 seconds before displaying its own "hello" message

#### 15.1.4 Synchronization: rendezvous

The only type of synchronization we've seen so far is the one that happens automatically at the end of a master construct with a subtask. You can also define custom synchronization points using the keyword **entry**. An **entry** can be viewed as a special kind of subprogram, which is called by another task using a similar syntax, as we will see later.

In the task body definition, you define which part of the task will accept the entries by using the keyword **accept**. A task proceeds until it reaches an **accept** statement and then waits for some other task to synchronize with it. Specifically,

- The task with the entry waits at that point (in the **accept** statement), ready to accept a call to the corresponding entry from the master task.
- The other task calls the task entry, in a manner similar to a procedure call, to synchronize with the entry.

This synchronization between tasks is called a *rendezvous*. Let's see an example:

Listing 212: show\_rendezvous.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Rendezvous is
4
5      task T is
6          entry Start;
7      end T;
8
9      task body T is
10     begin
11         accept Start;
12         -- ^ Waiting for somebody
13         -- to call the entry
14
15         Put_Line ("In T");
16     end T;
17
18 begin
19     Put_Line ("In Main");
20
21     -- Calling T's entry:
22     T.Start;
23 end Show_Rendezvous;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Rendezvous  
MD5: 479eea7adc876ac359ad20ac6e3acf66

### Runtime output

```
In Main
In T
```

In this example, we declare an entry Start for task T. In the task body, we implement this entry using `accept Start`. When task T reaches this point, it waits for some other task to call its entry. This synchronization occurs in the `T.Start` statement. After the rendezvous completes, the main task and task T again run concurrently until they synchronize one final time when the main subprogram `Show_Rendezvous` finishes.

An entry may be used to perform more than a simple task synchronization: it also may perform multiple statements during the time both tasks are synchronized. We do this with a `do ... end` block. For the previous example, we would simply write `accept Start do <statements>; end;`. We use this kind of block in the next example.

### 15.1.5 Select loop

There's no limit to the number of times an entry can be accepted. We could even create an infinite loop in the task and accept calls to the same entry over and over again. An infinite loop, however, prevents the subtask from finishing, so it blocks its master task when it reaches the end of its processing. Therefore, a loop containing `accept` statements in a task body can be used in conjunction with a `select ... or terminate` statement. In simple terms, this statement allows its master task to automatically terminate the subtask when the master construct reaches its end. For example:

Listing 213: show\_rendezvous\_loop.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
```

(continues on next page)

(continued from previous page)

```

3  procedure Show_Rendezvous_Loop is
4
5      task T is
6          entry Reset;
7          entry Increment;
8      end T;
9
10     task body T is
11         Cnt : Integer := 0;
12     begin
13         loop
14             select
15                 accept Reset do
16                     Cnt := 0;
17                     end Reset;
18                     Put_Line ("Reset");
19                 or
20                 accept Increment do
21                     Cnt := Cnt + 1;
22                     end Increment;
23                     Put_Line ("In T's loop (" & Integer'Image (Cnt) & ")");
24                 or
25                     terminate;
26             end select;
27         end loop;
28     end T;
29
30 begin
31     Put_Line ("In Main");
32
33     for I in 1 .. 4 loop
34         -- Calling T's entry multiple times
35         T.Increment;
36     end loop;
37
38     T.Reset;
39     for I in 1 .. 4 loop
40         -- Calling T's entry multiple times
41         T.Increment;
42     end loop;
43
44 end Show_Rendezvous_Loop;
45
46

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Rendezvous\_Loop  
MD5: 0542dbc029cffb9f794d761bab9f3a9d

### Runtime output

```

In Main
In T's loop ( 1)
In T's loop ( 2)
In T's loop ( 3)
In T's loop ( 4)
Reset
In T's loop ( 1)
In T's loop ( 2)
In T's loop ( 3)

```

(continues on next page)

(continued from previous page)

In T's loop ( 4 )

In this example, the task body implements an infinite loop that accepts calls to the Reset and Increment entry. We make the following observations:

- The `accept E do ... end` block is used to increment a counter.
  - As long as task T is performing the `do ... end` block, the main task waits for the block to complete.
- The main task is calling the Increment entry multiple times in the loop from `1 ... 4`. It is also calling the Reset entry before the second loop.
  - Because task T contains an infinite loop, it always accepts calls to the Reset and Increment entries.
  - When the master construct of the subtask (the `Show_Rendezvous_Loop` subprogram) completes, it checks the status of the T task. Even though task T could accept new calls to the Reset or Increment entries, the master construct is allowed to terminate task T due to the `or terminate` part of the `select` statement.

### 15.1.6 Cycling tasks

In a previous example, we saw how to delay a task a specified time by using the `delay` keyword. However, using delay statements in a loop is not enough to guarantee regular intervals between those delay statements. For example, we may have a call to a computationally intensive procedure between executions of successive delay statements:

```
while True loop
  delay 1.0;
  --  ^ Wait 1.0 seconds
  Computational_Intensive_App;
end loop;
```

In this case, we can't guarantee that exactly 10 seconds have elapsed after 10 calls to the `delay` statement because a time drift may be introduced by the `Computational_Intensive_App` procedure. In many cases, this time drift is not relevant, so using the `delay` keyword is good enough.

However, there are situations where a time drift isn't acceptable. In those cases, we need to use the `delay until` statement, which accepts a precise time for the end of the delay, allowing us to define a regular interval. This is useful, for example, in real-time applications.

We will soon see an example of how this time drift may be introduced and how the `delay until` statement circumvents the problem. But before we do that, we look at a package containing a procedure allowing us to measure the elapsed time (`Show_Elapsed_Time`) and a dummy `Computational_Intensive_App` procedure which is simulated by using a simple delay. This is the complete package:

Listing 214: `delay_aux_pkg.ads`

```
1  with Ada.Real_Time; use Ada.Real_Time;
2
3  package Delay_Aux_Pkg is
4
5    function Get_Start_Time return Time
6      with Inline;
7
8    procedure Show_Elapsed_Time
9      with Inline;
10
```

(continues on next page)

(continued from previous page)

```

11  procedure Computational_Intensive_App;
12  private
13      Start_Time    : Time := Clock;
14
15  function Get_Start_Time return Time is
16      (Start_Time);
17
18 end Delay_Aux_Pkg;

```

Listing 215: delay\_aux\_pkg.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  package body Delay_Aux_Pkg is
4
5      procedure Show_Elapsed_Time is
6          Now_Time      : Time;
7          Elapsed_Time : Time_Span;
8      begin
9          Now_Time      := Clock;
10         Elapsed_Time := Now_Time - Start_Time;
11         Put_Line ("Elapsed time "
12                     & Duration'Image
13                         (To_Duration (Elapsed_Time))
14                         & " seconds");
15     end Show_Elapsed_Time;
16
17     procedure Computational_Intensive_App is
18     begin
19         delay 0.5;
20     end Computational_Intensive_App;
21
22 end Delay_Aux_Pkg;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Time  
MD5: 422a38c1afa0bbd659ec81de88479e0a

Using this auxiliary package, we're now ready to write our time-drifting application:

Listing 216: show\_time\_task.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2  with Ada.Real_Time; use Ada.Real_Time;
3
4  with Delay_Aux_Pkg;
5
6  procedure Show_Time_Task is
7      package Aux renames Delay_Aux_Pkg;
8
9      task T;
10
11      task body T is
12          Cnt   : Integer := 1;
13      begin
14          for I in 1 .. 5 loop
15              delay 1.0;
16
17              Aux.Show_Elapsed_Time;
18              Aux.Computational_Intensive_App;

```

(continues on next page)

(continued from previous page)

```
19      Put_Line ("Cycle # "
20          & Integer'Image (Cnt));
21      Cnt := Cnt + 1;
22  end loop;
23  Put_Line ("Finished time-drifting loop");
24 end T;
25
26 begin
27     null;
28 end Show_Time_Task;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Time  
MD5: fe17c902fc127c0132677ea4005ff3f1

### Runtime output

```
Elapsed time 1.000549316 seconds
Cycle # 1
Elapsed time 2.502470007 seconds
Cycle # 2
Elapsed time 4.003457739 seconds
Cycle # 3
Elapsed time 5.505746683 seconds
Cycle # 4
Elapsed time 7.006942816 seconds
Cycle # 5
Finished time-drifting loop
```

We can see by running the application that we already have a time difference of about four seconds after three iterations of the loop due to the drift introduced by `Computational_Intensive_App`. Using the `delay until` statement, however, we're able to avoid this time drift and have a regular interval of exactly one second:

Listing 217: `show_time_task.adb`

```
1 with Ada.Text_IO;  use Ada.Text_IO;
2 with Ada.Real_Time; use Ada.Real_Time;
3
4 with Delay_Aux_Pkg;
5
6 procedure Show_Time_Task is
7     package Aux renames Delay_Aux_Pkg;
8
9     task T;
10
11    task body T is
12        Cycle : constant Time_Span :=
13            Milliseconds (1000);
14        Next : Time := Aux.Get_Start_Time
15            + Cycle;
16
17        Cnt : Integer := 1;
18    begin
19        for I in 1 .. 5 loop
20            delay until Next;
21
22            Aux.Show_Elapsed_Time;
23            Aux.Computational_Intensive_App;
```

(continues on next page)

(continued from previous page)

```

24      -- Calculate next execution time
25      -- using a cycle of one second
26      Next := Next + Cycle;
27
28      Put_Line ("Cycle # "
29                  & Integer'Image (Cnt));
30      Cnt := Cnt + 1;
31  end loop;
32  Put_Line ("Finished cycling");
33
34 end T;
35
36 begin
37   null;
38 end Show_Time_Task;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Time  
MD5: 1456c0feee6def8b370d994c0ab75a15

### Runtime output

```

Elapsed time 1.001883847 seconds
Cycle # 1
Elapsed time 2.001327181 seconds
Cycle # 2
Elapsed time 3.000699340 seconds
Cycle # 3
Elapsed time 4.001981741 seconds
Cycle # 4
Elapsed time 5.001690838 seconds
Cycle # 5
Finished cycling

```

Now, as we can see by running the application, the `delay until` statement ensures that the `Computational_Intensive_App` doesn't disturb the regular interval of one second between iterations.

## 15.2 Protected objects

When multiple tasks are accessing shared data, corruption of that data may occur. For example, data may be inconsistent if one task overwrites parts of the information that's being read by another task at the same time. In order to avoid these kinds of problems and ensure information is accessed in a coordinated way, we use *protected objects*.

Protected objects encapsulate data and provide access to that data by means of *protected operations*, which may be subprograms or protected entries. Using protected objects ensures that data is not corrupted by race conditions or other concurrent access.

### Important

Objects can be protected from concurrent access using Ada tasks. In fact, this was the *only* way of protecting objects from concurrent access in Ada 83 (the first version of the Ada language). However, the use of protected objects is much simpler than using similar mechanisms implemented using only tasks. Therefore, you should use protected objects when your main goal is only to protect data.

### 15.2.1 Simple object

You declare a protected object with the **protected** keyword. The syntax is similar to that used for packages: you can declare operations (e.g., procedures and functions) in the public part and data in the private part. The corresponding implementation of the operations is included in the **protected body** of the object. For example:

Listing 218: show\_protected\_objects.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Protected_Objects is
4
5      protected Obj is
6          -- Operations go here (only subprograms)
7          procedure Set (V : Integer);
8          function Get return Integer;
9      private
10         -- Data goes here
11         Local : Integer := 0;
12     end Obj;
13
14     protected body Obj is
15         -- procedures can modify the data
16         procedure Set (V : Integer) is
17             begin
18                 Local := V;
19             end Set;
20
21         -- functions cannot modify the data
22         function Get return Integer is
23             begin
24                 return Local;
25             end Get;
26     end Obj;
27
28 begin
29     Obj.Set (5);
30     Put_Line ("Number is: "
31               & Integer'Image (Obj.Get));
32 end Show_Protected_Objects;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Protected\_Objects  
MD5: dd97dd584ba2f13def3c04725d4e48a7

#### Runtime output

Number is: 5

In this example, we define two operations for Obj: Set and Get. The implementation of these operations is in the Obj body. The syntax used for writing these operations is the same as that for normal procedures and functions. The implementation of protected objects is straightforward — we simply access and update Local in these subprograms. To call these operations in the main application, we use prefixed notation, e.g., Obj.Get.

## 15.2.2 Entries

In addition to protected procedures and functions, you can also define protected entry points. Do this using the `entry` keyword. Protected entry points allow you to define barriers using the `when` keyword. Barriers are conditions that must be fulfilled before the entry can start performing its actual processing — we speak of *releasing* the barrier when the condition is fulfilled.

The previous example used procedures and functions to define operations on the protected objects. However, doing so permits reading protected information (via `Obj.Get`) before it's set (via `Obj.Set`). To allow that to be a defined operation, we specified a default value (0). Instead, by rewriting `Obj.Get` using an `entry` instead of a function, we implement a barrier, ensuring no task can read the information before it's been set.

The following example implements the barrier for the `Obj.Get` operation. It also contains two concurrent subprograms (main task and task T) that try to access the protected object.

Listing 219: `show_protected_objects_entries.adb`

```

1  with Ada.Text_Io; use Ada.Text_Io;
2
3  procedure Show_Protected_Objects_Entries is
4
5      protected Obj is
6          procedure Set (V : Integer);
7          entry Get (V : out Integer);
8      private
9          Local : Integer;
10         Is_Set : Boolean := False;
11     end Obj;
12
13     protected body Obj is
14         procedure Set (V : Integer) is
15             begin
16                 Local := V;
17                 Is_Set := True;
18             end Set;
19
20         entry Get (V : out Integer)
21             when Is_Set is
22             -- Entry is blocked until the
23             -- condition is true. The barrier
24             -- is evaluated at call of entries
25             -- and at exits of procedures and
26             -- entries. The calling task sleeps
27             -- until the barrier is released.
28             begin
29                 V := Local;
30                 Is_Set := False;
31             end Get;
32     end Obj;
33
34     N : Integer := 0;
35
36     task T;
37
38     task body T is
39         begin
40             Put_Line
41                 ("Task T will delay for 4 seconds...");
42             delay 4.0;
43
44             Put_Line

```

(continues on next page)

(continued from previous page)

```
45      ("Task T will set Obj...");  
46      Obj.Set (5);  
47  
48      Put_Line  
49      ("Task T has just set Obj...");  
50  end T;  
51 begin  
52      Put_Line  
53      ("Main application will get Obj...");  
54      Obj.Get (N);  
55  
56      Put_Line  
57      ("Main application has retrieved Obj...");  
58      Put_Line  
59      ("Number is: " & Integer'Image (N));  
60  
61 end Show_Protected_Objects_Entries;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Protected\_Objects\_Entries  
MD5: c1134445a96700b871fb76c4d6342359

### Runtime output

```
Task T will delay for 4 seconds...  
Main application will get Obj...  
Task T will set Obj...  
Task T has just set Obj...  
Main application has retrieved Obj...  
Number is: 5
```

As we see by running it, the main application waits until the protected object is set (by the call to `Obj.Set` in task T) before it reads the information (via `Obj.Get`). Because a 4-second delay has been added in task T, the main application is also delayed by 4 seconds. Only after this delay does task T set the object and release the barrier in `Obj.Get` so that the main application can then resume processing (after the information is retrieved from the protected object).

## 15.3 Task and protected types

In the previous examples, we defined single tasks and protected objects. We can, however, generalize tasks and protected objects using type definitions. This allows us, for example, to create multiple tasks based on just a single task type.

### 15.3.1 Task types

A task type is a generalization of a task. The declaration is similar to simple tasks: you replace `task` with `task type`. The difference between simple tasks and task types is that task types don't create actual tasks that automatically start. Instead, a task object declaration is needed. This is exactly the way normal variables and types work: objects are only created by variable definitions, not type definitions.

To illustrate this, we repeat our first example:

Listing 220: show\_simple\_task.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Simple_Task is
4      task T;
5
6      task body T is
7          begin
8              Put_Line ("In task T");
9          end T;
10
11 begin
12     Put_Line ("In main");
13 end Show_Simple_Task;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Simple\_Task  
MD5: b17d9b35b4b2b53bc59776749e1be219

### Runtime output

In task T  
In main

We now rewrite it by replacing `task T` with `task type TT`. We declare a task (A\_Task) based on the task type TT after its definition:

Listing 221: show\_simple\_task\_type.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Simple_Task_Type is
4      task type TT;
5
6      task body TT is
7          begin
8              Put_Line ("In task type TT");
9          end TT;
10
11     A_Task : TT;
12
13 begin
14     Put_Line ("In main");
15 end Show_Simple_Task_Type;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Simple\_Task\_Type  
MD5: 24c26dcba6f5c54f0a7d47c3c0da728

### Runtime output

In task type TT  
In main

We can extend this example and create an array of tasks. Since we're using the same syntax as for variable declarations, we use a similar syntax for task types: `array (<>) of Task_Type`. Also, we can pass information to the individual tasks by defining a Start entry. Here's the updated example:

Listing 222: show\_task\_type\_array.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Task_Type_Array is
4      task type TT is
5          entry Start (N : Integer);
6      end TT;
7
8      task body TT is
9          Task_N : Integer;
10     begin
11         accept Start (N : Integer) do
12             Task_N := N;
13         end Start;
14         Put_Line ("In task T: "
15                   & Integer'Image (Task_N));
16     end TT;
17
18     My_Tasks : array (1 .. 5) of TT;
19 begin
20     Put_Line ("In main");
21
22     for I in My_Tasks'Range loop
23         My_Tasks (I).Start (I);
24     end loop;
25 end Show_Task_Type_Array;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Task\_Type\_Array  
MD5: bba072dfc52fb2bfbef6e7b9f8191464

### Runtime output

```
In main
In task T: 1
In task T: 2
In task T: 3
In task T: 4
In task T: 5
```

In this example, we're declaring five tasks in the array `My_Tasks`. We pass the array index to the individual tasks in the entry point (`Start`). After the synchronization between the individual subtasks and the main task, each subtask calls `Put_Line` concurrently.

### 15.3.2 Protected types

A protected type is a generalization of a protected object. The declaration is similar to that for protected objects: you replace `protected` with `protected type`. Like task types, protected types require an object declaration to create actual objects. Again, this is similar to variable declarations and allows for creating arrays (or other composite objects) of protected objects.

We can reuse a previous example and rewrite it to use a protected type:

Listing 223: show\_protected\_object\_type.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Protected_Object_Type is
```

(continues on next page)

(continued from previous page)

```

4  protected type P_Obj_Type is
5      procedure Set (V : Integer);
6      function Get return Integer;
7  private
8      Local : Integer := 0;
9  end P_Obj_Type;
10
11 protected body P_Obj_Type is
12     procedure Set (V : Integer) is
13     begin
14         Local := V;
15     end Set;
16
17     function Get return Integer is
18     begin
19         return Local;
20     end Get;
21  end P_Obj_Type;
22
23  Obj : P_Obj_Type;
24
25 begin
26     Obj.Set (5);
27     Put_Line ("Number is: "
28             & Integer'Image (Obj.Get));
29 end Show_Protected_Object_Type;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Tasking.Show\_Protected\_Object\_Type  
MD5: c50321e55afef0d72f263fee0669e55f

### Runtime output

Number is: 5

In this example, instead of directly defining the protected object Obj, we first define a protected type P\_Obj\_Type and then declare Obj as an object of that protected type. Note that the main application hasn't changed: we still use Obj.Set and Obj.Get to access the protected object, just like in the original example.



## DESIGN BY CONTRACTS

Contracts are used in programming to codify expectations. Parameter modes of a subprogram can be viewed as a simple form of contracts. When the specification of subprogram  $0p$  declares a parameter using `in` mode, the caller of  $0p$  knows that the `in` argument won't be changed by  $0p$ . In other words, the caller expects that  $0p$  doesn't modify the argument it's providing, but just reads the information stored in the argument. Constraints and subtypes are other examples of contracts. In general, these specifications improve the consistency of the application.

*Design-by-contract* programming refers to techniques that include pre- and postconditions, subtype predicates, and type invariants. We study those topics in this chapter.

### 16.1 Pre- and postconditions

Pre- and postconditions provide expectations regarding input and output parameters of subprograms and return value of functions. If we say that certain requirements must be met before calling a subprogram  $0p$ , those are preconditions. Similarly, if certain requirements must be met after a call to the subprogram  $0p$ , those are postconditions. We can think of preconditions and postconditions as promises between the subprogram caller and the callee: a precondition is a promise from the caller to the callee, and a postcondition is a promise in the other direction.

Pre- and postconditions are specified using an aspect clause in the subprogram declaration. A `with Pre => <condition>` clause specifies a precondition and a `with Post => <condition>` clause specifies a postcondition.

The following code shows an example of preconditions:

Listing 224: show\_simple\_precondition.adb

```
1  procedure Show_Simple_Precondition is
2
3      procedure DB_Entry (Name : String;
4                          Age  : Natural)
5          with Pre => Name'Length > 0
6      is
7          begin
8              -- Missing implementation
9              null;
10             end DB_Entry;
11            begin
12                 DB_Entry ("John", 30);
13
14                 -- Precondition will fail!
15                 DB_Entry ("", 21);
16            end Show_Simple_Precondition;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Contracts.Show\_Simple\_Precondition  
MD5: 87b6e080555603111801a0fc2469acd

### Runtime output

```
raised ADA ASSERTIONS ASSERTION_ERROR : failed precondition from show_simple_
precondition.adb:5
```

In this example, we want to prevent the name field in our database from containing an empty string. We implement this requirement by using a precondition requiring that the length of the string used for the Name parameter of the DB\_Entry procedure is greater than zero. If the DB\_Entry procedure is called with an empty string for the Name parameter, the call will fail because the precondition is not met.

#### In the GNAT toolchain

GNAT handles pre- and postconditions by generating runtime assertions for them. By default, however, assertions aren't enabled. Therefore, in order to check pre- and post-conditions at runtime, you need to enable assertions by using the `-gnata` switch.

Before we get to our next example, let's briefly discuss quantified expressions, which are quite useful in concisely writing pre- and postconditions. Quantified expressions return a Boolean value indicating whether elements of an array or container match the expected condition. They have the form: `(for all I in A'Range => <condition on A(I)>)`, where A is an array and I is an index. Quantified expressions using `for all` check whether the condition is true for every element. For example:

```
(for all I in A'Range => A (I) = 0)
```

This quantified expression is only true when all elements of the array A have a value of zero.

Another kind of quantified expressions uses `for some`. The form looks similar: `(for some I in A'Range => <condition on A(I)>)`. However, in this case the qualified expression tests whether the condition is true only on *some* elements (hence the name) instead of all elements.

We illustrate postconditions using the following example:

Listing 225: show\_simple\_postcondition.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Show_Simple_Postcondition is
4
5   type Int_8 is range -2 ** 7 .. 2 ** 7 - 1;
6
7   type Int_8_Array is
8     array (Integer range <>) of Int_8;
9
10  function Square (A : Int_8) return Int_8 is
11    (A * A)
12    with Post => (if abs A in 0 | 1
13                  then Square'Result = abs A
14                  else Square'Result > A);
15
16  procedure Square (A : in out Int_8_Array)
17    with Post => (for all I in A'Range =>
18                  A (I) = A'Old (I) *
19                  A'Old (I))
```

(continues on next page)

(continued from previous page)

```

20  is
21  begin
22    for V of A loop
23      V := Square (V);
24    end loop;
25  end Square;

26
27  V : Int_8_Array := (-2, -1, 0, 1, 10, 11);
28 begin
29  for E of V loop
30    Put_Line ("Original: "
31              & Int_8'Image (E));
32  end loop;
33  New_Line;

34
35  Square (V);
36  for E of V loop
37    Put_Line ("Square:    "
38              & Int_8'Image (E));
39  end loop;
40 end Show_Simple_Postcondition;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Contracts>Show\_Simple\_Postcondition  
MD5: b9bae9fe09cefcb6769ad9cd6739e2a

### Runtime output

```

Original: -2
Original: -1
Original: 0
Original: 1
Original: 10
Original: 11

Square:    4
Square:    1
Square:    0
Square:    1
Square: 100
Square: 121

```

We declare a signed 8-bit type `Int_8` and an array of that type (`Int_8_Array`). We want to ensure each element of the array is squared after calling the procedure `Square` for an object of the `Int_8_Array` type. We do this with a postcondition using a `for all` expression. This postcondition also uses the `'Old` attribute to refer to the original value of the parameter (before the call).

We also want to ensure that the result of calls to the `Square` function for the `Int_8` type are greater than the input to that call. To do that, we write a postcondition using the `'Result` attribute of the function and comparing it to the input value.

We can use both pre- and postconditions in the declaration of a single subprogram. For example:

Listing 226: `show_simple_contract.adb`

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Simple_Contract is

```

(continues on next page)

(continued from previous page)

```

4  type Int_8 is range -2 ** 7 .. 2 ** 7 - 1;
5
6  function Square (A : Int_8) return Int_8 is
7      (A * A)
8      with
9          Pre  => (Integer'Size >= Int_8'Size * 2
10             and Integer (A) *
11                 Integer (A) <=
12                     Integer (Int_8'Last)),
13             Post => (if abs A in 0 | 1
14                     then Square'Result = abs A
15                     else Square'Result > A);
16
17
18  V : Int_8;
19 begin
20     V := Square (11);
21     Put_Line ("Square of 11 is "
22               & Int_8'Image (V));
23
24     -- Precondition will fail...
25     V := Square (12);
26     Put_Line ("Square of 12 is "
27               & Int_8'Image (V));
28 end Show_Simple_Contract;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Contracts.Show\_Simple\_Contract  
MD5: 1d928dd100704907c858562155f90ee2

### Runtime output

```

Square of 11 is 121

raised ADA ASSERTIONS ASSERTION_ERROR : failed precondition from show_simple_
contract.adb:10

```

In this example, we want to ensure that the input value of calls to the `Square` function for the `Int_8` type won't cause overflow in that function. We do this by converting the input value to the `Integer` type, which is used for the temporary calculation, and check if the result is in the appropriate range for the `Int_8` type. We have the same postcondition in this example as in the previous one.

## 16.2 Predicates

Predicates specify expectations regarding types. They're similar to pre- and postconditions, but apply to types instead of subprograms. Their conditions are checked for each object of a given type, which allows verifying that an object of type `T` is conformant to the requirements of its type.

There are two kinds of predicates: static and dynamic. In simple terms, static predicates are used to check objects at compile-time, while dynamic predicates are used for checks at run time. Normally, static predicates are used for scalar types and dynamic predicates for the more complex types.

Static and dynamic predicates are specified using the following clauses, respectively:

- `with Static_Predicate => <property>`
- `with Dynamic_Predicate => <property>`

Let's use the following example to illustrate dynamic predicates:

Listing 227: show\_dynamic\_predicate\_courses.adb

```

1  with Ada.Calendar; use Ada.Calendar;
2
3  with Ada.Containers.Vectors;
4
5  with Ada.Strings.Unbounded;
6  use Ada.Strings.Unbounded;
7
8  procedure Show_Dynamic_Predicate_Courses is
9
10 package Courses is
11   type Course_Container is private;
12
13   type Course is record
14     Name      : Unbounded_String;
15     Start_Date : Time;
16     End_Date   : Time;
17   end record;
18   with Dynamic_Predicate =>
19     Course.Start_Date <= Course.End_Date;
20
21   procedure Add (CC : in out Course_Container;
22                 C  :          Course);
23
24 private
25   package Course_Vectors is new
26     Ada.Containers.Vectors
27       (Index_Type  => Natural,
28        Element_Type => Course);
29
30   type Course_Container is record
31     V : Course_Vectors.Vector;
32   end record;
33 end Courses;
34
35 package body Courses is
36   procedure Add (CC : in out Course_Container;
37                 C  :          Course) is
38     begin
39       CC.V.Append (C);
40     end Add;
41   end Courses;
42
43 use Courses;
44
45 CC : Course_Container;
46 begin
47   Add (CC,
48         Course'(
49           Name      =>
50             To_Unbounded_String
51             ("Intro to Photography"),
52           Start_Date =>
53             Time_Of (2018, 5, 1),
54           End_Date   =>
55             Time_Of (2018, 5, 10)));
56
57   -- This should trigger an error in the
58   -- dynamic predicate check
59   Add (CC,
60         Course'(

```

(continues on next page)

(continued from previous page)

```
60  Name      =>
61    To_Unbounded_String
62      ("Intro to Video Recording"),
63  Start_Date =>
64    Time_Of (2019, 5, 1),
65  End_Date  =>
66    Time_Of (2018, 5, 10));
67
68 end Show_Dynamic_Predicate_Courses;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Contracts.Show\_Dynamic\_Predicate\_Courses  
MD5: 8bd6539e72995fececfcdf9666ffd04f

### Runtime output

```
raised ADA ASSERTIONS ASSERTION_ERROR : Dynamic_Predicate failed at show_dynamic_
→predicate_courses.adb:59
```

In this example, the package Courses defines a type Course and a type Course\_Container, an object of which contains all courses. We want to ensure that the dates of each course are consistent, specifically that the start date is no later than the end date. To enforce this rule, we declare a dynamic predicate for the Course type that performs the check for each object. The predicate uses the type name where a variable of that type would normally be used: this is a reference to the instance of the object being tested.

Note that the example above makes use of unbounded strings and dates. Both types are available in Ada's standard library. Please refer to the following sections for more information about:

- the unbounded string type (Unbounded\_String): [Unbounded Strings](#) (page 245) section;
- dates and times: [Dates & Times](#) (page 229) section.

Static predicates, as mentioned above, are mostly used for scalar types and checked during compilation. They're particularly useful for representing non-contiguous elements of an enumeration. A classic example is a list of week days:

```
type Week is (Mon, Tue, Wed, Thu, Fri, Sat, Sun);
```

We can easily create a sub-list of work days in the week by specifying a **subtype** with a range based on Week. For example:

```
subtype Work_Week is Week range Mon .. Fri;
```

Ranges in Ada can only be specified as contiguous lists: they don't allow us to pick specific days. However, we may want to create a list containing just the first, middle and last day of the work week. To do that, we use a static predicate:

```
subtype Check_Days is Work_Week
  with Static_Predicate =>
    Check_Days in Mon | Wed | Fri;
```

Let's look at a complete example:

Listing 228: show\_predicates.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Predicates is
4
5      type Week is (Mon, Tue, Wed, Thu,
6                     Fri, Sat, Sun);
7
8      subtype Work_Week is Week range Mon .. Fri;
9
10     subtype Test_Days is Work_Week
11         with Static_Predicate =>
12             Test_Days in Mon | Wed | Fri;
13
14     type Tests_Week is array (Week) of Natural
15         with Dynamic_Predicate =>
16             (for all I in Tests_Week'Range =>
17                 (case I is
18                     when Test_Days =>
19                         Tests_Week (I) > 0,
20                     when others      =>
21                         Tests_Week (I) = 0));
22
23     Num_Tests : Tests_Week := (
24         Mon => 3, Tue => 0,
25         Wed => 4, Thu => 0,
26         Fri => 2, Sat => 0,
27         Sun => 0);
28
29     procedure Display_Tests (N : Tests_Week) is
30 begin
31     for I in Test_Days loop
32         Put_Line ("# tests on "
33                     & Test_Days'Image (I)
34                     & " => "
35                     & Integer'Image (N (I)));
36     end loop;
37 end Display_Tests;
38
39 begin
40     Display_Tests (Num_Tests);
41
42     -- Assigning non-conformant values to
43     -- individual elements of the Tests_Week
44     -- type does not trigger a predicate
45     -- check:
46     Num_Tests (Tue) := 2;
47
48     -- However, assignments with the "complete"
49     -- Tests_Week type trigger a predicate
50     -- check. For example:
51
52     -- Num_Tests := (others => 0);
53
54     -- Also, calling any subprogram with
55     -- parameters of Tests_Week type
56     -- triggers a predicate check. Therefore,
57     -- the following line will fail:
58     Display_Tests (Num_Tests);
59 end Show_Predicates;

```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.Contracts.Show\_Predicates  
MD5: 126c47033fc67fc8b6d7f6479205e752

### Runtime output

```
# tests on MON => 3
# tests on WED => 4
# tests on FRI => 2

raised ADA ASSERTIONS ASSERTION_ERROR : Dynamic_Predicate failed at show_
→predicates.adb:58
```

Here we have an application that wants to perform tests only on three days of the work week. These days are specified in the `Test_Days` subtype. We want to track the number of tests that occur each day. We declare the type `Tests_Week` as an array, an object of which will contain the number of tests done each day. According to our requirements, these tests should happen only in the aforementioned three days; on other days, no tests should be performed. This requirement is implemented with a dynamic predicate of the type `Tests_Week`. Finally, the actual information about these tests is stored in the array `Num_Tests`, which is an instance of the `Tests_Week` type.

The dynamic predicate of the `Tests_Week` type is verified during the initialization of `Num_Tests`. If we have a non-conformant value there, the check will fail. However, as we can see in our example, individual assignments to elements of the array do not trigger a check. We can't check for consistency at this point because the initialization of the a complex data structure (such as arrays or records) may not be performed with a single assignment. However, as soon as the object is passed as an argument to a subprogram, the dynamic predicate is checked because the subprogram requires the object to be consistent. This happens in the last call to `Display_Tests` in our example. Here, the predicate check fails because the previous assignment has a non-conformant value.

## 16.3 Type invariants

Type invariants are another way of specifying expectations regarding types. While predicates are used for *non-private* types, type invariants are used exclusively to define expectations about private types. If a type `T` from a package `P` has a type invariant, the results of operations on objects of type `T` are always consistent with that invariant.

Type invariants are specified with a `with Type_Invariant => <property>` clause. Like predicates, the *property* defines a condition that allows us to check if an object of type `T` is conformant to its requirements. In this sense, type invariants can be viewed as a sort of predicate for private types. However, there are some differences in terms of checks. The following table summarizes the differences:

Element	Subprogram parameter checks	Assignment checks
Predicates	On all <code>in</code> and <code>out</code> parameters	On assignments and explicit initializations
Type invariants	On <code>out</code> parameters returned from subprograms declared in the same public scope	On all initializations

We could rewrite our previous example and replace dynamic predicates by type invariants. It would look like this:

Listing 229: show\_type\_invariant.adb

```

1  with Ada.Text_Io;  use Ada.Text_Io;
2  with Ada.Calendar; use Ada.Calendar;
3
4  with Ada.Containers.Vectors;
5
6  with Ada.Strings.Unbounded;
7  use Ada.Strings.Unbounded;
8
9  procedure Show_Type_Invariant is
10
11    package Courses is
12      type Course is private
13      with Type_Invariant => Check (Course);
14
15      type Course_Container is private;
16
17      procedure Add (CC : in out Course_Container;
18                     C  :          Course);
19
20      function Init
21        (Name          : String;
22         Start_Date, End_Date : Time)
23        return Course;
24
25      function Check (C : Course)
26        return Boolean;
27
28  private
29    type Course is record
30      Name          : Unbounded_String;
31      Start_Date : Time;
32      End_Date   : Time;
33    end record;
34
35    function Check (C : Course)
36      return Boolean is
37      (C.Start_Date <= C.End_Date);
38
39    package Course_Vectors is new
40      Ada.Containers.Vectors
41      (Index_Type  => Natural,
42       Element_Type => Course);
43
44    type Course_Container is record
45      V : Course_Vectors.Vector;
46    end record;
47  end Courses;
48
49  package body Courses is
50    procedure Add (CC : in out Course_Container;
51                  C  :          Course) is
52    begin
53      CC.V.Append (C);
54    end Add;
55
56    function Init
57      (Name          : String;
58       Start_Date, End_Date : Time)
59      return Course is
60    begin
61      return

```

(continues on next page)

(continued from previous page)

```
62  Course' (Name      =>
63      To_Unbounded_String (Name),
64      Start_Date => Start_Date,
65      End_Date   => End_Date);
66  end Init;
67 end Courses;
68
69 use Courses;
70
71 CC : Course_Container;
72 begin
73  Add (CC,
74      Init (Name      =>
75          "Intro to Photography",
76          Start_Date =>
77          Time_Of (2018, 5, 1),
78          End_Date   =>
79          Time_Of (2018, 5, 10)));
80
81  -- This should trigger an error in the
82  -- type-invariant check
83  Add (CC,
84      Init (Name      =>
85          "Intro to Video Recording",
86          Start_Date =>
87          Time_Of (2019, 5, 1),
88          End_Date   =>
89          Time_Of (2018, 5, 10)));
90 end Show_Type_Invariant;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Contracts.Show\_Type\_Invariant  
MD5: c6ef863da94285f927dd106645af8650

### Runtime output

```
raised ADA ASSERTIONS ASSERTION_ERROR : failed invariant from show_type_invariant.
  ↵adb:13
```

The major difference is that the Course type was a visible (public) type of the Courses package in the previous example, but in this example is a private type.

## INTERFACING WITH C

Ada allows us to interface with code in many languages, including C and C++. This section discusses how to interface with C.

### 17.1 Multi-language project

By default, when using **gprbuild** we only compile Ada source files. To compile C files as well, we need to modify the project file used by **gprbuild**. We use the Languages entry, as in the following example:

```
project Multilang is
  for Languages use ("ada", "c");
  for Source_Dirs use ("src");
  for Main use ("main.adb");
  for Object_Dir use "obj";
end Multilang;
```

### 17.2 Type convention

To interface with data types declared in a C application, you specify the Convention aspect on the corresponding Ada type declaration. In the following example, we interface with the `C_Enum` enumeration declared in a C source file:

Listing 230: `show_c_enum.adb`

```
1  procedure Show_C_Enum is
2
3    type C_Enum is (A, B, C)
4      with Convention => C;
5      -- Use C convention for C_Enum
6  begin
7    null;
8  end Show_C_Enum;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Enum  
MD5: a14d7d981fd7d6d806cf3c55f35e19c8

To interface with C's built-in types, we use the `Interfaces.C` package, which contains most of the type definitions we need. For example:

Listing 231: show\_c\_struct.adb

```
1 with Interfaces.C; use Interfaces.C;
2
3 procedure Show_C_Struct is
4
5   type c_struct is record
6     a : int;
7     b : long;
8     c : unsigned;
9     d : double;
10  end record
11  with Convention => C;
12
13 begin
14   null;
15 end Show_C_Struct;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Struct  
MD5: dda4d3f8e4ddf5c5138a990a9a8ac427

Here, we're interfacing with a C struct (C\_Struct) and using the corresponding data types in C (**int**, **long**, **unsigned** and **double**). This is the declaration in C:

Listing 232: c\_struct.h

```
1 struct c_struct
2 {
3   int          a;
4   long         b;
5   unsigned     c;
6   double       d;
7 };
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Struct  
MD5: 58709b6a9eea2606d7ec0aaca0a749ff

## 17.3 Foreign subprograms

### 17.3.1 Calling C subprograms in Ada

We use a similar approach when interfacing with subprograms written in C. Consider the following declaration in the C header file:

Listing 233: my\_func.h

```
1 int my_func (int a);
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Func  
MD5: 37b9d7ba668f7ec83c2b27ee33637937

Here's the corresponding C definition:

Listing 234: my\_func.c

```

1 #include "my_func.h"
2
3 int my_func (int a)
4 {
5     return a * 2;
6 }
```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Func  
MD5: 284b1639cb393fc14ed196d78429f3ba

We can interface this code in Ada using the Import aspect. For example:

Listing 235: show\_c\_func.adb

```

1 with Interfaces.C; use Interfaces.C;
2 with Ada.Text_Io;  use Ada.Text_Io;
3
4 procedure Show_C_Func is
5
6     function my_func (a : int) return int
7     with
8         Import      => True,
9         Convention => C;
10
11    -- Imports function 'my_func' from C.
12    -- You can now call it from Ada.
13
14    V : int;
15 begin
16    V := my_func (2);
17    Put_Line ("Result is " & int'Image (V));
18 end Show_C_Func;
```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Func  
MD5: 6c5d85c1debdeaa642946eacf413dfd2

If you want, you can use a different subprogram name in the Ada code. For example, we could call the C function Get\_Value:

Listing 236: show\_c\_func.adb

```

1 with Interfaces.C; use Interfaces.C;
2 with Ada.Text_Io;  use Ada.Text_Io;
3
4 procedure Show_C_Func is
5
6     function Get_Value (a : int) return int
7     with
8         Import      => True,
9         Convention => C,
10        External_Name => "my_func";
11
12    -- Imports function 'my_func' from C and
13    -- renames it to 'Get_Value'
```

(continues on next page)

(continued from previous page)

```
15  V : int;
16  begin
17    V := Get_Value (2);
18    Put_Line ("Result is " & int'Image (V));
19  end Show_C_Func;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Func  
MD5: 856b4d99dfa6946fb4597f254fd2f97

## 17.3.2 Calling Ada subprograms in C

You can also call Ada subprograms from C applications. You do this with the Export aspect. For example:

Listing 237: c\_api.ads

```
1  with Interfaces.C; use Interfaces.C;
2
3  package C_API is
4
5    function My_Func (a : int) return int
6      with
7        Export      => True,
8        Convention  => C,
9        External_Name => "my_func";
10
11 end C_API;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Ada\_Func  
MD5: 00aa4ec29fc551e710900e2ee7d96bc9

This is the corresponding body that implements that function:

Listing 238: c\_api.adb

```
1  package body C_API is
2
3    function My_Func (a : int) return int is
4      begin
5        return a * 2;
6      end My_Func;
7
8  end C_API;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Ada\_Func  
MD5: 2b999ab431bbc1ee223a654ad84b8248

On the C side, we do the same as we would if the function were written in C: simply declare it using the **extern** keyword. For example:

Listing 239: main.c

```
1  #include <stdio.h>
```

(continues on next page)

(continued from previous page)

```

3  extern int my_func (int a);
4
5  int main (int argc, char **argv) {
6
7      int v = my_func(2);
8
9      printf("Result is %d\n", v);
10
11     return 0;
12 }
```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Ada\_Func  
 MD5: 69301036be9be16ed45895c2a86bc352

## 17.4 Foreign variables

### 17.4.1 Using C global variables in Ada

To use global variables from C code, we use the same method as subprograms: we specify the Import and Convention aspects for each variable we want to import.

Let's reuse an example from the previous section. We'll add a global variable (func\_cnt) to count the number of times the function (my\_func) is called:

Listing 240: test.h

```

1  extern int func_cnt;
2
3  int my_func (int a);
```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Vars  
 MD5: 11ba8f7a72ce7058571994870a02b052

The variable is declared in the C file and incremented in my\_func:

Listing 241: test.c

```

1  #include "test.h"
2
3  int func_cnt = 0;
4
5  int my_func (int a)
6  {
7      func_cnt++;
8
9      return a * 2;
10 }
```

**Code block metadata**

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Vars  
 MD5: 23631537cb877a03d1243c94cb7b48e8

In the Ada application, we just reference the foreign variable:

Listing 242: show\_c\_func.adb

```
1 with Interfaces.C; use Interfaces.C;
2 with Ada.Text_Io;  use Ada.Text_Io;
3
4 procedure Show_C_Func is
5
6   function my_func (a : int) return int
7   with
8     Import      => True,
9     Convention  => C;
10
11   V : int;
12
13   func_cnt : int
14   with
15     Import      => True,
16     Convention  => C;
17   -- We can access the func_cnt variable
18   -- from test.c
19
20 begin
21   V := my_func (1);
22   V := my_func (2);
23   V := my_func (3);
24
25   Put_Line ("Result is "
26             & int'Image (V));
27
28   Put_Line ("Function was called "
29             & int'Image (func_cnt)
30             & " times");
31 end Show_C_Func;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.Ada\_C\_Vars  
MD5: cf64a9dfbc6be853ba19729fe55f0ba4

As we see by running the application, the value of the counter is the number of times `my_func` was called.

We can use the `External_Name` aspect to give a different name for the variable in the Ada application in the same way we do for subprograms.

### 17.4.2 Using Ada variables in C

You can also use variables declared in Ada files in C applications. In the same way as we did for subprograms, you do this with the `Export` aspect.

Let's reuse a past example and add a counter, as in the previous example, but this time have the counter incremented in Ada code:

Listing 243: c\_api.ads

```
1 with Interfaces.C; use Interfaces.C;
2
3 package C_API is
4
5   func_cnt : int := 0
6   with
```

(continues on next page)

(continued from previous page)

```

7   Export      => True,
8   Convention => C;
9
10  function My_Func (a : int) return int
11    with
12      Export      => True,
13      Convention => C,
14      External_Name => "my_func";
15
16 end C_API;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Ada\_Vars  
MD5: fc118cddd797b669d2c68e57f90f69b2

The variable is then incremented in My\_Func:

Listing 244: c\_api.adb

```

1 package body C_API is
2
3   function My_Func (a : int) return int is
4     begin
5       func_cnt := func_cnt + 1;
6       return a * 2;
7     end My_Func;
8
9 end C_API;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Ada\_Vars  
MD5: adff5f3088da8b0dd853f1fb8b1e204f

In the C application, we just need to declare the variable and use it:

Listing 245: main.c

```

1 #include <stdio.h>
2
3 extern int my_func (int a);
4
5 extern int func_cnt;
6
7 int main (int argc, char **argv) {
8
9   int v;
10
11   v = my_func(1);
12   v = my_func(2);
13   v = my_func(3);
14
15   printf("Result is %d\n", v);
16
17   printf("Function was called %d times\n",
18         func_cnt);
19
20   return 0;
21 }

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Ada\_Vars  
MD5: 07fb3fbadb8ed4c0543fbfd7b5ef5c57

Again, by running the application, we see that the value from the counter is the number of times that `my_func` was called.

## 17.5 Generating bindings

In the examples above, we manually added aspects to our Ada code to correspond to the C source-code we're interfacing with. This is called creating a *binding*. We can automate this process by using the `Ada spec dump` compiler option: `-fdump-ada-spec`. We illustrate this by revisiting our previous example.

This was our C header file:

Listing 246: test.h

```
1 extern int func_cnt;  
2  
3 int my_func (int a);
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds  
MD5: 11ba8f7a72ce7058571994870a02b052

To create Ada bindings, we'll call the compiler like this:

```
gcc -c -fdump-ada-spec -C ./test.h
```

The result is an Ada spec file called `test_h.ads`:

Listing 247: test\_h.ads

```
1 pragma Ada_2005;  
2 pragma Style_Checks (Off);  
3  
4 with Interfaces.C; use Interfaces.C;  
5  
6 package test_h is  
7  
8     func_cnt : aliased int; -- ./test.h:3  
9     pragma Import (C, func_cnt, "func_cnt");  
10  
11    function my_func (arg1 : int) return int; -- ./test.h:5  
12    pragma Import (C, my_func, "my_func");  
13  
14 end test_h;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds  
MD5: 8d18aeae72dba3a9ab4f9f3943fab839

Now we simply refer to this `test_h` package in our Ada application:

Listing 248: show\_c\_func.adb

```
1 with Interfaces.C; use Interfaces.C;  
2 with Ada.Text_IO; use Ada.Text_IO;
```

(continues on next page)

(continued from previous page)

```

3  with test_h;           use test_h;
4
5  procedure Show_C_Func is
6    V : int;
7  begin
8    V := my_func (1);
9    V := my_func (2);
10   V := my_func (3);
11
12   Put_Line ("Result is "
13             & int'Image (V));
14
15   Put_Line ("Function was called "
16             & int'Image (func_cnt)
17             & " times");
18 end Show_C_Func;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds  
MD5: 8a07aae87b9f36c3fce84b75e8388933

You can specify the name of the parent unit for the bindings you're creating as the operand to `fdump-ada-spec`:

```
gcc -c -fdump-ada-spec -fada-spec-parent=Ext_C_Code -C ./test.h
```

This creates the file `ext_c_code-test_h.ads`:

Listing 249: `ext_c_code-test_h.ads`

```

1 package Ext_C_Code.test_h is
2
3   -- automatic generated bindings...
4
5 end Ext_C_Code.test_h;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds\_2  
MD5: 3bd4087edff145a70d2a6db8543859ad

## 17.5.1 Adapting bindings

The compiler does the best it can when creating bindings for a C header file. However, sometimes it has to guess about the translation and the generated bindings don't always match our expectations. For example, this can happen when creating bindings for functions that have pointers as arguments. In this case, the compiler may use `System.Address` as the type of one or more pointers. Although this approach works fine (as we'll see later), this is usually not how a human would interpret the C header file. The following example illustrates this issue.

Let's start with this C header file:

Listing 250: `test.h`

```

1 struct test;
2
3 struct test * test_create(void);
4

```

(continues on next page)

(continued from previous page)

```
5 void test_destroy(struct test *t);
6
7 void test_reset(struct test *t);
8
9 void test_set_name(struct test *t,
10                   char      *name);
11
12 void test_set_address(struct test *t,
13                      char      *address);
14
15 void test_display(const struct test *t);
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds\_3  
MD5: af642d9ea995bf01f13f8ff41bb0f4f6

And the corresponding C implementation:

Listing 251: test.c

```
1 #include <stdlib.h>
2 #include <string.h>
3 #include <stdio.h>
4
5 #include "test.h"
6
7 struct test {
8     char name[80];
9     char address[120];
10};
11
12 static size_t
13 strlcpy_stat(char      *dst,
14               const char *src,
15               size_t      dstsize)
16 {
17     size_t len = strlen(src);
18     if (dstsize) {
19         size_t bl = (len < dstsize-1) ?
20                     len : dstsize-1;
21         ((char*)memcpy(dst, src, bl))[bl] = 0;
22     }
23     return len;
24 }
25
26 struct test * test_create(void)
27 {
28     return malloc (sizeof (struct test));
29 }
30
31 void test_destroy(struct test *t)
32 {
33     if (t != NULL) {
34         free(t);
35     }
36 }
37
38 void test_reset(struct test *t)
39 {
40     t->name[0]      = '\0';
```

(continues on next page)

(continued from previous page)

```

41     t->address[0] = '\0';
42 }
43
44 void test_set_name(struct test *t,
45                      char           *name)
46 {
47     strlcpy_stat(t->name,
48                  name,
49                  sizeof(t->name));
50 }
51
52 void test_set_address(struct test *t,
53                      char           *address)
54 {
55     strlcpy_stat(t->address,
56                  address,
57                  sizeof(t->address));
58 }
59
60 void test_display(const struct test *t)
61 {
62     printf("Name: %s\n", t->name);
63     printf("Address: %s\n", t->address);
64 }
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds\_3  
MD5: 32652eb76ad92212609680d64e5687d3

Next, we'll create our bindings:

```
gcc -c -fdump-ada-spec -C ./test.h
```

This creates the following specification in test\_h.ads:

Listing 252: test\_h.ads

```

1  pragma Ada_2005;
2  pragma Style_Checks (Off);
3
4  with Interfaces.C; use Interfaces.C;
5  with System;
6  with Interfaces.C.Strings;
7
8  package test_h is
9
10   -- skipped empty struct test
11
12   function test_create return System.Address; -- ./test.h:5
13   pragma Import (C, test_create, "test_create");
14
15   procedure test_destroy (arg1 : System.Address); -- ./test.h:7
16   pragma Import (C, test_destroy, "test_destroy");
17
18   procedure test_reset (arg1 : System.Address); -- ./test.h:9
19   pragma Import (C, test_reset, "test_reset");
20
21   procedure test_set_name (arg1 : System.Address; arg2 : Interfaces.C.Strings.
22   ↴chars_ptr); -- ./test.h:11
23   pragma Import (C, test_set_name, "test_set_name");
```

(continues on next page)

(continued from previous page)

```
23
24  procedure test_set_address (arg1 : System.Address; arg2 : Interfaces.C.Strings.
25  →chars_ptr);  -- ./test.h:13
26  pragma Import (C, test_set_address, "test_set_address");
27
28  procedure test_display (arg1 : System.Address);  -- ./test.h:15
29  pragma Import (C, test_display, "test_display");
30
end test_h;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds\_3  
MD5: 3bf8f01b94fd28594e4121a6a36afdf7

As we can see, the binding generator completely ignores the declaration **struct test** and all references to the test struct are replaced by addresses (System.Address). Nevertheless, these bindings are good enough to allow us to create a test application in Ada:

Listing 253: show\_automatic\_c\_struct\_bindings.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  with Interfaces.C;
4  use Interfaces.C;
5
6  with Interfaces.C.Strings;
7  use Interfaces.C.Strings;
8
9  with test_h; use test_h;
10
11 with System;
12
13 procedure Show_Automatic_C_Struct_Bindings is
14
15   Name    : constant chars_ptr :=
16     New_String ("John Doe");
17   Address : constant chars_ptr :=
18     New_String ("Small Town");
19
20   T : System.Address := test_create;
21
22 begin
23   test_reset (T);
24   test_set_name (T, Name);
25   test_set_address (T, Address);
26
27   test_display (T);
28   test_destroy (T);
29 end Show_Automatic_C_Struct_Bindings;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds\_3  
MD5: 99d64fb14d9c869d140dd2fb7d3888d7

We can successfully bind our C code with Ada using the automatically-generated bindings, but they aren't ideal. Instead, we would prefer Ada bindings that match our (human) interpretation of the C header file. This requires manual analysis of the header file. The good news is that we can use the automatically generated bindings as a starting point and adapt them to our needs. For example, we can:

1. Define a Test type based on System.Address and use it in all relevant functions.
2. Remove the test\_ prefix in all operations on the Test type.

This is the resulting specification:

Listing 254: adapted\_test\_h.ads

```

1  with System;
2
3  with Interfaces.C; use Interfaces.C;
4  with Interfaces.C.Strings;
5
6  package adapted_test_h is
7
8      type Test is new System.Address;
9
10     function Create return Test;
11     pragma Import (C, Create, "test_create");
12
13     procedure Destroy (T : Test);
14     pragma Import (C, Destroy, "test_destroy");
15
16     procedure Reset (T : Test);
17     pragma Import (C, Reset, "test_reset");
18
19     procedure Set_Name (T      : Test;
20                         Name : Interfaces.C.Strings.chars_ptr);  -- ./test.h:11
21     pragma Import (C, Set_Name, "test_set_name");
22
23     procedure Set_Address (T      : Test;
24                           Address : Interfaces.C.Strings.chars_ptr);
25     pragma Import (C, Set_Address, "test_set_address");
26
27     procedure Display (T : Test);  -- ./test.h:15
28     pragma Import (C, Display, "test_display");
29
30 end adapted_test_h;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds\_3  
MD5: 5cc875e1b01af839141e5e623f6c5b7a

And this is the corresponding Ada body:

Listing 255: show\_adapted\_c\_struct\_bindings.adb

```

1  with Interfaces.C;
2  use Interfaces.C;
3
4  with Interfaces.C.Strings;
5  use Interfaces.C.Strings;
6
7  with adapted_test_h; use adapted_test_h;
8
9  with System;
10
11  procedure Show_Adapted_C_Struct_Bindings is
12
13      Name      : constant chars_ptr := 
14          New_String ("John Doe");
15      Address : constant chars_ptr := 

```

(continues on next page)

(continued from previous page)

```
16  New_String ("Small Town");  
17  
18  T : Test := Create;  
19  
20 begin  
21  Reset (T);  
22  Set_Name (T, Name);  
23  Set_Address (T, Address);  
24  
25  Display (T);  
26  Destroy (T);  
27 end Show_Adapted_C_Struct_Bindings;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Interfacing\_With\_C.C\_Binds\_3  
MD5: 626d07b080fbbd2bf1d5f9140b64955c

Now we can use the Test type and its operations in a clean, readable way.

## OBJECT-ORIENTED PROGRAMMING

Object-oriented programming (OOP) is a large and ill-defined concept in programming languages and one that tends to encompass many different meanings because different languages often implement their own vision of it, with similarities and differences from the implementations in other languages.

However, one model mostly "won" the battle of what object-oriented means, if only by sheer popularity. It's the model used in the Java programming language, which is very similar to the one used by C++. Here are some defining characteristics:

- Type derivation and extension: Most object oriented languages allow the user to add fields to derived types.
- Subtyping: Objects of a type derived from a base type can, in some instances, be substituted for objects of the base type.
- Runtime polymorphism: Calling a subprogram, usually called a *method*, attached to an object type can dispatch at runtime depending on the exact type of the object.
- Encapsulation: Objects can hide some of their data.
- Extensibility: People from the "outside" of your package, or even your whole library, can derive from your object types and define their own behaviors.

Ada dates from before object-oriented programming was as popular as it is today. Some of the mechanisms and concepts from the above list were in the earliest version of Ada even before what we would call OOP was added:

- As we saw, encapsulation is not implemented at the type level in Ada, but instead at the package level.
- Subtyping can be implemented using, well, subtypes, which have a full and permissive static substitutability model. The substitution will fail at runtime if the dynamic constraints of the subtype are not fulfilled.
- Runtime polymorphism can be implemented using variant records.

However, this lists leaves out type extensions, if you don't consider variant records, and extensibility.

The 1995 revision of Ada added a feature filling the gaps, which allowed people to program following the object-oriented paradigm in an easier fashion. This feature is called *tagged types*.

### Note

It's possible to program in Ada without ever creating tagged types. If that's your preferred style of programming or you have no specific use for tagged types, feel free to not use them, as is the case for many features of Ada.

However, they can be the best way to express solutions to certain problems and they may be the best way to solve your problem. If that's the case, read on!

## 18.1 Derived types

Before presenting tagged types, we should discuss a topic we have brushed on, but not really covered, up to now:

You can create one or more new types from every type in Ada. Type derivation is built into the language.

Listing 256: newtypes.ads

```
1 package Newtypes is
2     type Point is record
3         X, Y : Integer;
4     end record;
5
6     type New_Point is new Point;
7 end Newtypes;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Newtypes  
MD5: 0d45096755b4bfb08ba8db19ecba3f57

Type derivation is useful to enforce strong typing because the type system treats the two types as incompatible.

But the benefits are not limited to that: you can inherit things from the type you derive from. You not only inherit the representation of the data, but you can also inherit behavior.

When you inherit a type you also inherit what are called *primitive operations*. A primitive operation (or just a *primitive*) is a subprogram attached to a type. Ada defines primitives as subprograms defined in the same scope as the type.

### Attention

A subprogram will only become a primitive of the type if:

1. The subprogram is declared in the same scope as the type and
2. The type and the subprogram are declared in a package

Listing 257: primitives.adb

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Primitives is
4     package Week is
5         type Days is (Monday, Tuesday, Wednesday,
6                        Thursday, Friday,
```

(continues on next page)

(continued from previous page)

```

7          Saturday, Sunday);
8
9      -- Print_Day is a primitive
10     -- of the type Days
11     procedure Print_Day (D : Days);
12 end Week;
13
14 package body Week is
15     procedure Print_Day (D : Days) is
16     begin
17         Put_Line (Days'Image (D));
18     end Print_Day;
19 end Week;
20
21 use Week;
22 type Weekend_Days is new
23     Days range Saturday .. Sunday;
24
25     -- A procedure Print_Day is automatically
26     -- inherited here. It is as if the procedure
27     --
28     -- procedure Print_Day (D : Weekend_Days);
29     --
30     -- has been declared with the same body
31
32 Sat : Weekend_Days := Saturday;
33 begin
34     Print_Day (Sat);
35 end Primitives;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Primitives  
 MD5: eb1b0eb66f03a4a17bd9686ec4e12e2e

### Runtime output

SATURDAY

This kind of inheritance can be very useful, and is not limited to record types (you can use it on discrete types, as in the example above), but it's only superficially similar to object-oriented inheritance:

- Records can't be extended using this mechanism alone. You also can't specify a new representation for the new type: it will **always** have the same representation as the base type.
- There's no facility for dynamic dispatch or polymorphism. Objects are of a fixed, static type.

There are other differences, but it's not useful to list them all here. Just remember that this is a kind of inheritance you can use if you only want to statically inherit behavior without duplicating code or using composition, but a kind you can't use if you want any dynamic features that are usually associated with OOP.

## 18.2 Tagged types

The 1995 revision of the Ada language introduced tagged types to fulfil the need for an unified solution that allows programming in an object-oriented style similar to the one described at the beginning of this chapter.

Tagged types are very similar to normal records except that some functionality is added:

- Types have a *tag*, stored inside each object, that identifies the *runtime type*<sup>19</sup> of that object.
- Primitives can dispatch. A primitive on a tagged type is what you would call a *method* in Java or C++. If you derive a base type and override a primitive of it, you can often call it on an object with the result that which primitive is called depends on the exact runtime type of the object.
- Subtyping rules are introduced allowing a tagged type derived from a base type to be statically compatible with the base type.

Let's see our first tagged type declarations:

Listing 258: p.ads

```
1 package P is
2     type My_Class is tagged null record;
3         -- Just like a regular record, but
4         -- with tagged qualifier
5
6         -- Methods are outside of the type
7         -- definition:
8
9     procedure Foo (Self : in out My_Class);
10        -- If you define a procedure taking a
11        -- My_Class argument in the same package,
12        -- it will be a method.
13
14        -- Here's how you derive a tagged type:
15
16     type Derived is new My_Class with record
17         A : Integer;
18             -- You can add fields in derived types.
19     end record;
20
21     overriding
22     procedure Foo (Self : in out Derived);
23         -- The "overriding" qualifier is optional,
24         -- but if it is present, it must be valid.
25 end P;
```

Listing 259: p.adb

```
1 with Ada.Text_Io; use Ada.Text_Io;
2
3 package body P is
4     procedure Foo (Self : in out My_Class) is
5     begin
6         Put_Line ("In My_Class.Foo");
7     end Foo;
8
9     procedure Foo (Self : in out Derived) is
10    begin
11        Put_Line ("In Derived.Foo, A = "
12                  & Integer'Image (Self.A));
13    end Foo;
14 end P;
```

### Code block metadata

<sup>19</sup> [https://en.wikipedia.org/wiki/Run-time\\_type\\_information](https://en.wikipedia.org/wiki/Run-time_type_information)

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Types  
MD5: 45baaad66a1047358addb574d0fa00bc

## 18.3 Classwide types

To remain consistent with the rest of the language, a new notation needed to be introduced to say "This object is of this type or any descendant derives tagged type".

In Ada, we call this the *classwide type*. It's used in OOP as soon as you need polymorphism. For example, you can't do the following:

Listing 260: main.adb

```

1  with P; use P;
2
3  procedure Main is
4
5    01 : My_Class;
6    -- Declaring an object of type My_Class
7
8    02 : Derived := (A => 12);
9    -- Declaring an object of type Derived
10
11   03 : My_Class := 02;
12   -- INVALID: Trying to assign a value
13   -- of type derived to a variable of
14   -- type My_Class.
15 begin
16   null;
17 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Types  
MD5: c87ad8bb686cb1763740750846258357

### Build output

```

main.adb:11:21: error: expected type "My_Class" defined at p.ads:2
main.adb:11:21: error: found type "Derived" defined at p.ads:16
gprbuild: *** compilation phase failed

```

This is because an object of a type T is exactly of the type T, whether T is tagged or not. What you want to say as a programmer is "I want 03 to be able to hold an object of type My\_Class or any type descending from My\_Class". Here's how you do that:

Listing 261: main.adb

```

1  with P; use P;
2
3  procedure Main is
4    01 : My_Class;
5    -- Declare an object of type My_Class
6
7    02 : Derived := (A => 12);
8    -- Declare an object of type Derived
9
10   03 : My_Class'Class := 02;
11   -- Now valid: My_Class'Class designates
12   -- the classwide type for My_Class,

```

(continues on next page)

(continued from previous page)

```
13  -- which is the set of all types
14  -- descending from My_Class (including
15  -- My_Class).
16 begin
17   null;
18 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Types  
MD5: 35412176a248015a26e507164ce526af

#### ⚠️ Attention

Because an object of a classwide type can be the size of any descendant of its base type, it has an unknown size. It's therefore an indefinite type, with the expected restrictions:

- It can't be stored as a field/component of a record
- An object of a classwide type needs to be initialized immediately (you can't specify the constraints of such a type in any way other than by initializing it).

## 18.4 Dispatching operations

We saw that you can override operations in types derived from another tagged type. The eventual goal of OOP is to make a dispatching call: a call to a primitive (method) that depends on the exact type of the object.

But, if you think carefully about it, a variable of type `My_Class` always contains an object of exactly that type. If you want to have a variable that can contain a `My_Class` or any derived type, it has to be of type `My_Class'Class`.

In other words, to make a dispatching call, you must first have an object that can be either of a type or any type derived from this type, namely an object of a classwide type.

Listing 262: main.adb

```
1 with P; use P;
2
3 procedure Main is
4   01 : My_Class;
5   -- Declare an object of type My_Class
6
7   02 : Derived := (A => 12);
8   -- Declare an object of type Derived
9
10  03 : My_Class'Class := 02;
11
12  04 : My_Class'Class := 01;
13 begin
14   Foo (01);
15   -- Non dispatching: Calls My_Class.Foo
16   Foo (02);
17   -- Non dispatching: Calls Derived.Foo
18   Foo (03);
19   -- Dispatching: Calls Derived.Foo
20   Foo (04);
```

(continues on next page)

(continued from previous page)

```

21  -- Dispatching: Calls My_Class.Foo
22 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Types  
MD5: 7631f823b0dd9e5474f6bb2dc35af2a2

### Runtime output

```

In My_Class.Foo
In Derived.Foo, A = 12
In Derived.Foo, A = 12
In My_Class.Foo

```

#### Attention

You can convert an object of type `Derived` to an object of type `My_Class`. This is called a *view conversion* in Ada parlance and is useful, for example, if you want to call a parent method.

In that case, the object really is converted to a `My_Class` object, which means its tag is changed. Since tagged objects are always passed by reference, you can use this kind of conversion to modify the state of an object: changes to converted object will affect the original one.

Listing 263: main.adb

```

1  with P; use P;
2
3  procedure Main is
4    01 : Derived := (A => 12);
5    -- Declare an object of type Derived
6
7    02 : My_Class := My_Class (01);
8
9    03 : My_Class'Class := 02;
10 begin
11   Foo (01);
12   -- Non dispatching: Calls Derived.Foo
13   Foo (02);
14   -- Non dispatching: Calls My_Class.Foo
15
16   Foo (03);
17   -- Dispatching: Calls My_Class.Foo
18 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Types  
MD5: b92112b05201ff14789baca258fa0cbc

### Runtime output

```

In Derived.Foo, A = 12
In My_Class.Foo
In My_Class.Foo

```

## 18.5 Dot notation

You can also call primitives of tagged types with a notation that's more familiar to object oriented programmers. Given the `Foo` primitive above, you can also write the above program this way:

Listing 264: main.adb

```

1  with P; use P;
2
3  procedure Main is
4      01 : My_Class;
5      -- Declare an object of type My_Class
6
7      02 : Derived := (A => 12);
8      -- Declare an object of type Derived
9
10     03 : My_Class'Class := 02;
11
12     04 : My_Class'Class := 01;
13 begin
14     01.Foo;
15     -- Non dispatching: Calls My_Class.Foo
16     02.Foo;
17     -- Non dispatching: Calls Derived.Foo
18     03.Foo;
19     -- Dispatching: Calls Derived.Foo
20     04.Foo;
21     -- Dispatching: Calls My_Class.Foo
22 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Types  
MD5: 9c6ebdfec9ceeb986d92eb90ec9ff59b

### Runtime output

```

In My_Class.Foo
In Derived.Foo, A = 12
In Derived.Foo, A = 12
In My_Class.Foo

```

If the dispatching parameter of a primitive is the first parameter, which is the case in our examples, you can call the primitive using the dot notation. Any remaining parameter are passed normally:

Listing 265: main.adb

```

1  with P; use P;
2
3  procedure Main is
4      package Extend is
5          type D2 is new Derived with null record;
6
7          procedure Bar (Self : in out D2;
8                          Val : Integer);
9      end Extend;
10
11     package body Extend is
12         procedure Bar (Self : in out D2;

```

(continues on next page)

(continued from previous page)

```

13           Val : Integer) is
14   begin
15     Self.A := Self.A + Val;
16   end Bar;
17 end Extend;
18
19 use Extend;
20
21 Obj : D2 := (A => 15);
22 begin
23   Obj.Bar (2);
24   Obj.Foo;
25 end Main;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Types  
MD5: fec4f5cc4213cc111708dcc276e870c2

### Runtime output

In Derived.Foo, A = 17

## 18.6 Private & Limited

We've seen previously (in the *Privacy* (page 111) chapter) that types can be declared limited or private. These encapsulation techniques can also be applied to tagged types, as we'll see in this section.

This is an example of a tagged private type:

Listing 266: p.ads

```

1 package P is
2   type T is tagged private;
3   private
4     type T is tagged record
5       E : Integer;
6     end record;
7 end P;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Private\_Types  
MD5: 4cd4bcd1a54d5f6407a500558b5da417

This is an example of a tagged limited type:

Listing 267: p.ads

```

1 package P is
2   type T is tagged limited record
3     E : Integer;
4   end record;
5 end P;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Limited\_Types  
MD5: 13228777133aa6db97dalc29f732459c

Naturally, you can combine both *limited* and *private* types and declare a tagged limited private type:

Listing 268: p.ads

```
1 package P is
2     type T is tagged limited private;
3
4     procedure Init (A : in out T);
5 private
6     type T is tagged limited record
7         E : Integer;
8     end record;
9 end P;
```

Listing 269: p.adb

```
1 package body P is
2
3     procedure Init (A : in out T) is
4 begin
5     A.E := 0;
6 end Init;
7
8 end P;
```

Listing 270: main.adb

```
1 with P; use P;
2
3 procedure Main is
4     T1, T2 : T;
5 begin
6     T1.Init;
7     T2.Init;
8
9     -- The following line doesn't work
10    -- because type T is private:
11    --
12    -- T1.E := 0;
13
14    -- The following line doesn't work
15    -- because type T is limited:
16    --
17    -- T2 := T1;
18 end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Tagged\_Limited\_Private\_Types  
MD5: 68240374505bcacf7aad4ebaed3b9127b

Note that the code in the Main procedure above presents two assignments that trigger compilation errors because type T is limited private. In fact, you cannot:

- assign to T1.E directly because type T is private;
- assign T1 to T2 because type T is limited.

In this case, there's no distinction between tagged and non-tagged types: these compilation errors would also occur for non-tagged types.

## 18.7 Classwide access types

In this section, we'll discuss an useful pattern for object-oriented programming in Ada: class-wide access type. Let's start with an example where we declare a tagged type T and a derived type T\_New:

Listing 271: p.ads

```

1 package P is
2     type T is tagged null record;
3
4     procedure Show (Dummy : T);
5
6     type T_New is new T with null record;
7
8     procedure Show (Dummy : T_New);
9 end P;

```

Listing 272: p.adb

```

1 with Ada.Text_IO; use Ada.Text_IO;
2
3 package body P is
4
5     procedure Show (Dummy : T) is
6     begin
7         Put_Line ("Using type "
8                   & T'External_Tag);
9     end Show;
10
11    procedure Show (Dummy : T_New) is
12    begin
13        Put_Line ("Using type "
14                   & T_New'External_Tag);
15    end Show;
16
17 end P;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Classwide\_Error  
MD5: fd5cb99925d3c88536546aa0be8104b7

Note that we're using null records for both types T and T\_New. Although these types don't actually have any component, we can still use them to demonstrate dispatching. Also note that the example above makes use of the 'External\_Tag' attribute in the implementation of the Show procedure to get a string for the corresponding tagged type.

As we've seen before, we must use a classwide type to create objects that can make dispatching calls. In other words, objects of type T'Class will dispatch. For example:

Listing 273: dispatching\_example.adb

```

1 with P; use P;
2
3 procedure Dispatching_Example is
4     T2          : T_New;
5     T_Dispatch : constant T'Class := T2;
6 begin
7     T_Dispatch.Show;
8 end Dispatching_Example;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Classwide\_Error  
MD5: f8957b31c9c62db23759baad7b867a57

### Runtime output

Using type P.T\_NEW

A more useful application is to declare an array of objects that can dispatch. For example, we'd like to declare an array T\_Arr, loop over this array and dispatch according to the actual type of each individual element:

```
for I in T_Arr'Range loop
    T_Arr (I).Show;
    -- Call Show procedure according
    -- to actual type of T_Arr (I)
end loop;
```

However, it's not possible to declare an array of type T'Class directly:

Listing 274: classwide\_compilation\_error.adb

```
1 with P; use P;
2
3 procedure Classwide_Compilation_Error is
4     T_Arr : array (1 .. 2) of T'Class;
5     --
6     --          Compilation Error!
7 begin
8     for I in T_Arr'Range loop
9         T_Arr (I).Show;
10    end loop;
11 end Classwide_Compilation_Error;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Classwide\_Error  
MD5: e86f6c6ee35dced8f330bf6177d178fd

### Build output

```
classwide_compilation_error.adb:4:32: error: unconstrained element type in array
  declaration
gprbuild: *** compilation phase failed
```

In fact, it's impossible for the compiler to know which type would actually be used for each element of the array. However, if we use dynamic allocation via access types, we can allocate objects of different types for the individual elements of an array T\_Arr. We do this by using classwide access types, which have the following format:

```
type T_Class is access T'Class;
```

We can rewrite the previous example using the T\_Class type. In this case, dynamically allocated objects of this type will dispatch according to the actual type used during the allocation. Also, let's introduce an Init procedure that won't be overridden for the derived T\_New type. This is the adapted code:

Listing 275: p.ads

```
1 package P is
2     type T is tagged record
```

(continues on next page)

(continued from previous page)

```

3   E : Integer;
4 end record;
5
6 type T_Class is access T'Class;
7
8 procedure Init (A : in out T);
9
10 procedure Show (Dummy : T);
11
12 type T_New is new T with null record;
13
14 procedure Show (Dummy : T_New);
15
16 end P;

```

Listing 276: p.adb

```

1 with Ada.Text_IO; use Ada.Text_IO;
2
3 package body P is
4
5   procedure Init (A : in out T) is
6   begin
7     Put_Line ("Initializing type T...");
8     A.E := 0;
9   end Init;
10
11  procedure Show (Dummy : T) is
12  begin
13    Put_Line ("Using type "
14      & T'External_Tag);
15  end Show;
16
17  procedure Show (Dummy : T_New) is
18  begin
19    Put_Line ("Using type "
20      & T_New'External_Tag);
21  end Show;
22
23 end P;

```

Listing 277: main.adb

```

1 with Ada.Text_IO; use Ada.Text_IO;
2 with P;           use P;
3
4 procedure Main is
5   T_Arr : array (1 .. 2) of T_Class;
6 begin
7   T_Arr (1) := new T;
8   T_Arr (2) := new T_New;
9
10  for I in T_Arr'Range loop
11    Put_Line ("Element # "
12      & Integer'Image (I));
13
14  T_Arr (I).Init;
15  T_Arr (I).Show;
16
17  Put_Line ("-----");

```

(continues on next page)

(continued from previous page)

```
18      end loop;  
19  end Main;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Object\_Oriented\_Programming.Classwide\_Access  
MD5: 97c05a8f911d0a0e39c0cc90fae184a7

### Runtime output

```
Element # 1  
Initializing type T...  
Using type P.T  
-----  
Element # 2  
Initializing type T...  
Using type P.T_NEW  
-----
```

In this example, the first element (T\_Arr (1)) is of type T, while the second element is of type T\_New. When running the example, the Init procedure of type T is called for both elements of the T\_Arr array, while the call to the Show procedure selects the corresponding procedure according to the type of each element of T\_Arr.

## STANDARD LIBRARY: CONTAINERS

In previous chapters, we've used arrays as the standard way to group multiple objects of a specific data type. In many cases, arrays are good enough for manipulating those objects. However, there are situations that require more flexibility and more advanced operations. For those cases, Ada provides support for containers — such as vectors and sets — in its standard library.

We present an introduction to containers here. For a list of all containers available in Ada, see [Appendix B](#) (page 269).

### 19.1 Vectors

In the following sections, we present a general overview of vectors, including instantiation, initialization, and operations on vector elements and vectors.

#### 19.1.1 Instantiation

Here's an example showing the instantiation and declaration of a vector V:

Listing 278: show\_vector\_inst.adb

```
1  with Ada.Containers.Vectors;
2
3  procedure Show_Vector_Inst is
4
5    package Integer_Vectors is new
6      Ada.Containers.Vectors
7        (Index_Type    => Natural,
8         Element_Type => Integer);
9
10   V : Integer_Vectors.Vector;
11 begin
12   null;
13 end Show_Vector_Inst;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_Inst  
MD5: 8b737842d2784f25502990f21e1cf6de

Containers are based on generic packages, so we can't simply declare a vector as we would declare an array of a specific type:

```
A : array (1 .. 10) of Integer;
```

Instead, we first need to instantiate one of those packages. We **with** the container package (Ada.Containers.Vectors in this case) and instantiate it to create an instance of the

generic package for the desired type. Only then can we declare the vector using the type from the instantiated package. This instantiation needs to be done for any container type from the standard library.

In the instantiation of `Integer_Vectors`, we indicate that the vector contains elements of `Integer` type by specifying it as the `Element_Type`. By setting `Index_Type` to `Natural`, we specify that the allowed range includes all natural numbers. We could have used a more restrictive range if desired.

### 19.1.2 Initialization

One way to initialize a vector is from a concatenation of elements. We use the `&` operator, as shown in the following example:

Listing 279: `show_vector_init.adb`

```
1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Vectors;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5
6  procedure Show_Vector_Init is
7
8    package Integer_Vectors is new
9      Ada.Containers.Vectors
10     (Index_Type    => Natural,
11      Element_Type => Integer);
12
13    use Integer_Vectors;
14
15    V : Vector := 20 & 10 & 0 & 13;
16  begin
17    Put_Line ("Vector has "
18              & Count_Type'Image (V.Length)
19              & " elements");
20  end Show_Vector_Init;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_Init  
MD5: 0087b0a15e0c88b27ac36c3b27159a17

#### Runtime output

Vector has 4 elements

We specify `use Integer_Vectors`, so we have direct access to the types and operations from the instantiated package. Also, the example introduces another operation on the vector: `Length`, which retrieves the number of elements in the vector. We can use the dot notation because `Vector` is a tagged type, allowing us to write either `V.Length` or `Length (V)`.

### 19.1.3 Appending and prepending elements

You add elements to a vector using the `Prepend` and `Append` operations. As the names suggest, these operations add elements to the beginning or end of a vector, respectively. For example:

Listing 280: show\_vector\_append.adb

```

1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Vectors;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5
6  procedure Show_Vector_Append is
7
8    package Integer_Vectors is new
9      Ada.Containers.Vectors
10     (Index_Type    => Natural,
11      Element_Type => Integer);
12
13  use Integer_Vectors;
14
15  V : Vector;
16  begin
17    Put_Line ("Appending some elements "
18              & "to the vector...");
19    V.Append (20);
20    V.Append (10);
21    V.Append (0);
22    V.Append (13);
23    Put_Line ("Finished appending.");
24
25    Put_Line ("Prepending some elements"
26              & "to the vector...");
27    V.Prepend (30);
28    V.Prepend (40);
29    V.Prepend (100);
30    Put_Line ("Finished prepending.");
31
32    Put_Line ("Vector has "
33              & Count_Type'Image (V.Length)
34              & " elements");
35  end Show_Vector_Append;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_Append  
MD5: f88d393ba96a7950f58d9f1c0c74a021

### Runtime output

```

Appending some elements to the vector...
Finished appending.
Prepending some elements to the vector...
Finished prepending.
Vector has 7 elements

```

This example puts elements into the vector in the following sequence: (100, 40, 30, 20, 10, 0, 13).

The Reference Manual specifies that the worst-case complexity must be:

- $O(\log N)$  for the Append operation, and
- $O(N \log N)$  for the Prepend operation.

### 19.1.4 Accessing first and last elements

We access the first and last elements of a vector using the `First_Element` and `Last_Element` functions. For example:

Listing 281: `show_vector_first_last_element.adb`

```
1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Vectors;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5
6  procedure Show_Vector_First_Last_Element is
7
8    package Integer_Vectors is new
9      Ada.Containers.Vectors
10     (Index_Type  => Natural,
11      Element_Type => Integer);
12
13  use Integer_Vectors;
14
15  function Img (I : Integer)      return String
16    renames Integer'Image;
17  function Img (I : Count_Type) return String
18    renames Count_Type'Image;
19
20  V : Vector := 20 & 10 & 0 & 13;
21 begin
22  Put_Line ("Vector has "
23            & Img (V.Length)
24            & " elements");
25
26  -- Using V.First_Element to
27  -- retrieve first element
28  Put_Line ("First element is "
29            & Img (V.First_Element));
30
31  -- Using V.Last_Element to
32  -- retrieve last element
33  Put_Line ("Last element is "
34            & Img (V.Last_Element));
35 end Show_Vector_First_Last_Element;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_First\_Last\_Element  
MD5: 602255760d0017ced6b4115c845cd48d

#### Runtime output

```
Vector has 4 elements
First element is 20
Last element is 13
```

You can swap elements by calling the procedure `Swap` and retrieving a reference (a *cursor*) to the first and last elements of the vector by calling `First` and `Last`. A cursor allows us to iterate over a container and process individual elements from it.

With these operations, we're able to write code to swap the first and last elements of a vector:

Listing 282: show\_vector\_first\_last\_element.adb

```

1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Vectors;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5
6  procedure Show_Vector_First_Last_Element is
7
8    package Integer_Vectors is new
9      Ada.Containers.Vectors
10     (Index_Type    => Natural,
11      Element_Type => Integer);
12
13  use Integer_Vectors;
14
15  function Img (I : Integer) return String
16    renames Integer'Image;
17
18  V : Vector := 20 & 10 & 0 & 13;
19 begin
20  -- We use V.First and V.Last to retrieve
21  -- cursor for first and last elements.
22  -- We use V.Swap to swap elements.
23  V.Swap (V.First, V.Last);
24
25  Put_Line ("First element is now "
26            & Img (V.First_Element));
27  Put_Line ("Last element is now "
28            & Img (V.Last_Element));
29 end Show_Vector_First_Last_Element;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_First\_Last\_Element  
MD5: 1a0c0bf28bb661b3f328473ac3c2eb54

### Runtime output

First element is now 13  
Last element is now 20

## 19.1.5 Iterating

The easiest way to iterate over a container is to use a **for E of Our\_Container** loop. This gives us a reference (E) to the element at the current position. We can then use E directly. For example:

Listing 283: show\_vector\_iteration.adb

```

1  with Ada.Containers.Vectors;
2
3  with Ada.Text_IO; use Ada.Text_IO;
4
5  procedure Show_Vector_Iteration is
6
7    package Integer_Vectors is new
8      Ada.Containers.Vectors
9      (Index_Type    => Natural,
10       Element_Type => Integer);
10

```

(continues on next page)

(continued from previous page)

```
11  use Integer_Vectors;
12
13
14  function Img (I : Integer) return String
15      renames Integer'Image;
16
17  V : Vector := 20 & 10 & 0 & 13;
18 begin
19     Put_Line ("Vector elements are: ");
20
21     -- Using for ... of loop to iterate:
22     --
23     for E of V loop
24         Put_Line ("- " & Img (E));
25     end loop;
26
27
28 end Show_Vector_Iteration;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_Iteration  
MD5: 4fc9a939aa822097d3a937646d3e2910

### Runtime output

```
Vector elements are:
- 20
- 10
- 0
- 13
```

This code displays each element from the vector V.

Because we're given a reference, we can display not only the value of an element but also modify it. For example, we could easily write a loop to add one to each element of vector V:

```
for E of V loop
    E := E + 1;
end loop;
```

We can also use indices to access vector elements. The format is similar to a loop over array elements: we use a **for I in <range>** loop. The range is provided by V.First\_Index and V.Last\_Index. We can access the current element by using it as an array index: V (I). For example:

Listing 284: show\_vector\_index\_iteration.adb

```
1 with Ada.Containers.Vectors;
2
3 with Ada.Text_IO; use Ada.Text_IO;
4
5 procedure Show_Vector_Index_Iteration is
6
7     package Integer_Vectors is new
8         Ada.Containers.Vectors
9             (Index_Type    => Natural,
10              Element_Type => Integer);
11
12     use Integer_Vectors;
```

(continues on next page)

(continued from previous page)

```

13      V : Vector := 20 & 10 & 0 & 13;
14  begin
15      Put_Line ("Vector elements are: ");
16
17
18      -- Using indices in a "for I in ..." loop
19      -- to iterate:
20      --
21      for I in V.First_Index .. V.Last_Index loop
22          -- Displaying current index I
23          Put (" - ["
24              & Extended_Index'Image (I)
25              & "] ");
26
27          Put (Integer'Image (V (I)));
28
29          -- We could also use the V.Element (I)
30          -- function to retrieve the element at
31          -- the current index I
32
33          New_Line;
34      end loop;
35
36  end Show_Vector_Index_Iteration;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_Index\_Iteration  
MD5: f5600bbcc53d6d6887a771b1505676e9

### Runtime output

```

Vector elements are:
- [ 0] 20
- [ 1] 10
- [ 2] 0
- [ 3] 13

```

Here, in addition to displaying the vector elements, we're also displaying each index, I, just like what we can do for array indices. Also, we can access the element by using either the short form `V (I)` or the longer form `V.Element (I)` but not `V.I`.

As mentioned in the previous section, you can use cursors to iterate over containers. For this, use the function `Iterate`, which retrieves a cursor for each position in the vector. The corresponding loop has the format `for C in V.Iterate loop`. Like the previous example using indices, you can again access the current element by using the cursor as an array index: `V (C)`. For example:

Listing 285: `show_vector_cursor_iteration.adb`

```

1  with Ada.Containers.Vectors;
2
3  with Ada.Text_IO; use Ada.Text_IO;
4
5  procedure Show_Vector_Cursor_Iteration is
6
7      package Integer_Vectors is new
8          Ada.Containers.Vectors
9          (Index_Type    => Natural,
10           Element_Type => Integer);

```

(continues on next page)

(continued from previous page)

```
11  use Integer_Vectors;
12
13
14  V : Vector := 20 & 10 & 0 & 13;
15  begin
16      Put_Line ("Vector elements are: ");
17
18      --
19      -- Use a cursor to iterate in a loop:
20      --
21      for C in V.Iterate loop
22          -- Using To_Index function to retrieve
23          -- the index for the cursor position
24          Put (" - ["
25              & Extended_Index'Image (To_Index (C))
26              & "] ");
27
28          Put (Integer'Image (V (C)));
29
30          -- We could use Element (C) to retrieve
31          -- the vector element for the cursor
32          -- position
33
34          New_Line;
35      end loop;
36
37      -- Alternatively, we could iterate with a
38      -- while-loop:
39      --
40      -- declare
41      --     C : Cursor := V.First;
42      -- begin
43      --     while C /= No_Element loop
44      --         some processing here...
45      --
46      --         C := Next (C);
47      --     end loop;
48      -- end;
49
50  end Show_Vector_Cursor_Iteration;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_Cursor\_Iteration  
MD5: de789bbd2e1814aae3fb5213c99ac25c

### Runtime output

```
Vector elements are:
- [ 0] 20
- [ 1] 10
- [ 2] 0
- [ 3] 13
```

Instead of accessing an element in the loop using `V (C)`, we could also have used the longer form `Element (C)`. In this example, we're using the function `To_Index` to retrieve the index corresponding to the current cursor.

As shown in the comments after the loop, we could also use a `while ... loop` to iterate over the vector. In this case, we would start with a cursor for the first element (retrieved by calling `V.First`) and then call `Next (C)` to retrieve a cursor for subsequent elements. `Next (C)` returns `No_Element` when the cursor reaches the end of the vector.

You can directly modify the elements using a reference. This is what it looks like when using both indices and cursors:

```
-- Modify vector elements using index
for I in V.First_Index .. V.Last_Index loop
    V (I) := V (I) + 1;
end loop;

-- Modify vector elements using cursor
for C in V.Iterate loop
    V (C) := V (C) + 1;
end loop;
```

The Reference Manual requires that the worst-case complexity for accessing an element be  $O(\log N)$ .

Another way of modifying elements of a vector is using a *process procedure*, which takes an individual element and does some processing on it. You can call `Update_Element` and pass both a cursor and an access to the process procedure. For example:

Listing 286: `show_vector_update.adb`

```
1  with Ada.Containers.Vectors;
2
3  with Ada.Text_IO; use Ada.Text_IO;
4
5  procedure Show_Vector_Update is
6
7      package Integer_Vectors is new
8          Ada.Containers.Vectors
9          (Index_Type  => Natural,
10           Element_Type => Integer);
11
12      use Integer_Vectors;
13
14      procedure Add_One (I : in out Integer) is
15      begin
16          I := I + 1;
17      end Add_One;
18
19      V : Vector := 20 & 10 & 12;
20  begin
21      --
22      -- Use V.Update_Element to process elements
23      --
24      for C in V.Iterate loop
25          V.Update_Element (C, Add_One'Access);
26      end loop;
27
28  end Show_Vector_Update;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_Update  
MD5: 5dcc3dd8020632a8ea2ce975ecd8f4da

## 19.1.6 Finding and changing elements

You can locate a specific element in a vector by retrieving its index. `Find_Index` retrieves the index of the first element matching the value you're looking for. Alternatively, you can use `Find` to retrieve a cursor referencing that element. For example:

Listing 287: show\_find\_vector\_element.adb

```
1  with Ada.Containers.Vectors;
2
3  with Ada.Text_IO; use Ada.Text_IO;
4
5  procedure Show_Find_Vector_Element is
6
7    package Integer_Vectors is new
8      Ada.Containers.Vectors
9      (Index_Type  => Natural,
10       Element_Type => Integer);
11
12   use Integer_Vectors;
13
14   V : Vector := 20 & 10 & 0 & 13;
15   Idx : Extended_Index;
16   C : Cursor;
17
18 begin
19   -- Using Find_Index to retrieve the index
20   -- of element with value 10
21   Idx := V.Find_Index (10);
22   Put_Line ("Index of element with value 10 is "
23             & Extended_Index'Image (Idx));
24
25   -- Using Find to retrieve the cursor for
26   -- the element with value 13
27   C := V.Find (13);
28   Idx := To_Index (C);
29   Put_Line ("Index of element with value 13 is "
30             & Extended_Index'Image (Idx));
31 end Show_Find_Vector_Element;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Find\_Vector\_Element  
MD5: c3da01cd66c8705a7cbccae8390d5f81

### Runtime output

```
Index of element with value 10 is 1
Index of element with value 13 is 3
```

As we saw in the previous section, we can directly access vector elements by using either an index or cursor. However, an exception is raised if we try to access an element with an invalid index or cursor, so we must check whether the index or cursor is valid before using it to access an element. In our example, `Find_Index` or `Find` might not have found the element in the vector. We check for this possibility by comparing the index to `No_Index` or the cursor to `No_Element`. For example:

```
-- Modify vector element using index
if Idx /= No_Index then
  V (Idx) := 11;
end if;

-- Modify vector element using cursor
if C /= No_Element then
  V (C) := 14;
end if;
```

Instead of writing `V (C) := 14`, we could use the longer form `V.Replace_Element (C, 14)`.

### 19.1.7 Inserting elements

In the previous sections, we've seen examples of how to add elements to a vector:

- using the concatenation operator (&) at the vector declaration, or
- calling the Prepend and Append procedures.

You may want to insert an element at a specific position, e.g. before a certain element in the vector. You do this by calling `Insert`. For example:

Listing 288: `show_vector_insert.adb`

```

1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Vectors;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5
6  procedure Show_Vector_Insert is
7
8    package Integer_Vectors is new
9      Ada.Containers.Vectors
10     (Index_Type  => Natural,
11      Element_Type => Integer);
12
13  use Integer_Vectors;
14
15  procedure Show_Elements (V : Vector) is
16  begin
17    New_Line;
18    Put_Line ("Vector has "
19              & Count_Type'Image (V.Length)
20              & " elements");
21
22    if not V.Is_Empty then
23      Put_Line ("Vector elements are: ");
24      for E of V loop
25        Put_Line ("- " & Integer'Image (E));
26      end loop;
27    end if;
28  end Show_Elements;
29
30  V : Vector := 20 & 10 & 12;
31  C : Cursor;
32
33 begin
34  Show_Elements (V);
35
36  New_Line;
37  Put_Line ("Adding element with value 9");
38  Put_Line (" (before 10)...");
39
40  -- Using V.Insert to insert the element
41  -- into the vector
42
43  C := V.Find (10);
44  if C /= No_Element then
45    V.Insert (C, 9);
46  end if;
47
48  Show_Elements (V);
49
50 end Show_Vector_Insert;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_Insert  
MD5: af49f390388896c51ab97541036fbcaf

### Runtime output

```
Vector has 3 elements
Vector elements are:
```

- 20
- 10
- 12

```
Adding element with value 9
(before 10)...
```

```
Vector has 4 elements
Vector elements are:
```

- 20
- 9
- 10
- 12

In this example, we're looking for an element with the value of 10. If we find it, we insert an element with the value of 9 before it.

### 19.1.8 Removing elements

You can remove elements from a vector by passing either a valid index or cursor to the `Delete` procedure. If we combine this with the functions `Find_Index` and `Find` from the previous section, we can write a program that searches for a specific element and deletes it, if found:

Listing 289: `show_remove_vector_element.adb`

```
1  with Ada.Containers.Vectors;
2
3  with Ada.Text_IO; use Ada.Text_IO;
4
5  procedure Show_Remove_Vector_Element is
6    package Integer_Vectors is new
7      Ada.Containers.Vectors
8        (Index_Type  => Natural,
9         Element_Type => Integer);
10
11  use Integer_Vectors;
12
13  V : Vector := 20 & 10 & 0 & 13 & 10 & 13;
14  Idx : Extended_Index;
15  C : Cursor;
16
17  begin
18    -- Use Find_Index to retrieve index of
19    -- the element with value 10
20    Idx := V.Find_Index (10);
21
22    -- Checking whether index is valid
23    if Idx /= No_Index then
24      -- Removing element using V.Delete
25      V.Delete (Idx);
26    end if;
```

(continues on next page)

(continued from previous page)

```

26
27  -- Use Find to retrieve cursor for
28  -- the element with value 13
29  C := V.Find (13);
30
31  -- Check whether index is valid
32  if C /= No_Element then
33    -- Remove element using V.Delete
34    V.Delete (C);
35  end if;
36
37 end Show_Remove_Vector_Element;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Remove\_Vector\_Element  
MD5: 540d0dc5715e58926e9dc4600bd6ad5d

We can extend this approach to delete all elements matching a certain value. We just need to keep searching for the element in a loop until we get an invalid index or cursor. For example:

Listing 290: show\_remove\_vector\_elements.adb

```

1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Vectors;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5
6  procedure Show_Remove_Vector_Elements is
7
8    package Integer_Vectors is new
9      Ada.Containers.Vectors
10     (Index_Type  => Natural,
11      Element_Type => Integer);
12
13  use Integer_Vectors;
14
15  procedure Show_Elements (V : Vector) is
16  begin
17    New_Line;
18    Put_Line ("Vector has "
19              & Count_Type'Image (V.Length)
20              & " elements");
21
22    if not V.Is_Empty then
23      Put_Line ("Vector elements are: ");
24      for E of V loop
25        Put_Line ("- " & Integer'Image (E));
26      end loop;
27    end if;
28  end Show_Elements;
29
30  V : Vector := 20 & 10 & 0 & 13 & 10 & 14 & 13;
31 begin
32  Show_Elements (V);
33
34  -- Remove elements using an index
35  --
36  declare

```

(continues on next page)

(continued from previous page)

```
38  E : constant Integer := 10;
39  I : Extended_Index;
40  begin
41    New_Line;
42    Put_Line
43      ("Removing all elements with value of "
44       & Integer'Image (E) & "...");
45  loop
46    I := V.Find_Index (E);
47    exit when I = No_Index;
48    V.Delete (I);
49  end loop;
50 end;
51
52 -- Remove elements using a cursor
53 --
54 declare
55  E : constant Integer := 13;
56  C : Cursor;
57 begin
58  New_Line;
59  Put_Line
60    ("Removing all elements with value of "
61     & Integer'Image (E) & "...");
62  loop
63    C := V.Find (E);
64    exit when C = No_Element;
65    V.Delete (C);
66  end loop;
67 end;
68
69 Show_Elements (V);
70 end Show_Remove_Vector_Elements;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Remove\_Vector\_Elements  
MD5: 6e364843b9638224bd9a36eb9d45e446

### Runtime output

```
Vector has 7 elements
Vector elements are:
- 20
- 10
- 0
- 13
- 10
- 14
- 13

Removing all elements with value of 10...

Removing all elements with value of 13...

Vector has 3 elements
Vector elements are:
- 20
- 0
```

(continues on next page)

(continued from previous page)

- 14

In this example, we remove all elements with the value 10 from the vector by retrieving their index. Likewise, we remove all elements with the value 13 by retrieving their cursor.

### 19.1.9 Other Operations

We've seen some operations on vector elements. Here, we'll see operations on the vector as a whole. The most prominent is the concatenation of multiple vectors, but we'll also see operations on vectors, such as sorting and sorted merging operations, that view the vector as a sequence of elements and operate on the vector considering the element's relations to each other.

We do vector concatenation using the `&` operator on vectors. Let's consider two vectors `V1` and `V2`. We can concatenate them by doing `V := V1 & V2`. `V` contains the resulting vector.

The generic package `Generic_Sorting` is a child package of `Ada.Containers.Vectors`. It contains sorting and merging operations. Because it's a generic package, you can't use it directly, but have to instantiate it. In order to use these operations on a vector of integer values (`Integer_Vectors`, in our example), you need to instantiate it directly as a child of `Integer_Vectors`. The next example makes it clear how to do this.

After instantiating `Generic_Sorting`, we make all the operations available to us with the `use` statement. We can then call `Sort` to sort the vector and `Merge` to merge one vector into another.

The following example presents code that manipulates three vectors (`V1`, `V2`, `V3`) using the concatenation, sorting and merging operations:

Listing 291: `show_vector_ops.adb`

```

1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Vectors;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5
6  procedure Show_Vector_Ops is
7
8    package Integer_Vectors is new
9      Ada.Containers.Vectors
10     (Index_Type  => Natural,
11      Element_Type => Integer);
12
13    package Integer_Vectors_Sorting is
14      new Integer_Vectors.Generic_Sorting;
15
16    use Integer_Vectors;
17    use Integer_Vectors_Sorting;
18
19    procedure Show_Elements (V : Vector) is
20    begin
21      New_Line;
22      Put_Line ("Vector has "
23                 & Count_Type'Image (V.Length)
24                 & " elements");
25
26      if not V.Is_Empty then
27        Put_Line ("Vector elements are: ");
28        for E of V loop
29          Put_Line ("- " & Integer'Image (E));
30        end loop;

```

(continues on next page)

(continued from previous page)

```
31      end if;
32  end Show_Elements;
33
34  V, V1, V2, V3 : Vector;
35 begin
36    V1 := 10 & 12 & 18;
37    V2 := 11 & 13 & 19;
38    V3 := 15 & 19;
39
40    New_Line;
41    Put_Line ("---- V1 ----");
42    Show_Elements (V1);
43
44    New_Line;
45    Put_Line ("---- V2 ----");
46    Show_Elements (V2);
47
48    New_Line;
49    Put_Line ("---- V3 ----");
50    Show_Elements (V3);
51
52    New_Line;
53    Put_Line
54      ("Concatenating V1, V2 and V3 into V:");
55
56    V := V1 & V2 & V3;
57
58    Show_Elements (V);
59
60    New_Line;
61    Put_Line ("Sorting V:");
62
63    Sort (V);
64
65    Show_Elements (V);
66
67    New_Line;
68    Put_Line ("Merging V2 into V1:");
69
70    Merge (V1, V2);
71
72    Show_Elements (V1);
73
74 end Show_Vector_Ops;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Vector\_Ops  
MD5: 3301513e4e7fd2f2848896e5b24e448

### Runtime output

```
---- V1 ----

Vector has 3 elements
Vector elements are:
- 10
- 12
- 18
```

(continues on next page)

(continued from previous page)

---- V2 ----

Vector has 3 elements  
Vector elements are:

- 11
- 13
- 19

---- V3 ----

Vector has 2 elements  
Vector elements are:

- 15
- 19

Concatenating V1, V2 and V3 into V:

Vector has 8 elements  
Vector elements are:

- 10
- 12
- 18
- 11
- 13
- 19
- 15
- 19

Sorting V:

Vector has 8 elements  
Vector elements are:

- 10
- 11
- 12
- 13
- 15
- 18
- 19
- 19

Merging V2 into V1:

Vector has 6 elements  
Vector elements are:

- 10
- 11
- 12
- 13
- 18
- 19

The Reference Manual requires that the worst-case complexity of a call to Sort be  $O(N^2)$  and the average complexity be better than  $O(N^2)$ .

## 19.2 Sets

Sets are another class of containers. While vectors allow duplicated elements to be inserted, sets ensure that no duplicated elements exist.

In the following sections, we'll see operations you can perform on sets. However, since many of the operations on vectors are similar to the ones used for sets, we'll cover them more quickly here. Please refer back to the section on vectors for a more detailed discussion.

### 19.2.1 Initialization and iteration

To initialize a set, you can call the `Insert` procedure. However, if you do, you need to ensure no duplicate elements are being inserted: if you try to insert a duplicate, you'll get an exception. If you have less control over the elements to be inserted so that there may be duplicates, you can use another option instead:

- a version of `Insert` that returns a Boolean value indicating whether the insertion was successful;
- the `Include` procedure, which silently ignores any attempt to insert a duplicated element.

To iterate over a set, you can use a `for E of S` loop, as you saw for vectors. This gives you a reference to each element in the set.

Let's see an example:

Listing 292: `show_set_init.adb`

```
1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Ordered_Sets;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5
6  procedure Show_Set_Init is
7
8    package Integer_Sets is new
9      Ada.Containers.Ordered_Sets
10     (Element_Type => Integer);
11
12    use Integer_Sets;
13
14    S : Set;
15    -- Same as: S : Integer_Sets.Set;
16    C : Cursor;
17    Ins : Boolean;
18  begin
19    S.Insert (20);
20    S.Insert (10);
21    S.Insert (0);
22    S.Insert (13);
23
24    -- Calling S.Insert(0) now would raise
25    -- Constraint_Error because this element
26    -- is already in the set. We instead call a
27    -- version of Insert that doesn't raise an
28    -- exception but instead returns a Boolean
29    -- indicating the status
30
31    S.Insert (0, C, Ins);
32    if not Ins then
33      Put_Line
34        ("Error while inserting 0 into set");
35    end if;
36
37    -- We can also call S.Include instead
```

(continues on next page)

(continued from previous page)

```

38  -- If the element is already present,
39  -- the set remains unchanged
40  S.Include (0);
41  S.Include (13);
42  S.Include (14);
43
44  Put_Line ("Set has "
45      & Count_Type'Image (S.Length)
46      & " elements");
47
48  --
49  -- Iterate over set using for .. of loop
50  --
51  Put_Line ("Elements:");
52  for E of S loop
53      Put_Line ("- " & Integer'Image (E));
54  end loop;
55 end Show_Set_Init;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Set\_Init  
MD5: b87f6729fea278396347248b95a30cb6

### Runtime output

```

Error while inserting 0 into set
Set has 5 elements
Elements:
- 0
- 10
- 13
- 14
- 20

```

## 19.2.2 Operations on elements

In this section, we briefly explore the following operations on sets:

- Delete and Exclude to remove elements;
- Contains and Find to verify the existence of elements.

To delete elements, you call the procedure `Delete`. However, analogously to the `Insert` procedure above, `Delete` raises an exception if the element to be deleted isn't present in the set. If you want to permit the case where an element might not exist, you can call `Exclude`, which silently ignores any attempt to delete a non-existent element.

`Contains` returns a Boolean value indicating whether a value is contained in the set. `Find` also looks for an element in a set, but returns a cursor to the element or `No_Element` if the element doesn't exist. You can use either function to search for elements in a set.

Let's look at an example that makes use of these operations:

Listing 293: `show_set_element_ops.adb`

```

1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Ordered_Sets;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5

```

(continues on next page)

(continued from previous page)

```
6  procedure Show_Set_Element_Ops is
7
8      package Integer_Sets is new
9          Ada.Containers.Ordered_Sets
10         (Element_Type => Integer);
11
12      use Integer_Sets;
13
14      procedure Show_Elements (S : Set) is
15      begin
16          New_Line;
17          Put_Line ("Set has "
18                  & Count_Type'Image (S.Length)
19                  & " elements");
20          Put_Line ("Elements:");
21          for E of S loop
22              Put_Line ("- " & Integer'Image (E));
23          end loop;
24      end Show_Elements;
25
26      S : Set;
27  begin
28      S.Insert (20);
29      S.Insert (10);
30      S.Insert (0);
31      S.Insert (13);
32
33      S.Delete (13);
34
35      -- Calling S.Delete (13) again raises
36      -- Constraint_Error because the element
37      -- is no longer present in the set, so
38      -- it can't be deleted. We can call
39      -- V.Exclude instead:
40      S.Exclude (13);
41
42      if S.Contains (20) then
43          Put_Line ("Found element 20 in set");
44      end if;
45
46      -- Alternatively, we could use S.Find
47      -- instead of S.Contains
48      if S.Find (0) /= No_Element then
49          Put_Line ("Found element 0 in set");
50      end if;
51
52      Show_Elements (S);
53  end Show_Set_Element_Ops;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Set\_Element\_Ops  
MD5: 77fb2aaba4221e337b0f90dd1a49c556

### Runtime output

```
Found element 20 in set
Found element 0 in set
```

```
Set has 3 elements
Elements:
```

(continues on next page)

(continued from previous page)

- 0
- 10
- 20

In addition to ordered sets used in the examples above, the standard library also offers hashed sets. The Reference Manual requires the following average complexity of each operation:

Operations	Ordered_Sets	Hashed_Sets
<ul style="list-style-type: none"> <li>• Insert</li> <li>• Include</li> <li>• Replace</li> <li>• Delete</li> <li>• Exclude</li> <li>• Find</li> </ul>	$O((\log N)^2)$ or better	$O(\log N)$
Subprogram using cursor	$O(1)$	$O(1)$

### 19.2.3 Other Operations

The previous sections mostly dealt with operations on individual elements of a set. But Ada also provides typical set operations: union, intersection, difference and symmetric difference. In contrast to some vector operations we've seen before (e.g. Merge), here you can use built-in operators, such as `-`. The following table lists the operations and its associated operator:

Set Operation	Operator
Union	<code>or</code>
Intersection	<code>and</code>
Difference	<code>-</code>
Symmetric difference	<code>xor</code>

The following example makes use of these operators:

Listing 294: show\_set\_ops.adb

```

1  with Ada.Containers; use Ada.Containers;
2  with Ada.Containers.Ordered_Sets;
3
4  with Ada.Text_IO; use Ada.Text_IO;
5
6  procedure Show_Set_Ops is
7
8    package Integer_Sets is new
9      Ada.Containers.Ordered_Sets
10     (Element_Type => Integer);
11
12    use Integer_Sets;
13
14    procedure Show_Elements (S : Set) is
15    begin
16      Put_Line ("Elements:");
17      for E of S loop
18        Put_Line ("- " & Integer'Image (E));

```

(continues on next page)

(continued from previous page)

```

19      end loop;
20  end Show_Elements;

21
22  procedure Show_Op (S          : Set;
23                      Op_Name : String) is
24  begin
25      New_Line;
26      Put_Line (Op_Name
27                  & "(set #1, set #2) has "
28                  & Count_Type'Image (S.Length)
29                  & " elements");
30  end Show_Op;

31
32  S1, S2, S3 : Set;
33 begin
34  S1.Insert (0);
35  S1.Insert (10);
36  S1.Insert (13);

37  S2.Insert (0);
38  S2.Insert (10);
39  S2.Insert (14);

40  S3.Insert (0);
41  S3.Insert (10);

42  New_Line;
43  Put_Line ("---- Set #1 ----");
44  Show_Elements (S1);

45  New_Line;
46  Put_Line ("---- Set #2 ----");
47  Show_Elements (S2);

48  New_Line;
49  Put_Line ("---- Set #3 ----");
50  Show_Elements (S3);

51
52  New_Line;
53  if S3.Is_Subset (S1) then
54      Put_Line ("S3 is a subset of S1");
55  else
56      Put_Line ("S3 is not a subset of S1");
57  end if;

58  S3 := S1 and S2;
59  Show_Op (S3, "Intersection");
60  Show_Elements (S3);

61  S3 := S1 or S2;
62  Show_Op (S3, "Union");
63  Show_Elements (S3);

64  S3 := S1 - S2;
65  Show_Op (S3, "Difference");
66  Show_Elements (S3);

67  S3 := S1 xor S2;
68  Show_Op (S3, "Symmetric difference");
69  Show_Elements (S3);
70
71
72
73
74
75
76
77
78
79

```

(continues on next page)

(continued from previous page)

80 `end Show_Set_Ops;`**Code block metadata**

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Set\_Ops  
 MD5: be9086591fc643e53facaf2ffea6c26d

**Runtime output**

```
---- Set #1 ----
Elements:
- 0
- 10
- 13

---- Set #2 ----
Elements:
- 0
- 10
- 14

---- Set #3 ----
Elements:
- 0
- 10

S3 is a subset of S1

Intersection(set #1, set #2) has 2 elements
Elements:
- 0
- 10

Union(set #1, set #2) has 4 elements
Elements:
- 0
- 10
- 13
- 14

Difference(set #1, set #2) has 1 elements
Elements:
- 13

Symmetric difference(set #1, set #2) has 2 elements
Elements:
- 13
- 14
```

## 19.3 Indefinite maps

The previous sections presented containers for elements of definite types. Although most examples in those sections presented **Integer** types as element type of the containers, containers can also be used with indefinite types, an example of which is the **String** type. However, indefinite types require a different kind of containers designed specially for them.

We'll also be exploring a different class of containers: maps. They associate a key with a specific value. An example of a map is the one-to-one association between a person and their age. If we consider a person's name to be the key, the value is the person's age.

### 19.3.1 Hashed maps

Hashed maps are maps that make use of a hash as a key. The hash itself is calculated by a function you provide.

#### In other languages

Hashed maps are similar to dictionaries in Python and hashes in Perl. One of the main differences is that these scripting languages allow using different types for the values contained in a single map, while in Ada, both the type of key and value are specified in the package instantiation and remains constant for that specific map. You can't have a map where two elements are of different types or two keys are of different types. If you want to use multiple types, you must create a different map for each and use only one type in each map.

When instantiating a hashed map from `Ada.Containers.Indefinite_Hashed_Maps`, we specify following elements:

- `Key_Type`: type of the key
- `Element_Type`: type of the element
- `Hash`: hash function for the `Key_Type`
- `Equivalent_Keys`: an equality operator (e.g. `=`) that indicates whether two keys are to be considered equal.
  - If the type specified in `Key_Type` has a standard operator, you can use it, which you do by specifying that operator as the value of `Equivalent_Keys`.

In the next example, we'll use a string as a key type. We'll use the `Hash` function provided by the standard library for strings (in the `Ada.Strings` package) and the standard equality operator.

You add elements to a hashed map by calling `Insert`. If an element is already contained in a map `M`, you can access it directly by using its key. For example, you can change the value of an element by calling `M ("My_Key") := 10`. If the key is not found, an exception is raised. To verify if a key is available, use the function `Contains` (as we've seen above in the section on sets).

Let's see an example:

Listing 295: `show_hashed_map.adb`

```
1 with Ada.Containers.Indefinite_Hashed_Maps;
2 with Ada.Strings.Hash;
3
4 with Ada.Text_IO; use Ada.Text_IO;
5
6 procedure Show_Hashed_Map is
7
8   package Integer_Hashed_Maps is new
9     Ada.Containers.Indefinite_Hashed_Maps
10    (Key_Type      => String,
11     Element_Type  => Integer,
12     Hash          => Ada.Strings.Hash,
13     Equivalent_Keys => "=");
14
15   use Integer_Hashed_Maps;
16
17   M : Map;
18   -- Same as:
```

(continues on next page)

(continued from previous page)

```

19  --
20  --  M : Integer_Hashed_Maps.Map;
21  begin
22    M.Include ("Alice", 24);
23    M.Include ("John", 40);
24    M.Include ("Bob", 28);
25
26    if M.Contains ("Alice") then
27      Put_Line ("Alice's age is "
28                & Integer'Image (M ("Alice")));
29    end if;
30
31    -- Update Alice's age
32    -- Key must already exist in M.
33    -- Otherwise an exception is raised.
34    M ("Alice") := 25;
35
36    New_Line; Put_Line ("Name & Age:");
37    for C in M.Iterate loop
38      Put_Line (Key (C) & ": "
39                & Integer'Image (M (C)));
40    end loop;
41
42  end Show_Hashed_Map;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Hashed\_Map  
MD5: 6117775bd9ce2b1466f448b100117ded

### Runtime output

Alice's age is 24

Name & Age:  
John: 40  
Bob: 28  
Alice: 25

### 19.3.2 Ordered maps

Ordered maps share many features with hashed maps. The main differences are:

- A hash function isn't needed. Instead, you must provide an ordering function (< operator), which the ordered map will use to order elements and allow fast access,  $O(\log N)$ , using a binary search.
  - If the type specified in Key\_Type has a standard < operator, you can use it in a similar way as we did for Equivalent\_Keys above for hashed maps.

Let's see an example:

Listing 296: show\_ordered\_map.adb

```

1  with Ada.Containers.Indefinite_Ordered_Maps;
2
3  with Ada.Text_IO; use Ada.Text_IO;
4
5  procedure Show_Ordered_Map is
6
7    package Integer_Ordered_Maps is new

```

(continues on next page)

(continued from previous page)

```
8  Ada.Containers.Indefinite_Ordered_Maps
9      (Key_Type      => String,
10       Element_Type  => Integer);
11
12  use Integer_Ordered_Maps;
13
14  M : Map;
15 begin
16      M.Include ("Alice", 24);
17      M.Include ("John", 40);
18      M.Include ("Bob", 28);
19
20      if M.Contains ("Alice") then
21          Put_Line ("Alice's age is "
22                     & Integer'Image (M ("Alice")));
23      end if;
24
25      -- Update Alice's age
26      -- Key must already exist in M
27      M ("Alice") := 25;
28
29      New_Line; Put_Line ("Name & Age:");
30      for C in M.Iterate loop
31          Put_Line (Key (C) & ": "
32                     & Integer'Image (M (C)));
33      end loop;
34
35 end Show_Ordered_Map;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Ordered\_Map  
MD5: 3deb3c685e767cee271b06e87727b086

### Runtime output

```
Alice's age is 24
```

```
Name & Age:
Alice: 25
Bob: 28
John: 40
```

You can see a great similarity between the examples above and from the previous section. In fact, since both kinds of maps share many operations, we didn't need to make extensive modifications when we changed our example to use ordered maps instead of hashed maps. The main difference is seen when we run the examples: the output of a hashed map is usually unordered, but the output of a ordered map is always ordered, as implied by its name.

### 19.3.3 Complexity

Hashed maps are generally the fastest data structure available to you in Ada if you need to associate heterogeneous keys to values and search for them quickly. In most cases, they are slightly faster than ordered maps. So if you don't need ordering, use hashed maps.

The Reference Manual requires the following average complexity of operations:

Operations	Ordered_Maps	Hashed_Maps
<ul style="list-style-type: none"><li>• Insert</li><li>• Include</li><li>• Replace</li><li>• Delete</li><li>• Exclude</li><li>• Find</li></ul>	$O((\log N)^2)$ or better	$O(\log N)$
Subprogram using cursor	$O(1)$	$O(1)$



## STANDARD LIBRARY: DATES & TIMES

The standard library supports processing of dates and times using two approaches:

- *Calendar* approach, which is suitable for handling dates and times in general;
- *Real-time* approach, which is better suited for real-time applications that require enhanced precision — for example, by having access to an absolute clock and handling time spans. Note that this approach only supports times, not dates.

The following sections present these two approaches.

### 20.1 Date and time handling

The Ada.Calendar package supports handling of dates and times. Let's look at a simple example:

Listing 297: display\_current\_time.adb

```
1 with Ada.Text_Io;  use Ada.Text_Io;
2 with Ada.Calendar; use Ada.Calendar;
3
4 with Ada.Calendar.Formatting;
5 use Ada.Calendar.Formatting;
6
7 procedure Display_Current_Time is
8     Now : Time := Clock;
9 begin
10     Put_Line ("Current time: " & Image (Now));
11 end Display_Current_Time;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Display\_Current\_Time  
MD5: 4a88069b33ecf80314b0164a472ff606

#### Runtime output

Current time: 2025-12-27 15:41:31

This example displays the current date and time, which is retrieved by a call to the `Clock` function. We call the function `Image` from the `Ada.Calendar.Formatting` package to get a **String** for the current date and time. We could instead retrieve each component using the `Split` function. For example:

Listing 298: display\_current\_year.adb

```
1 with Ada.Text_Io;  use Ada.Text_Io;
2 with Ada.Calendar; use Ada.Calendar;
```

(continues on next page)

(continued from previous page)

```
3
4  procedure Display_Current_Year is
5      Now          : Time := Clock;
6
7      Now_Year      : Year_Number;
8      Now_Month    : Month_Number;
9      Now_Day       : Day_Number;
10     Now_Seconds : Day_Duration;
11
12    begin
13        Split (Now,
14                  Now_Year,
15                  Now_Month,
16                  Now_Day,
17                  Now_Seconds);
18
19        Put_Line ("Current year is: "
20                  & Year_Number'Image (Now_Year));
21        Put_Line ("Current month is: "
22                  & Month_Number'Image (Now_Month));
23        Put_Line ("Current day   is: "
24                  & Day_Number'Image (Now_Day));
25    end Display_Current_Year;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Display\_Current\_Year  
MD5: fdf298ee97f225261ce3839ebd833bbe

### Runtime output

```
Current year  is: 2025
Current month is: 12
Current day   is: 27
```

Here, we're retrieving each element and displaying it separately.

#### 20.1.1 Delaying using date

You can delay an application so that it restarts at a specific date and time. We saw something similar in the chapter on tasking. You do this using a **delay until** statement. For example:

Listing 299: display\_delay\_next\_specific\_time.adb

```
1  with Ada.Text_IO;  use Ada.Text_IO;
2  with Ada.Calendar; use Ada.Calendar;
3
4  with Ada.Calendar.Formatting;
5  use Ada.Calendar.Formatting;
6
7  with Ada.Calendar.Time_Zones;
8  use Ada.Calendar.Time_Zones;
9
10 procedure Display_Delay_Next_Specific_Time is
11     TZ   : Time_Offset := UTC_Time_Offset;
12     Next : Time        :=
13         Ada.Calendar.Formatting.Time_Of
14         (Year      => 2018,
15          Month    => 5,
16          Day      => 1,
```

(continues on next page)

(continued from previous page)

```

17      Hour      => 15,
18      Minute    => 0,
19      Second    => 0,
20      Sub_Second => 0.0,
21      Leap_Second => False,
22      Time_Zone  => TZ);
23
24      -- Next = 2018-05-01 15:00:00.00
25      --          (local time-zone)
26 begin
27      Put_Line ("Let's wait until..."); 
28      Put_Line (Image (Next, True, TZ));
29
30      delay until Next;
31
32      Put_Line ("Enough waiting!");
33 end Display_Delay_Next_Specific_Time;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Display\_Delay\_Next\_Specific\_Time  
MD5: 36ec2bdce7c1e8d107fae54ef9852d3f

### Runtime output

```

Let's wait until...
2018-05-01 15:00:00.00
Enough waiting!

```

In this example, we specify the date and time by initializing `Next` using a call to `Time_0f`, a function taking the various components of a date (year, month, etc) and returning an element of the `Time` type. Because the date specified is in the past, the `delay until` statement won't produce any noticeable effect. However, if we passed a date in the future, the program would wait until that specific date and time arrived.

Here we're converting the time to the local timezone. If we don't specify a timezone, *Coordinated Universal Time* (abbreviated to UTC) is used by default. By retrieving the time offset to UTC with a call to `UTC_Time_Offset` from the `Ada.Calendar.Time_Zones` package, we can initialize `TZ` and use it in the call to `Time_0f`. This is all we need do to make the information provided to `Time_0f` relative to the local time zone.

We could achieve a similar result by initializing `Next` with a `String`. We can do this with a call to `Value` from the `Ada.Calendar.Formatting` package. This is the modified code:

Listing 300: `display_delay_next_specific_time.adb`

```

1  with Ada.Text_Io;  use Ada.Text_Io;
2  with Ada.Calendar; use Ada.Calendar;
3
4  with Ada.Calendar.Formatting;
5  use Ada.Calendar.Formatting;
6
7  with Ada.Calendar.Time_Zones;
8  use Ada.Calendar.Time_Zones;
9
10 procedure Display_Delay_Next_Specific_Time is
11     TZ : Time_Offset := UTC_Time_Offset;
12     Next : Time      :=
13         Ada.Calendar.Formatting.Value
14         ("2018-05-01 15:00:00.00", TZ);
15

```

(continues on next page)

(continued from previous page)

```
16  -- Next = 2018-05-01 15:00:00.00
17  -- (local time-zone)
18 begin
19  Put_Line ("Let's wait until...");
20  Put_Line (Image (Next, True, TZ));
21
22  delay until Next;
23
24  Put_Line ("Enough waiting!");
25 end Display_Delay_Next_Specific_Time;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Display\_Delay\_Next\_Specific\_Time  
MD5: fdf6ad7fca303d4d7bd444c23e11c7bd

### Runtime output

```
Let's wait until...
2018-05-01 15:00:00.00
Enough waiting!
```

In this example, we're again using TZ in the call to Value to adjust the input time to the current time zone.

In the examples above, we were delaying to a specific date and time. Just like we saw in the tasking chapter, we could instead specify the delay relative to the current time. For example, we could delay by 5 seconds, using the current time:

Listing 301: display\_delay\_next.adb

```
1 with Ada.Calendar; use Ada.Calendar;
2 with Ada.Text_IO;  use Ada.Text_IO;
3
4 procedure Display_Delay_Next is
5   D : Duration := 5.0;
6   --          ^ seconds
7   Now : Time    := Clock;
8   Next : Time   := Now + D;
9   --          ^ use duration to
10  --          specify next
11  --          point in time
12 begin
13   Put_Line ("Let's wait "
14     & Duration'Image (D)
15     & " seconds...");
16   delay until Next;
17   Put_Line ("Enough waiting!");
18 end Display_Delay_Next;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Display\_Delay\_Next
MD5: 58360d93388c3fe027c3d9d67389efc7

### Runtime output

```
Let's wait 5.000000000 seconds...
Enough waiting!
```

Here, we're specifying a duration of 5 seconds in D, adding it to the current time from Now, and storing the sum in Next. We then use it in the `delay until` statement.

## 20.2 Real-time

In addition to Ada.Calendar, the standard library also supports time operations for real-time applications. These are included in the Ada.Real\_Time package. This package also include a Time type. However, in the Ada.Real\_Time package, the Time type is used to represent an absolute clock and handle a time span. This contrasts with the Ada.Calendar, which uses the Time type to represent dates and times.

In the previous section, we used the Time type from the Ada.Calendar and the **delay until** statement to delay an application by 5 seconds. We could have used the Ada.Real\_Time package instead. Let's modify that example:

Listing 302: display\_delay\_next\_real\_time.adb

```

1  with Ada.Text_IO;    use Ada.Text_IO;
2  with Ada.Real_Time; use Ada.Real_Time;
3
4  procedure Display_Delay_Next_Real_Time is
5      D      : Time_Span := Seconds (5);
6      Next   : Time      := Clock + D;
7  begin
8      Put_Line ("Let's wait "
9                  & Duration'Image (To_Duration (D))
10                 & " seconds...");
11      delay until Next;
12      Put_Line ("Enough waiting!");
13  end Display_Delay_Next_Real_Time;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Display\_Delay\_Next\_Real\_Time  
MD5: a80e96c4ac7bd3ba7813f983b10cb038

### Runtime output

Let's wait 5.000000000 seconds...  
Enough waiting!

The main difference is that D is now a variable of type Time\_Span, defined in the Ada.Real\_Time package. We call the function Seconds to initialize D, but could have gotten a finer granularity by calling Nanoseconds instead. Also, we need to first convert D to the Duration type using To\_Duration before we can display it.

### 20.2.1 Benchmarking

One interesting application using the Ada.Real\_Time package is benchmarking. We've used that package before in a previous section when discussing tasking. Let's look at an example of benchmarking:

Listing 303: display\_benchmarking.adb

```

1  with Ada.Text_IO;    use Ada.Text_IO;
2  with Ada.Real_Time; use Ada.Real_Time;
3
4  procedure Display_Benchmarking is
5
6      procedure Computational_Intensive_App is
7      begin
8          delay 5.0;
9      end Computational_Intensive_App;

```

(continues on next page)

(continued from previous page)

```
10      Start_Time, Stop_Time : Time;
11      Elapsed_Time          : Time_Span;
12
13  begin
14      Start_Time := Clock;
15
16      Computational_Intensive_App;
17
18      Stop_Time    := Clock;
19      Elapsed_Time := Stop_Time - Start_Time;
20
21      Put_Line ( "Elapsed time: "
22                  & Duration'Image
23                  (To_Duration (Elapsed_Time))
24                  & " seconds");
25
26  end Display_Benchmarking;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Display\_Benchmarking  
MD5: 4b20940cb613d3f634be5224f409efeb

### Runtime output

```
Elapsed time: 5.002554479 seconds
```

This example defines a dummy Computational\_Intensive\_App implemented using a simple **delay** statement. We initialize Start\_Time and Stop\_Time from the then-current clock and calculate the elapsed time. By running this program, we see that the time is roughly 5 seconds, which is expected due to the **delay** statement.

A similar application is benchmarking of CPU time. We can implement this using the Execution\_Time package. Let's modify the previous example to measure CPU time:

Listing 304: display\_benchmarking\_cpu\_time.adb

```
1  with Ada.Text_IO;           use Ada.Text_IO;
2  with Ada.Real_Time;         use Ada.Real_Time;
3  with Ada.Execution_Time;   use Ada.Execution_Time;
4
5  procedure Display_Benchmarking_CPU_Time is
6
7      procedure Computational_Intensive_App is
8      begin
9          delay 5.0;
10         end Computational_Intensive_App;
11
12     Start_Time, Stop_Time : CPU_Time;
13     Elapsed_Time          : Time_Span;
14
15  begin
16      Start_Time := Clock;
17
18      Computational_Intensive_App;
19
20      Stop_Time    := Clock;
21      Elapsed_Time := Stop_Time - Start_Time;
22
23      Put_Line ( "CPU time: "
24                  & Duration'Image
```

(continues on next page)

(continued from previous page)

```

25      (To_Duration (Elapsed_Time))
26      & " seconds");
27 end Display_Benchmarking_CPU_Time;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Display\_Benchmarking\_CPU\_Time  
MD5: ba83ddbd05db523479be5692c4134901

### Runtime output

CPU time: 0.000050028 seconds

In this example, `Start_Time` and `Stop_Time` are of type `CPU_Time` instead of `Time`. However, we still call the `Clock` function to initialize both variables and calculate the elapsed time in the same way as before. By running this program, we see that the CPU time is significantly lower than the 5 seconds we've seen before. This is because the `delay` statement doesn't require much CPU time. The results will be different if we change the implementation of `Computational_Intensive_App` to use a mathematical function in a long loop. For example:

Listing 305: `display_benchmarking_math.adb`

```

1  with Ada.Text_IO;           use Ada.Text_IO;
2  with Ada.Real_Time;         use Ada.Real_Time;
3  with Ada.Execution_Time;   use Ada.Execution_Time;
4
5  with Ada.Numerics.Generic_Elementary_Functions;
6
7  procedure Display_Benchmarking_Math is
8
9    procedure Computational_Intensive_App is
10      package Funcs is new
11        Ada.Numerics.Generic_Elementary_Functions
12        (Float_Type => Long_Long_Float);
13      use Funcs;
14
15      X : Long_Long_Float;
16    begin
17      for I in 0 .. 1_000_000 loop
18        X := Tan (Arctan
19                  (Tan (Arctan
20                  (Tan (Arctan
21                  (Tan (Arctan
22                  (Tan (Arctan
23                  (Tan (Arctan
24                  (0.577))))))))));
25    end loop;
26  end Computational_Intensive_App;
27
28  procedure Benchm_Elapsed_Time is
29    Start_Time, Stop_Time : Time;
30    Elapsed_Time          : Time_Span;
31
32  begin
33    Start_Time := Clock;
34
35    Computational_Intensive_App;
36
37    Stop_Time    := Clock;
38    Elapsed_Time := Stop_Time - Start_Time;

```

(continues on next page)

(continued from previous page)

```
39      Put_Line ("Elapsed time: "
40                  & Duration'Image
41                  (To_Duration (Elapsed_Time))
42                  & " seconds");
43  end Benchm_Elapsed_Time;
44
45
46  procedure Benchm_CPU_Time is
47      Start_Time, Stop_Time : CPU_Time;
48      Elapsed_Time          : Time_Span;
49
50  begin
51      Start_Time := Clock;
52
53      Computational_Intensive_App;
54
55      Stop_Time    := Clock;
56      Elapsed_Time := Stop_Time - Start_Time;
57
58      Put_Line ("CPU time: "
59                  & Duration'Image
60                  (To_Duration (Elapsed_Time))
61                  & " seconds");
62  end Benchm_CPU_Time;
63 begin
64  Benchm_Elapsed_Time;
65  Benchm_CPU_Time;
66 end Display_Benchmarking_Math;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Display\_Benchmarking\_Math  
MD5: 06fe96bf03321c248dd1ed843648cf0b

### Runtime output

```
Elapsed time: 0.574971306 seconds
CPU time: 0.574500344 seconds
```

Now that our dummy Computational\_Intensive\_App involves mathematical operations requiring significant CPU time, the measured elapsed and CPU time are much closer to each other than before.

## STANDARD LIBRARY: STRINGS

In previous chapters, we've seen source-code examples using the **String** type, which is a fixed-length string type — essentially, it's an array of characters. In many cases, this data type is good enough to deal with textual information. However, there are situations that require more advanced text processing. Ada offers alternative approaches for these cases:

- *Bounded strings*: similar to fixed-length strings, bounded strings have a maximum length, which is set at its instantiation. However, bounded strings are not arrays of characters. At any time, they can contain a string of varied length — provided this length is below or equal to the maximum length.
- *Unbounded strings*: similar to bounded strings, unbounded strings can contain strings of varied length. However, in addition to that, they don't require a maximum length to be specified at the declaration of a string. In this sense, they are very flexible.

### For further reading...

Although we don't specify a maximum length for unbounded strings, the limit is defined by the Reference Manual<sup>20</sup>:

An object of type `Unbounded_String` represents a **String** whose low bound is 1 and whose length can vary conceptually between 0 and `Natural'Last`.

Therefore, the implicit maximum length is `Natural'Last`. In contrast, bounded strings have an explicit maximum length that is specified when the `Generic_Bounded_Length` package is instantiated (as we'll see *later on* (page 242)).

Another difference between bounded and unbounded strings is the strategy that is used by the compiler to allocate memory for those strings. When using GNAT, bounded strings are allocated on the stack, while unbounded strings are allocated on the heap.

The following sections present an overview of the different string types and common operations for string types.

### 21.1 String operations

Operations on standard (fixed-length) strings are available in the `Ada.Strings.Fixed` package. As mentioned previously, standard strings are arrays of elements of **Character** type with a *fixed-length*. That's why this child package is called `Fixed`.

One of the simplest operations provided is counting the number of substrings available in a string (**Count**) and finding their corresponding indices (**Index**). Let's look at an example:

<sup>20</sup> <http://www.adacore.com/standards/12rm/html/RM-A-4-5.html>

Listing 306: show\_find\_substring.adb

```
1 with Ada.Strings.Fixed; use Ada.Strings.Fixed;
2 with Ada.Text_Io;      use Ada.Text_Io;
3
4 procedure Show_Find_Substring is
5
6   S   : String := "Hello" & 3 * "World";
7   P   : constant String := "World";
8   Idx : Natural;
9   Cnt : Natural;
10 begin
11   Cnt := Ada.Strings.Fixed.Count
12     (Source => S,
13      Pattern => P);
14
15   Put_Line ("String: " & S);
16   Put_Line ("Count for '" & P & "' is "
17             & Natural'Image (Cnt));
18
19   Idx := 0;
20   for I in 1 .. Cnt loop
21     Idx := Index
22       (Source => S,
23        Pattern => P,
24        From    => Idx + 1);
25
26     Put_Line ("Found instance of '"
27               & P & "' at position: "
28               & Natural'Image (Idx));
29   end loop;
30
31 end Show_Find_Substring;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Find\_Substring  
MD5: faa8373bf9aec9f9f5507cf55590b0c0

### Runtime output

```
String: Hello World World World
Count for 'World': 3
Found instance of 'World' at position: 7
Found instance of 'World' at position: 13
Found instance of 'World' at position: 19
```

We initialize the string S using a multiplication. Writing `"Hello" & 3 * "World"` creates the string Hello World World World. We then call the function Count to get the number of instances of the word World in S. Next we call the function Index in a loop to find the index of each instance of World in S.

That example looked for instances of a specific substring. In the next example, we retrieve all the words in the string. We do this using `Find_Token` and specifying whitespaces as separators. For example:

Listing 307: show\_find\_words.adb

```
1 with Ada.Strings;      use Ada.Strings;
2 with Ada.Strings.Fixed; use Ada.Strings.Fixed;
3 with Ada.Strings.Maps;  use Ada.Strings.Maps;
4 with Ada.Text_Io;       use Ada.Text_Io;
```

(continues on next page)

(continued from previous page)

```

5  procedure Show_Find_Words is
6
7      S  : String := "Hello" & 3 * "World";
8      F  : Positive;
9      L  : Natural;
10     I  : Natural := 1;
11
12     Whitespace : constant Character_Set :=
13         To_Set (' ');
14
15 begin
16     Put_Line ("String: " & S);
17     Put_Line ("String length: "
18             & Integer'Image (S'Length));
19
20     while I in S'Range loop
21         Find_Token
22             (Source  => S,
23              Set     => Whitespace,
24              From    => I,
25              Test    => Outside,
26              First   => F,
27              Last    => L);
28
29     exit when L = 0;
30
31     Put_Line ("Found word instance at position "
32             & Natural'Image (F)
33             & ":" & S (F .. L) & ")");
34     --  & "-" & F'Img & "-" & L'Img
35
36     I := L + 1;
37 end loop;
38 end Show_Find_Words;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Find\_Words  
MD5: e622f489af5901e5d31f314efc3324d2

### Runtime output

```

String: Hello World World World
String length: 23
Found word instance at position 1: 'Hello'
Found word instance at position 7: 'World'
Found word instance at position 13: 'World'
Found word instance at position 19: 'World'

```

We pass a set of characters to be used as delimitators to the procedure `Find_Token`. This set is a member of the `Character_Set` type from the `Ada.Strings.Maps` package. We call the `To_Set` function (from the same package) to initialize the set to `Whitespace` and then call `Find_Token` to loop over each valid index and find the starting index of each word. We pass `Outside` to the `Test` parameter of the `Find_Token` procedure to indicate that we're looking for indices that are outside the `Whitespace` set, i.e. actual words. The `First` and `Last` parameters of `Find_Token` are output parameters that indicate the valid range of the substring. We use this information to display the string (`S (F .. L)`).

The operations we've looked at so far read strings, but don't modify them. We next discuss operations that change the content of strings:

Operation	Description
Insert	Insert substring in a string
Overwrite	Overwrite a string with a substring
Delete	Delete a substring
Trim	Remove whitespaces from a string

All these operations are available both as functions or procedures. Functions create a new string but procedures perform the operations in place. The procedure will raise an exception if the constraints of the string are not satisfied. For example, if we have a string *S* containing 10 characters, inserting a string with two characters (e.g. "!!") into it produces a string containing 12 characters. Since it has a fixed length, we can't increase its size. One possible solution in this case is to specify that truncation should be applied while inserting the substring. This keeps the length of *S* fixed. Let's see an example that makes use of both function and procedure versions of Insert, Overwrite, and Delete:

Listing 308: show\_adapted\_strings.adb

```

1  with Ada.Strings;      use Ada.Strings;
2  with Ada.Strings.Fixed; use Ada.Strings.Fixed;
3  with Ada.Text_IO;       use Ada.Text_IO;
4
5  procedure Show_Adapted_Strings is
6
7      S  : String := "Hello World";
8      P  : constant String := "World";
9      N  : constant String := "Beautiful";
10
11     procedure Display_Adapted_String
12         (Source  : String;
13          Before   : Positive;
14          New_Item : String;
15          Pattern  : String)
16     is
17         S_Ins_In : String := Source;
18         S_Ovr_In : String := Source;
19         S_Del_In : String := Source;
20
21         S_Ins : String :=
22             Insert (Source,
23                     Before,
24                     New_Item & " ");
25         S_Ovr : String :=
26             Overwrite (Source,
27                         Before,
28                         New_Item);
29         S_Del : String :=
30             Trim (Delete (Source,
31                           Before,
32                           Before +
33                           Pattern'Length - 1),
34                           Ada.Strings.Right);
35
36     begin
37         Insert (S_Ins_In,
38                 Before,
39                 New_Item,
40                 Right);
41
42         Overwrite (S_Ovr_In,
43                     Before,
44                     New_Item);

```

(continues on next page)

(continued from previous page)

```

43         New_Item,
44         Right);
45
46     Delete (S_Del_In,
47             Before,
48             Before + Pattern'Length - 1);
49
50     Put_Line ("Original:  ''"
51               & Source & "'");
52
53     Put_Line ("Insert:    ''"
54               & S_Ins & "'");
55     Put_Line ("Overwrite: ''"
56               & S_Ovr & "'");
57     Put_Line ("Delete:    ''"
58               & S_Del & "'");
59
60     Put_Line ("Insert  (in-place):  ''"
61               & S_Ins_In & "'");
62     Put_Line ("Overwrite (in-place):  ''"
63               & S_Ovr_In & "'");
64     Put_Line ("Delete   (in-place):  ''"
65               & S_Del_In & "'");
66 end Display_Adapted_String;
67
68
69  Idx : Natural;
70 begin
71   Idx := Index
72     (Source => S,
73      Pattern => P);
74
75   if Idx > 0 then
76     Display_Adapted_String (S, Idx, N, P);
77   end if;
78 end Show_Adapted.Strings;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Adapted.Strings  
MD5: b31b6bc94d8bdbec717c6b6b2534beb6

### Runtime output

```

Original: 'Hello World'
Insert:  'Hello Beautiful World'
Overwrite: 'Hello Beautiful'
Delete:  'Hello'
Insert  (in-place): 'Hello Beaut'
Overwrite (in-place): 'Hello Beaut'
Delete   (in-place): 'Hello  '

```

In this example, we look for the index of the substring `World` and perform operations on this substring within the outer string. The procedure `Display_Adapted_String` uses both versions of the operations. For the procedural version of `Insert` and `Overwrite`, we apply truncation to the right side of the string (`Right`). For the `Delete` procedure, we specify the range of the substring, which is replaced by whitespaces. For the function version of `Delete`, we also call `Trim` which trims the trailing whitespace.

## 21.2 Limitation of fixed-length strings

Using fixed-length strings is usually good enough for strings that are initialized when they are declared. However, as seen in the previous section, procedural operations on strings cause difficulties when done on fixed-length strings because fixed-length strings are arrays of characters. The following example shows how cumbersome the initialization of fixed-length strings can be when it's not performed in the declaration:

Listing 309: show\_char\_array.adb

```
1  with Ada.Text_IO;          use Ada.Text_IO;
2
3  procedure Show_Char_Array is
4      S : String (1 .. 15);
5      -- Strings are arrays of Character
6  begin
7      S := "Hello          ";
8      -- Alternatively:
9      --
10     -- #1:
11     --     S (1 .. 5)      := "Hello";
12     --     S (6 .. S'Last) := (others => ' ');
13     --
14     -- #2:
15     --     S := ('H', 'e', 'l', 'l', 'o',
16     --             others => ' ');
17
18     Put_Line ("String: " & S);
19     Put_Line ("String Length: "
20               & Integer'Image (S'Length));
21 end Show_Char_Array;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Char\_Array  
MD5: 9f3df03c9c5336184139cf2a22f2cb7e

### Runtime output

```
String: Hello
String Length: 15
```

In this case, we can't simply write `S := "Hello"` because the resulting array of characters for the `Hello` constant has a different length than the `S` string. Therefore, we need to include trailing whitespaces to match the length of `S`. As shown in the example, we could use an exact range for the initialization (`S (1 .. 5)`) or use an explicit array of individual characters.

When strings are initialized or manipulated at run-time, it's usually better to use bounded or unbounded strings. An important feature of these types is that they aren't arrays, so the difficulties presented above don't apply. Let's start with bounded strings.

## 21.3 Bounded strings

Bounded strings are defined in the `Ada.Strings.Bounded.Generic_Bounded_Length` package. Because this is a generic package, you need to instantiate it and set the maximum length of the bounded string. You can then declare bounded strings of the `Bounded_String` type.

Both bounded and fixed-length strings have a maximum length that they can hold. How-

ever, bounded strings are not arrays, so initializing them at run-time is much easier. For example:

Listing 310: show\_bounded\_string.adb

```

1  with Ada.Strings;           use Ada.Strings;
2  with Ada.Strings.Bounded;
3  with Ada.Text_IO;           use Ada.Text_IO;
4
5  procedure Show_Bounded_String is
6    package B_Str is new
7      Ada.Strings.Bounded.Generic_Bounded_Length
8      (Max => 15);
9    use B_Str;
10
11   S1, S2 : Bounded_String;
12
13  procedure Display_String_Info
14    (S : Bounded_String)
15  is
16  begin
17    Put_Line ("String: " & To_String (S));
18    Put_Line ("String Length: "
19              & Integer'Image (Length (S)));
20    -- String:
21    --      S'Length => ok
22    -- Bounded_String:
23    --      S'Length => compilation error:
24    --                      bounded strings are
25    --                      not arrays!
26
27    Put_Line ("Max. Length: "
28              & Integer'Image (Max_Length));
29  end Display_String_Info;
30
31 begin
32   S1 := To_Bounded_String ("Hello");
33   Display_String_Info (S1);
34
35   S2 := To_Bounded_String ("Hello World");
36   Display_String_Info (S2);
37
38   S1 := To_Bounded_String
39     ("Something longer to say here...",
40      Right);
41   Display_String_Info (S1);
42 end Show_Bounded_String;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Bounded\_String  
MD5: a51fdeacfd43923145ee92bf5c72ecd6

### Runtime output

```

String: Hello
String Length: 5
Max. Length: 15
String: Hello World
String Length: 11
Max. Length: 15
String: Something longer

```

(continues on next page)

(continued from previous page)

```
String Length: 15
Max. Length: 15
```

By using bounded strings, we can easily assign to S1 and S2 multiple times during execution. We use the `To_Bounded_String` and `To_String` functions to convert, in the respective direction, between fixed-length and bounded strings. A call to `To_Bounded_String` raises an exception if the length of the input string is greater than the maximum capacity of the bounded string. To avoid this, we can use the truncation parameter (`Right` in our example).

Bounded strings are not arrays, so we can't use the '`Length`' attribute as we did for fixed-length strings. Instead, we call the `Length` function, which returns the length of the bounded string. The `Max_Length` constant represents the maximum length of the bounded string that we set when we instantiated the package.

After initializing a bounded string, we can manipulate it. For example, we can append a string to a bounded string using `Append` or concatenate bounded strings using the `&` operator. Like so:

Listing 311: `show_bounded_string_op.adb`

```
1  with Ada.Strings;           use Ada.Strings;
2  with Ada.Strings.Bounded;
3  with Ada.Text_IO;           use Ada.Text_IO;
4
5  procedure Show_Bounded_String_Op is
6    package B_Str is new
7      Ada.Strings.Bounded.Generic_Bounded_Length
8      (Max => 30);
9    use B_Str;
10
11   S1, S2 : Bounded_String;
12
13   S1 := To_Bounded_String ("Hello");
14   -- Alternatively:
15   --
16   -- A := Null_Bounded_String & "Hello";
17
18   Append (S1, " World");
19   -- Alternatively:
20   -- Append (A, " World", Right);
21
22   Put_Line ("String: " & To_String (S1));
23
24   S2 := To_Bounded_String ("Hello!");
25   S1 := S1 & " " & S2;
26   Put_Line ("String: " & To_String (S1));
27 end Show_Bounded_String_Op;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Bounded\_String\_Op  
MD5: c7c6a840c314a9cd9f75aac082a63159

### Runtime output

```
String: Hello World
String: Hello World Hello!
```

We can initialize a bounded string with an empty string using the `Null_Bounded_String` constant. Also, we can use the `Append` procedure and specify the truncation mode like we do with the `To_Bounded_String` function.

## 21.4 Unbounded strings

Unbounded strings are defined in the `Ada.Strings.Unbounded` package. This is *not* a generic package, so we don't need to instantiate it before using the `Unbounded_String` type. As you may recall from the previous section, bounded strings require a package instantiation.

Unbounded strings are similar to bounded strings. The main difference is that they can hold strings of any size and adjust according to the input string: if we assign, e.g., a 10-character string to an unbounded string and later assign a 50-character string, internal operations in the container ensure that memory is allocated to store the new string. In most cases, developers don't need to worry about these operations. Also, no truncation is necessary.

Initialization of unbounded strings is very similar to bounded strings. Let's look at an example:

Listing 312: `show_unbounded_string.adb`

```

1  with Ada.Text_Io; use Ada.Text_Io;
2  with Ada.Strings; use Ada.Strings;
3
4  with Ada.Strings.Unbounded;
5  use Ada.Strings.Unbounded;
6
7  procedure Show_Unbounded_String is
8    S1, S2 : Unbounded_String;
9
10   procedure Display_String_Info
11     (S : Unbounded_String)
12   is
13   begin
14     Put_Line ("String: " & To_String (S));
15     Put_Line ("String Length: "
16               & Integer'Image (Length (S)));
17   end Display_String_Info;
18 begin
19   S1 := To_Unbounded_String ("Hello");
20   -- Alternatively:
21   --
22   -- A := Null_Unbounded_String & "Hello";
23
24   Display_String_Info (S1);
25
26   S2 := To_Unbounded_String ("Hello World");
27   Display_String_Info (S2);
28
29   S1 := To_Unbounded_String
30         ("Something longer to say here... ");
31   Display_String_Info (S1);
32 end Show_Unbounded_String;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Unbounded\_String  
MD5: 904402992c96eb393b875d1b7cf49c1b

### Runtime output

```

String: Hello
String Length: 5
String: Hello World

```

(continues on next page)

(continued from previous page)

```
String Length: 11
String: Something longer to say here...
String Length: 31
```

Like bounded strings, we can assign to S1 and S2 multiple times during execution and use the To\_Unbounded\_String and To\_String functions to convert back-and-forth between fixed-length strings and unbounded strings. However, in this case, truncation is not needed.

And, just like for bounded strings, you can use the Append procedure and the & operator for unbounded strings. For example:

Listing 313: show\_unbounded\_string\_op.adb

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  with Ada.Strings.Unbounded;
4  use Ada.Strings.Unbounded;
5
6  procedure Show_Unbounded_String_Op is
7      S1, S2 : Unbounded_String := Null_Unbounded_String;
8
9  begin
10     S1 := S1 & "Hello";
11     S2 := S2 & "Hello!";
12
13     Append (S1, " World");
14     Put_Line ("String: " & To_String (S1));
15
16     S1 := S1 & " " & S2;
17     Put_Line ("String: " & To_String (S1));
18 end Show_Unbounded_String_Op;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Unbounded\_String\_Op  
MD5: 806e24a6dd0bc87e76f73a22e42ba390

### Runtime output

```
String: Hello World
String: Hello World Hello!
```

In this example, we're concatenating the unbounded S1 and S2 strings with the "Hello" and "Hello!" strings, respectively. Also, we're using the Append procedure, just like we did with bounded strings.

## STANDARD LIBRARY: FILES AND STREAMS

Ada provides different approaches for file input/output (I/O):

- *Text I/O*, which supports file I/O in text format, including the display of information on the console.
- *Sequential I/O*, which supports file I/O in binary format written in a sequential fashion for a specific data type.
- *Direct I/O*, which supports file I/O in binary format for a specific data type, but also supporting access to any position of a file.
- *Stream I/O*, which supports I/O of information for multiple data types, including objects of unbounded types, using files in binary format.

This table presents a summary of the features we've just seen:

File I/O option	Format	Random access	Data types
Text I/O	text		string type
Sequential I/O	binary		single type
Direct I/O	binary	✓	single type
Stream I/O	binary	✓	multiple types

In the following sections, we discuss details about these I/O approaches.

### 22.1 Text I/O

In most parts of this course, we used the `Put_Line` procedure to display information on the console. However, this procedure also accepts a **File\_Type** parameter. For example, you can select between standard output and standard error by setting this parameter explicitly:

Listing 314: `show_std_text_out.adb`

```
1 with Ada.Text_IO; use Ada.Text_IO;
2
3 procedure Show_Std_Text_Out is
4 begin
5   Put_Line (Standard_Output, "Hello World #1");
6   Put_Line (Standard_Error, "Hello World #2");
7 end Show_Std_Text_Out;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Std\_Text\_Out  
MD5: 4d75bd2906226897244e3d2a611c9725

#### Runtime output

```
Hello World #1
Hello World #2
```

You can also use this parameter to write information to any text file. To create a new file for writing, use the `Create` procedure, which initializes a `File_Type` element that you can later pass to `Put_Line` (instead of, e.g., `Standard_Output`). After you finish writing information, you can close the file by calling the `Close` procedure.

You use a similar method to read information from a text file. However, when opening the file, you must specify that it's an input file (`In_File`) instead of an output file. Also, instead of calling the `Put_Line` procedure, you call the `Get_Line` function to read information from the file.

Let's see an example that writes information into a new text file and then reads it back from the same file:

Listing 315: `show_simple_text_file_io.adb`

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Simple_Text_File_IO is
4      F      : File_Type;
5      File_Name : constant String := "simple.txt";
6  begin
7      Create (F, Out_File, File_Name);
8      Put_Line (F, "Hello World #1");
9      Put_Line (F, "Hello World #2");
10     Put_Line (F, "Hello World #3");
11     Close (F);
12
13     Open (F, In_File, File_Name);
14     while not End_Of_File (F) loop
15         Put_Line (Get_Line (F));
16     end loop;
17     Close (F);
18 end Show_Simple_Text_File_IO;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Simple\_Text\_File\_IO  
MD5: 7461e946eef18c93219fa4ce3afbb1ea

### Runtime output

```
Hello World #1
Hello World #2
Hello World #3
```

In addition to the `Create` and `Close` procedures, the standard library also includes a `Reset` procedure, which, as the name implies, resets (erases) all the information from the file. For example:

Listing 316: `show_text_file_reset.adb`

```
1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Text_File_Reset is
4      F      : File_Type;
5      File_Name : constant String := "simple.txt";
6  begin
7      Create (F, Out_File, File_Name);
8      Put_Line (F, "Hello World #1");
```

(continues on next page)

(continued from previous page)

```

9  Reset (F);
10 Put_Line (F, "Hello World #2");
11 Close (F);
12
13 Open (F, In_File, File_Name);
14 while not End_Of_File (F) loop
15   Put_Line (Get_Line (F));
16 end loop;
17 Close (F);
18 end Show_Text_File_Reset;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Text\_File\_Reset  
MD5: 5e5498f03b2c829513af062c5959fc93

### Runtime output

Hello World #2

By running this program, we notice that, although we've written the first string ("Hello World #1") to the file, it has been erased because of the call to Reset.

In addition to opening a file for reading or writing, you can also open an existing file and append to it. Do this by calling the Open procedure with the Append\_File option.

When calling the Open procedure, an exception is raised if the specified file isn't found. Therefore, you should handle exceptions in that context. The following example deletes a file and then tries to open the same file for reading:

Listing 317: show\_text\_file\_input\_except.adb

```

1  with Ada.Text_IO; use Ada.Text_IO;
2
3  procedure Show_Text_File_Input_Except is
4    F        : File_Type;
5    File_Name : constant String := "simple.txt";
6  begin
7    -- Open output file and delete it
8    Create (F, Out_File, File_Name);
9    Delete (F);
10
11   -- Try to open deleted file
12   Open (F, In_File, File_Name);
13   Close (F);
14 exception
15   when Name_Error =>
16     Put_Line ("File does not exist");
17   when others =>
18     Put_Line
19       ("Error while processing input file");
20 end Show_Text_File_Input_Except;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Text\_File\_Input\_Except  
MD5: c8d257091831c48d10b6e70e34b4261b

### Runtime output

File does not exist

In this example, we create the file by calling `Create` and then delete it by calling `Delete`. After the call to `Delete`, we can no longer use the `File_Type` element. After deleting the file, we try to open the non-existent file, which raises a `Name_Error` exception.

## 22.2 Sequential I/O

The previous section presented details about text file I/O. Here, we discuss doing file I/O in binary format. The first package we'll explore is the `Ada.Sequential_IO` package. Because this package is a generic package, you need to instantiate it for the data type you want to use for file I/O. Once you've done that, you can use the same procedures we've seen in the previous section: `Create`, `Open`, `Close`, `Reset` and `Delete`. However, instead of calling the `Get_Line` and `Put_Line` procedures, you'd call the `Read` and `Write` procedures.

In the following example, we instantiate the `Ada.Sequential_IO` package for floating-point types:

Listing 318: `show_seq_float_io.adb`

```
1  with Ada.Text_IO;
2  with Ada.Sequential_IO;
3
4  procedure Show_Seq_Float_IO is
5    package Float_IO is
6      new Ada.Sequential_IO (Float);
7    use Float_IO;
8
9    F      : Float_IO.File_Type;
10   File_Name : constant String :=
11     "float_file.bin";
12
13  begin
14    Create (F, Out_File, File_Name);
15    Write (F, 1.5);
16    Write (F, 2.4);
17    Write (F, 6.7);
18    Close (F);
19
20    declare
21      Value : Float;
22    begin
23      Open (F, In_File, File_Name);
24      while not End_Of_File (F) loop
25        Read (F, Value);
26        Ada.Text_IO.Put_Line
27          (Float'Image (Value));
28      end loop;
29      Close (F);
30    end;
31  end Show_Seq_Float_IO;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Seq\_Float\_IO  
MD5: 27aa5daf92cba5df23fdc55c3578aa34

### Runtime output

```
1.50000E+00
2.40000E+00
6.70000E+00
```

We use the same approach to read and write complex information. The following example

uses a record that includes a Boolean and a floating-point value:

Listing 319: show\_seq\_rec\_io.adb

```

1  with Ada.Text_IO;
2  with Ada.Sequential_IO;
3
4  procedure Show_Seq_Rec_IO is
5    type Num_Info is record
6      Valid : Boolean := False;
7      Value : Float;
8    end record;
9
10 procedure Put_Line (N : Num_Info) is
11 begin
12   if N.Valid then
13     Ada.Text_IO.Put_Line
14     ("(ok,
15      & Float'Image (N.Value) & ")");
16   else
17     Ada.Text_IO.Put_Line
18     ("(not ok,  -----)");
19   end if;
20 end Put_Line;
21
22 package Num_Info_IO is new
23   Ada.Sequential_IO (Num_Info);
24 use Num_Info_IO;
25
26 F      : Num_Info_IO.File_Type;
27 File_Name : constant String :=
28   "float_file.bin";
29
30 begin
31   Create (F, Out_File, File_Name);
32   Write (F, (True, 1.5));
33   Write (F, (False, 2.4));
34   Write (F, (True, 6.7));
35   Close (F);
36
37 declare
38   Value : Num_Info;
39 begin
40   Open (F, In_File, File_Name);
41   while not End_Of_File (F) loop
42     Read (F, Value);
43     Put_Line (Value);
44   end loop;
45   Close (F);
46 end;
47 end Show_Seq_Rec_IO;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Seq\_Rec\_IO  
MD5: a88b1428cc50745dce0509087e74adb7

### Runtime output

```
(ok,      1.50000E+00)
(not ok,  -----)
(ok,      6.70000E+00)
```

As the example shows, we can use the same approach we used for floating-point types to

perform file I/O for this record. Once we instantiate the Ada.Sequential\_IO package for the record type, file I/O operations are performed the same way.

### 22.3 Direct I/O

Direct I/O is available in the Ada.Direct\_IO package. This mechanism is similar to the sequential I/O approach just presented, but allows us to access any position in the file. The package instantiation and most operations are very similar to sequential I/O. To rewrite the Show\_Seq\_Float\_IO application presented in the previous section to use the Ada.Direct\_IO package, we just need to replace the instances of the Ada.Sequential\_IO package by the Ada.Direct\_IO package. This is the new source code:

Listing 320: show\_dir\_float\_io.adb

```
1 with Ada.Text_IO;
2 with Ada.Direct_IO;
3
4 procedure Show_Dir_Float_IO is
5   package Float_IO is new Ada.Direct_IO (Float);
6   use Float_IO;
7
8   F      : Float_IO.File_Type;
9   File_Name : constant String :=
10      "float_file.bin";
11
12 begin
13   Create (F, Out_File, File_Name);
14   Write (F, 1.5);
15   Write (F, 2.4);
16   Write (F, 6.7);
17   Close (F);
18
19   declare
20     Value : Float;
21   begin
22     Open (F, In_File, File_Name);
23     while not End_Of_File (F) loop
24       Read (F, Value);
25       Ada.Text_IO.Put_Line
26         (Float'Image (Value));
27     end loop;
28     Close (F);
29   end;
30 end Show_Dir_Float_IO;
```

#### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Dir\_Float\_IO  
MD5: e4e5855976de44f53a821eb90dcbb206

#### Runtime output

```
1.50000E+00
2.40000E+00
6.70000E+00
```

Unlike sequential I/O, direct I/O allows you to access any position in the file. However, it doesn't offer an option to append information to a file. Instead, it provides an Inout\_File mode allowing reading and writing to a file via the same **File\_Type** element.

To access any position in the file, call the Set\_Index procedure to set the new position / index. You can use the Index function to retrieve the current index. Let's see an example:

Listing 321: show\_dir\_float\_in\_out\_file.adb

```

1  with Ada.Text_IO;
2  with Ada.Direct_IO;
3
4  procedure Show_Dir_Float_In_Out_File is
5    package Float_IO is new Ada.Direct_IO (Float);
6    use Float_IO;
7
8    F      : Float_IO.File_Type;
9    File_Name : constant String :=
10      "float_file.bin";
11
12  begin
13    -- Open file for input / output
14    Create (F, Inout_File, File_Name);
15    Write (F, 1.5);
16    Write (F, 2.4);
17    Write (F, 6.7);
18
19    -- Set index to previous position
20    -- and overwrite value
21    Set_Index (F, Index (F) - 1);
22    Write (F, 7.7);
23
24  declare
25    Value : Float;
26  begin
27    -- Set index to start of file
28    Set_Index (F, 1);
29
30    while not End_Of_File (F) loop
31      Read (F, Value);
32      Ada.Text_IO.Put_Line
33        (Float'Image (Value));
34    end loop;
35    Close (F);
36  end;
37 end Show_Dir_Float_In_Out_File;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Dir\_Float\_In\_Out\_File  
MD5: 17b83a16ab8fa30f07cf8a0bd54078a1

### Runtime output

```

1.50000E+00
2.40000E+00
7.70000E+00

```

By running this example, we see that the file contains 7.7, rather than the previous 6.7 that we wrote. We overwrote the value by changing the index to the previous position before doing another write.

In this example we used the `Inout_File` mode. Using that mode, we just changed the index back to the initial position before reading from the file (`Set_Index (F, 1)`) instead of closing the file and reopening it for reading.

## 22.4 Stream I/O

All the previous approaches for file I/O in binary format (sequential and direct I/O) are specific for a single data type (the one we instantiate them with). You can use these approaches to write objects of a single data type that may be an array or record (potentially with many fields), but if you need to create and process files that include different data types, or any objects of an unbounded type, these approaches are not sufficient. Instead, you should use stream I/O.

Stream I/O shares some similarities with the previous approaches. We still use the `Create`, `Open` and `Close` procedures. However, instead of accessing the file directly via a `File_Type` element, you use a `Stream_Access` element. To read and write information, you use the `'Read` or `'Write` attributes of the data types you're reading or writing.

Let's look at a version of the `Show_Dir_Float_I0` procedure from the previous section that makes use of stream I/O instead of direct I/O:

Listing 322: `show_float_stream.adb`

```

1  with Ada.Text_I0;
2
3  with Ada.Streams.Stream_I0;
4  use Ada.Streams.Stream_I0;
5
6  procedure Show_Float_Stream is
7      F      : File_Type;
8      S      : Stream_Access;
9      File_Name : constant String := "float_file.bin";
10
11 begin
12     Create (F, Out_File, File_Name);
13     S := Stream (F);
14
15     Float'Write (S, 1.5);
16     Float'Write (S, 2.4);
17     Float'Write (S, 6.7);
18
19     Close (F);
20
21  declare
22      Value : Float;
23  begin
24      Open (F, In_File, File_Name);
25      S := Stream (F);
26
27      while not End_Of_File (F) loop
28          Float'Read (S, Value);
29          Ada.Text_I0.Put_Line
30              (Float'Image (Value));
31      end loop;
32      Close (F);
33  end;
34 end Show_Float_Stream;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Float\_Stream  
MD5: 34ccf04b0821074a332019ac0e38bb3e

### Runtime output

```
1.50000E+00
2.40000E+00
6.70000E+00
```

After the call to `Create`, we retrieve the corresponding `Stream_Access` element by calling the `Stream` function. We then use this stream to write information to the file via the '`Write`' attribute of the `Float` type. After closing the file and reopening it for reading, we again retrieve the corresponding `Stream_Access` element and processed to read information from the file via the '`Read`' attribute of the `Float` type.

You can use streams to create and process files containing different data types within the same file. You can also read and write unbounded data types such as strings. However, when using unbounded data types you must call the '`Input`' and '`Output`' attributes of the unbounded data type: these attributes write information about bounds or discriminants in addition to the object's actual data.

The following example shows file I/O that mixes both strings of different lengths and floating-point values:

Listing 323: `show_string_stream.adb`

```
1  with Ada.Text_IO;
2
3  with Ada.Streams.Stream_IO;
4  use Ada.Streams.Stream_IO;
5
6  procedure Show_String_Stream is
7      F      : File_Type;
8      S      : Stream_Access;
9      File_Name : constant String :=
10        "float_file.bin";
11
12  procedure Output (S : Stream_Access;
13                    FV : Float;
14                    SV : String) is
15  begin
16      String'Output (S, SV);
17      Float'Output (S, FV);
18  end Output;
19
20  procedure Input_Display (S : Stream_Access) is
21      SV : String := String'Input (S);
22      FV : Float  := Float'Input (S);
23  begin
24      Ada.Text_IO.Put_Line (Float'Image (FV)
25                            & " --- " & SV);
26  end Input_Display;
27
28 begin
29     Create (F, Out_File, File_Name);
30     S := Stream (F);
31
32     Output (S, 1.5, "Hi!!!");
33     Output (S, 2.4, "Hello world!");
34     Output (S, 6.7, "Something longer here... ");
35
36     Close (F);
37
38     Open (F, In_File, File_Name);
39     S := Stream (F);
40
41     while not End_Of_File (F) loop
```

(continues on next page)

(continued from previous page)

```
42     Input_Display (S);
43   end loop;
44   Close (F);
45
46 end Show_String_Stream;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_String\_Stream  
MD5: 3ae8276ada5f24cab49994e368e0fa34

### Runtime output

```
1.50000E+00 --- Hi!!
2.40000E+00 --- Hello world!
6.70000E+00 --- Something longer here...
```

When you use Stream I/O, no information is written into the file indicating the type of the data that you wrote. If a file contains data from different types, you must reference types in the same order when reading a file as when you wrote it. If not, the information you get will be corrupted. Unfortunately, strong data typing doesn't help you in this case. Writing simple procedures for file I/O (as in the example above) may help ensuring that the file format is consistent.

Like direct I/O, stream I/O support also allows you to access any location in the file. However, when doing so, you need to be extremely careful that the position of the new index is consistent with the data types you're expecting.

## STANDARD LIBRARY: NUMERICS

The standard library provides support for common numeric operations on floating-point types as well as on complex types and matrices. In the sections below, we present a brief introduction to these numeric operations.

### 23.1 Elementary Functions

The `Ada.Numerics.Elementary_Functions` package provides common operations for floating-point types, such as square root, logarithm, and the trigonometric functions (e.g., `sin`, `cos`). For example:

Listing 324: `show_elem_math.adb`

```
1  with Ada.Text_Io;  use Ada.Text_Io;
2  with Ada.Numerics; use Ada.Numerics;
3
4  with Ada.Numerics.Elementary_Functions;
5  use Ada.Numerics.Elementary_Functions;
6
7  procedure Show_Elem_Math is
8      X : Float;
9  begin
10     X := 2.0;
11     Put_Line ("Square root of "
12               & Float'Image (X)
13               & " is "
14               & Float'Image (Sqrt (X)));
15
16     X := e;
17     Put_Line ("Natural log of "
18               & Float'Image (X)
19               & " is "
20               & Float'Image (Log (X)));
21
22     X := 10.0 ** 6.0;
23     Put_Line ("Log_10      of "
24               & Float'Image (X)
25               & " is "
26               & Float'Image (Log (X, 10.0)));
27
28     X := 2.0 ** 8.0;
29     Put_Line ("Log_2      of "
30               & Float'Image (X)
31               & " is "
32               & Float'Image (Log (X, 2.0)));
33
34     X := Pi;
```

(continues on next page)

(continued from previous page)

```

35 Put_Line ("Cos      of "
36      & Float'Image (X)
37      & " is "
38      & Float'Image (Cos (X)));
39
40 X := -1.0;
41 Put_Line ("Arccos      of "
42      & Float'Image (X)
43      & " is "
44      & Float'Image (Arccos (X)));
45 end Show_Elem_Math;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Elem\_Math  
MD5: 17511d7e17cd98d4b6e49ad302d6dcb6

### Runtime output

```

Square root of 2.00000E+00 is 1.41421E+00
Natural log of 2.71828E+00 is 1.00000E+00
Log_10      of 1.00000E+06 is 6.00000E+00
Log_2      of 2.56000E+02 is 8.00000E+00
Cos        of 3.14159E+00 is -1.00000E+00
Arccos     of -1.00000E+00 is 3.14159E+00

```

Here we use the standard e and Pi constants from the Ada.Numerics package.

The Ada.Numerics.Elementary\_Functions package provides operations for the **Float** type. Similar packages are available for **Long\_Float** and **Long\_Long\_Float** types. For example, the Ada.Numerics.Long\_Elementary\_Functions package offers the same set of operations for the **Long\_Float** type. In addition, the Ada.Numerics.**Generic\_Elementary\_Functions** package is a generic version of the package that you can instantiate for custom floating-point types. In fact, the **Elementary\_Functions** package can be defined as follows:

```

package Elementary_Functions is new
  Ada.Numerics.Generic_Elementary_Functions (Float);

```

## 23.2 Random Number Generation

The Ada.Numerics.Float\_Random package provides a simple random number generator for the range between 0.0 and 1.0. To use it, declare a generator G, which you pass to Random. For example:

Listing 325: show\_float\_random\_num.adb

```

1 with Ada.Text_IO;  use Ada.Text_IO;
2
3 with Ada.Numerics.Float_Random;
4 use Ada.Numerics.Float_Random;
5
6 procedure Show_Float_Random_Num is
7   G : Generator;
8   X : Uniformly_Distributed;
9 begin
10   Reset (G);
11

```

(continues on next page)

(continued from previous page)

```

12 Put_Line ("Some random numbers between "
13   & Float'Image
14   (Uniformly_Distributed'First)
15   & " and "
16   & Float'Image
17   (Uniformly_Distributed'Last)
18   & ":");
19 for I in 1 .. 15 loop
20   X := Random (G);
21   Put_Line (Float'Image (X));
22 end loop;
23 end Show_Float_Random_Num;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Float\_Random\_Num  
MD5: cf38ab00e27bad4309010e678113dd36

### Runtime output

```
Some random numbers between 0.00000E+00 and 1.00000E+00:
9.80034E-01
2.85511E-01
6.52899E-01
2.38608E-02
2.95568E-02
9.90054E-01
8.09696E-01
1.17431E-02
3.45056E-01
5.46639E-01
8.32065E-01
6.83566E-01
4.59634E-01
7.73473E-01
7.50087E-01
```

The standard library also includes a random number generator for discrete numbers, which is part of the Ada.Numerics.Discrete\_Random package. Since it's a generic package, you have to instantiate it for the desired discrete type. This allows you to specify a range for the generator. In the following example, we create an application that displays random integers between 1 and 10:

Listing 326: show\_discrete\_random\_num.adb

```

1 with Ada.Text_Io;  use Ada.Text_Io;
2 with Ada.Numerics.Discrete_Random;
3
4 procedure Show_Discrete_Random_Num is
5
6   subtype Random_Range is Integer range 1 .. 10;
7
8   package R is new
9     Ada.Numerics.Discrete_Random (Random_Range);
10  use R;
11
12  G : Generator;
13  X : Random_Range;
14 begin
15   Reset (G);

```

(continues on next page)

(continued from previous page)

```
17 Put_Line ("Some random numbers between "
18     & Integer'Image (Random_Range'First)
19     & " and "
20     & Integer'Image (Random_Range'Last)
21     & ":");

22 for I in 1 .. 15 loop
23     X := Random (G);
24     Put_Line (Integer'Image (X));
25 end loop;
26 end Show_Discrete_Random_Num;
```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Discrete\_Random\_Num  
MD5: 892f6525477f9a2c56f88885de011fba

### Runtime output

```
Some random numbers between 1 and 10:
4
4
10
8
8
2
2
2
4
5
7
2
1
8
5
```

Here, package R is instantiated with the Random\_Range type, which has a constrained range between 1 and 10. This allows us to control the range used for the random numbers. We could easily modify the application to display random integers between 0 and 20 by changing the specification of the Random\_Range type. We can also use floating-point or fixed-point types.

## 23.3 Complex Types

The Ada.Numerics.Complex\_Types package provides support for complex number types and the Ada.Numerics.Complex\_Elementary\_Functions package provides support for common operations on complex number types, similar to the Ada.Numerics.Elementary\_Functions package. Finally, you can use the Ada.Text\_Io.Complex\_Io package to perform I/O operations on complex numbers. In the following example, we declare variables of the Complex type and initialize them using an aggregate:

Listing 327: show\_elem\_math.adb

```
1 with Ada.Text_Io;  use Ada.Text_Io;
2 with Ada.Numerics; use Ada.Numerics;
3
4 with Ada.Numerics.Complex_Types;
5 use Ada.Numerics.Complex_Types;
```

(continues on next page)

(continued from previous page)

```

6
7 with Ada.Numerics.Complex_Elementary_Functions;
8 use Ada.Numerics.Complex_Elementary_Functions;
9
10 with Ada.Text_Io.Complex_Io;
11
12 procedure Show_Elem_Math is
13
14     package C_Io is new
15         Ada.Text_Io.Complex_Io (Complex_Types);
16     use C_Io;
17
18     X, Y : Complex;
19     R, Th : Float;
20
21 begin
22     X := (2.0, -1.0);
23     Y := (3.0, 4.0);
24
25     Put (X);
26     Put (" * ");
27     Put (Y);
28     Put (" is ");
29     Put (X * Y);
30     New_Line;
31     New_Line;
32
33     R := 3.0;
34     Th := Pi / 2.0;
35     X := Compose_From_Polar (R, Th);
36     -- Alternatively:
37     -- X := R * Exp ((0.0, Th));
38     -- X := R * e ** Complex'(0.0, Th);
39
40     Put ("Polar form:      "
41           & Float'Image (R) & " * e**(" & Float'Image (Th) & ")");
42     New_Line;
43
44     Put ("Modulus      of ");
45     Put (X);
46     Put (" is ");
47     Put (Float'Image (abs X));
48     New_Line;
49
50     Put ("Argument      of ");
51     Put (X);
52     Put (" is ");
53     Put (Float'Image (Argument (X)));
54     New_Line;
55     New_Line;
56
57     Put ("Sqrt      of ");
58     Put (X);
59     Put (" is ");
60     Put (Sqrt (X));
61     New_Line;
62 end Show_Elem_Math;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Elem\_Math

(continues on next page)

(continued from previous page)

MD5: 73dfc0021c804fa1c5727e272546f172

### Runtime output

```
( 2.00000E+00, -1.00000E+00) * ( 3.00000E+00, 4.00000E+00) is ( 1.00000E+01, 5.  
00000E+00)  
  
Polar form: 3.00000E+00 * e**(i * 1.57080E+00)  
Modulus of (-1.31134E-07, 3.00000E+00) is 3.00000E+00  
Argument of (-1.31134E-07, 3.00000E+00) is 1.57080E+00  
  
Sqrt of (-1.31134E-07, 3.00000E+00) is ( 1.22474E+00, 1.22474E+00)
```

As we can see from this example, all the common operators, such as `*` and `+`, are available for complex types. You also have typical operations on complex numbers, such as `Argument` and `Exp`. In addition to initializing complex numbers in the cartesian form using aggregates, you can do so from the polar form by calling the `Compose_From_Polar` function.

The `Ada.Numerics.Complex_Types` and `Ada.Numerics.Complex_Elementary_Functions` packages provide operations for the `Float` type. Similar packages are available for `Long_Float` and `Long_Long_Float` types. In addition, the `Ada.Numerics.Generic_Complex_Types` and `Ada.Numerics.Generic_Complex_Elementary_Functions` packages are generic versions that you can instantiate for custom or pre-defined floating-point types. For example:

```
with Ada.Numerics.Generic_Complex_Types;  
with Ada.Numerics.Generic_Complex_Elementary_Functions;  
with Ada.Text_Io.Complex_Io;  
  
procedure Show_Elem_Math is  
  
  package Complex_Types is new  
    Ada.Numerics.Generic_Complex_Types (Float);  
  use Complex_Types;  
  
  package Elementary_Functions is new  
    Ada.Numerics.Generic_Complex_Elementary_Functions  
    (Complex_Types);  
  use Elementary_Functions;  
  
  package C_Io is new Ada.Text_Io.Complex_Io  
    (Complex_Types);  
  use C_Io;  
  
  X, Y : Complex;  
  R, Th : Float;
```

## 23.4 Vector and Matrix Manipulation

The `Ada.Numerics.Real_Arrays` package provides support for vectors and matrices. It includes common matrix operations such as inverse, determinant, eigenvalues in addition to simpler operators such as matrix addition and multiplication. You can declare vectors and matrices using the `Real_Vector` and `Real_Matrix` types, respectively.

The following example uses some of the operations from the `Ada.Numerics.Real_Arrays` package:

Listing 328: show\_matrix.adb

```

1  with Ada.Text_Io;  use Ada.Text_Io;
2
3  with Ada.Numerics.Real_Arrays;
4  use Ada.Numerics.Real_Arrays;
5
6  procedure Show_Matrix is
7
8      procedure Put_Vector (V : Real_Vector) is
9  begin
10         Put ("  ");
11         for I in V'Range loop
12             Put (Float'Image (V (I)) & " ");
13         end loop;
14         Put_Line ("");
15     end Put_Vector;
16
17     procedure Put_Matrix (M : Real_Matrix) is
18  begin
19         for I in M'Range (1) loop
20             Put ("  ");
21             for J in M'Range (2) loop
22                 Put (Float'Image (M (I, J)) & " ");
23             end loop;
24             Put_Line ("");
25         end loop;
26     end Put_Matrix;
27
28     V1      : Real_Vector := (1.0, 3.0);
29     V2      : Real_Vector := (75.0, 11.0);
30
31     M1      : Real_Matrix :=
32         ((1.0, 5.0, 1.0),
33          (2.0, 2.0, 1.0));
34     M2      : Real_Matrix :=
35         ((31.0, 11.0, 10.0),
36          (34.0, 16.0, 11.0),
37          (32.0, 12.0, 10.0),
38          (31.0, 13.0, 10.0));
39     M3      : Real_Matrix := ((1.0, 2.0),
40                               (2.0, 3.0));
41
42 begin
43     Put_Line ("V1");
44     Put_Vector (V1);
45     Put_Line ("V2");
46     Put_Vector (V2);
47     Put_Line ("V1 * V2 =");
48     Put_Line ("  " & Float'Image (V1 * V2));
49     Put_Line ("V1 * V2 =");
50     Put_Matrix (V1 * V2);
51     New_Line;
52
53     Put_Line ("M1");
54     Put_Matrix (M1);
55     Put_Line ("M2");
56     Put_Matrix (M2);
57     Put_Line ("M2 * Transpose(M1) =");
58     Put_Matrix (M2 * Transpose (M1));
59     New_Line;
60

```

(continues on next page)

(continued from previous page)

```

61  Put_Line ("M3");
62  Put_Matrix (M3);
63  Put_Line ("Inverse (M3) =");
64  Put_Matrix (Inverse (M3));
65  Put_Line ("abs Inverse (M3) =");
66  Put_Matrix (abs Inverse (M3));
67  Put_Line ("Determinant (M3) =");
68  Put_Line (""
69      & Float'Image (Determinant (M3)));
70  Put_Line ("Solve (M3, V1) =");
71  Put_Vector (Solve (M3, V1));
72  Put_Line ("Eigenvalues (M3) =");
73  Put_Vector (Eigenvalues (M3));
74  New_Line;
75 end Show_Matrix;

```

### Code block metadata

Project: Courses.Intro\_To\_Ada.Standard\_Library.Show\_Matrix  
MD5: c9df45a742a42bd47e03fbf2d0282238

### Runtime output

```

V1
  ( 1.00000E+00  3.00000E+00 )
V2
  ( 7.50000E+01  1.10000E+01 )
V1 * V2 =
  1.08000E+02
V1 * V2 =
  ( 7.50000E+01  1.10000E+01 )
  ( 2.25000E+02  3.30000E+01 )

M1
  ( 1.00000E+00  5.00000E+00  1.00000E+00 )
  ( 2.00000E+00  2.00000E+00  1.00000E+00 )
M2
  ( 3.10000E+01  1.10000E+01  1.00000E+01 )
  ( 3.40000E+01  1.60000E+01  1.10000E+01 )
  ( 3.20000E+01  1.20000E+01  1.00000E+01 )
  ( 3.10000E+01  1.30000E+01  1.00000E+01 )
M2 * Transpose(M1) =
  ( 9.60000E+01  9.40000E+01 )
  ( 1.25000E+02  1.11000E+02 )
  ( 1.02000E+02  9.80000E+01 )
  ( 1.06000E+02  9.80000E+01 )

M3
  ( 1.00000E+00  2.00000E+00 )
  ( 2.00000E+00  3.00000E+00 )
Inverse (M3) =
  (-3.00000E+00  2.00000E+00 )
  ( 2.00000E+00 -1.00000E+00 )
abs Inverse (M3) =
  ( 3.00000E+00  2.00000E+00 )
  ( 2.00000E+00  1.00000E+00 )
Determinant (M3) =
  -1.00000E+00
Solve (M3, V1) =
  ( 3.00000E+00 -1.00000E+00 )
Eigenvalues (M3) =

```

(continues on next page)

(continued from previous page)

( 4.23607E+00 -2.36068E-01 )

Matrix dimensions are automatically determined from the aggregate used for initialization when you don't specify them. You can, however, also use explicit ranges. For example:

```
M1      : Real_Matrix (1 .. 2, 1 .. 3) :=
      ((1.0, 5.0, 1.0),
       (2.0, 2.0, 1.0));
```

The Ada.Numerics.Real\_Arrays package implements operations for the **Float** type. Similar packages are available for **Long\_Float** and **Long\_Long\_Float** types. In addition, the Ada.Numerics.**Generic**\_Real\_Arrays package is a generic version that you can instantiate with custom floating-point types. For example, the Real\_Arrays package can be defined as follows:

```
package Real_Arrays is new
  Ada.Numerics.Generic_Real_Arrays (Float);
```



APPENDICES

## 24.1 Appendix A: Generic Formal Types

The following tables contain examples of available formal types for generics:

Formal type	Actual type
Incomplete type <b>Format:</b> <code>type T;</code>	Any type
Discrete type <b>Format:</b> <code>type T is (&lt;&gt;);</code>	Any integer, modular or enumeration type
Range type <b>Format:</b> <code>type T is range &lt;&gt;;</code>	Any signed integer type
Modular type <b>Format:</b> <code>type T is mod &lt;&gt;;</code>	Any modular type
Floating-point type <b>Format:</b> <code>type T is digits &lt;&gt;;</code>	Any floating-point type
Binary fixed-point type <b>Format:</b> <code>type T is delta &lt;&gt;;</code>	Any binary fixed-point type
Decimal fixed-point type <b>Format:</b> <code>type T is delta &lt;&gt; digits &lt;&gt;;</code>	Any decimal fixed-point type
Definite nonlimited private type <b>Format:</b> <code>type T is private;</code>	Any nonlimited, definite type
Nonlimited Private type with discriminant <b>Format:</b> <code>type T (D : DT) is private;</code>	Any nonlimited type with discriminant
Access type <b>Format:</b> <code>type A is access T;</code>	Any access type for type T
Definite derived type <b>Format:</b> <code>type T is new B;</code>	Any concrete type derived from base type B
Limited private type <b>Format:</b> <code>type T is limited private;</code>	Any definite type, limited or not
Incomplete tagged type <b>Format:</b> <code>type T is tagged;</code>	Any concrete, definite, tagged type
Definite tagged private type <b>Format:</b> <code>type T is tagged private;</code>	Any concrete, definite, tagged type
Definite tagged limited private type <b>Format:</b> <code>type T is tagged limited private;</code>	Any concrete definite tagged type, limited or not
Definite abstract tagged private type <b>Format:</b> <code>type T is abstract tagged private;</code>	Any nonlimited, definite tagged type, abstract or concrete

continues on next page

Table 1 – continued from previous page

Formal type	Actual type
Definite abstract tagged limited private type <b>Format: type T is abstract tagged limited private;</b>	Any definite tagged type, limited or not, abstract or concrete
Definite derived tagged type <b>Format: type T is new B with private;</b>	Any concrete tagged type derived from base type B
Definite abstract derived tagged type <b>Format: type T is abstract new B with private;</b>	Any tagged type derived from base type B abstract or concrete
Array type <b>Format: type A is array (R) of T;</b>	Any array type with range R containing elements of type T
Interface type <b>Format: type T is interface;</b>	Any interface type T
Limited interface type <b>Format: type T is limited interface;</b>	Any limited interface type T
Task interface type <b>Format: type T is task interface;</b>	Any task interface type T
Synchronized interface type <b>Format: type T is synchronized interface;</b>	Any synchronized interface type T
Protected interface type <b>Format: type T is protected interface;</b>	Any protected interface type T
Derived interface type <b>Format: type T is new B and I with private;</b>	Any type T derived from base type B and interface I
Derived type with multiple interfaces <b>Format: type T is new B and I1 and I2 with private;</b>	Any type T derived from base type B and interfaces I1 and I2
Abstract derived interface type <b>Format: type T is abstract new B and I with private;</b>	Any type T derived from abstract base type B and interface I
Limited derived interface type <b>Format: type T is limited new B and I with private;</b>	Any type T derived from limited base type B and limited interface I
Abstract limited derived interface type <b>Format: type T is abstract limited new B and I with private;</b>	Any type T derived from abstract limited base type B and limited interface I
Synchronized interface type <b>Format: type T is synchronized new SI with private;</b>	Any type T derived from synchronized interface SI
Abstract synchronized interface type <b>Format: type T is abstract synchronized new SI with private;</b>	Any type T derived from synchronized interface SI

### 24.1.1 Indefinite version

Many of the examples above can be used for formal indefinite types:

Formal type	Actual type
Indefinite incomplete type <b>Format:</b> <code>type T (&lt;&gt;);</code>	Any type
Indefinite nonlimited private type <b>Format:</b> <code>type T (&lt;&gt;) is private;</code>	Any nonlimited type indefinite or definite
Indefinite limited private type <b>Format:</b> <code>type T (&lt;&gt;) is limited private;</code>	Any type, limited or not, indefinite or definite
Incomplete indefinite tagged private type <b>Format:</b> <code>type T (&lt;&gt;) is tagged;</code>	Any concrete tagged type, indefinite or definite
Indefinite tagged private type <b>Format:</b> <code>type T (&lt;&gt;) is tagged private;</code>	Any concrete, nonlimited tagged type, indefinite or definite
Indefinite tagged limited private type <b>Format:</b> <code>type T (&lt;&gt;) is tagged limited private;</code>	Any concrete tagged type, limited or not, indefinite or definite
Indefinite abstract tagged private type <b>Format:</b> <code>type T (&lt;&gt;) is abstract tagged private;</code>	Any nonlimited tagged type, indefinite or definite, abstract or concrete
Indefinite abstract tagged limited private type <b>Format:</b> <code>type T (&lt;&gt;) is abstract tagged limited private;</code>	Any tagged type, limited or not, indefinite or definite abstract or concrete
Indefinite derived tagged type <b>Format:</b> <code>type T (&lt;&gt;) is new B with private;</code>	Any tagged type derived from base type B, indefinite or definite
Indefinite abstract derived tagged type <b>Format:</b> <code>type T (&lt;&gt;) is abstract new B with private;</code>	Any tagged type derived from base type B, indefinite or definite abstract or concrete

The same examples could also contain discriminants. In this case, `(<>)` is replaced by a list of discriminants, e.g.: `(D: DT)`.

## 24.2 Appendix B: Containers

The following table shows all containers available in Ada, including their versions (standard, bounded, unbounded, indefinite):

Category	Container	Std	Bounded	Unbounded	Indefinite
Vector	Vectors	Y	Y		Y
List	Doubly Linked Lists	Y	Y		Y
Map	Hashed Maps	Y	Y		Y
Map	Ordered Maps	Y	Y		Y
Set	Hashed Sets	Y	Y		Y
Set	Ordered Sets	Y	Y		Y
Tree	Multiway Trees	Y	Y		Y
Generic	Holders				Y
Queue	Synchronized Queue Interfaces	Y			
Queue	Synchronized Queues		Y	Y	
Queue	Priority Queues		Y	Y	

### Note

To get the correct container name, replace the whitespace by `_` in the names above. (For example, Hashed Maps becomes Hashed\_Maps.)

The following table presents the prefixing applied to the container name that depends on its version. As indicated in the table, the standard version does not have a prefix associated with it.

Version	Naming prefix
Std	
Bounded	Bounded_
Unbounded	Unbounded_
Indefinite	Indefinite_